Mit *Mathematica* durchs Abitur

Flächenstück unter einer Kurve

Aufgabe:

Gegeben ist für x>0 die Funktion $f[x] = \frac{x}{8} + \frac{2}{x}$. Zeichnen Sie den Graphen der Kurve.

In die Fläche zwischen Kurve und *x*−Achse ist ein Streifen mit der Breite 3 parallel zur *y*−Achse so einzufügen, dass seine Fläche möglichst klein wird. Berechnen Sie den Ort der beiden Parallelen und die resultierende minimale Fläche.

```
ClearAll["Global`*"]
f[x_] := x/8 + 2/x
```

Um ein Bild der Situation mit **Plot**, **FillingStyle** und verschiedenen Streifen erstellen, definieren wir die folgende Funktion. Sie erzeugt den Plot zweier Funktionen, die in der Komposition genau das gewünschte Bild ergeben. Mit den **Filling**−Optionen wird genau die Fläche zwischen den beiden Kurven gefärbt. **PlotRange** fixiert den Wertebereich auf einen für unsere Zwecke sinnvollen Bereich.

```
fplot[u_] := Plot[{f[x], Piecewise[{{f[x], u < x < u + 3}}]}, {x, 0, 11},
  Filling → {2 -> Axis}, FillingStyle → Blue, PlotRange → {0, 5}]

fplot[1]
```



Und so lässt sich daraus in *Mathematica* 6.0 eine Animation erstellen.

```
Animate[fplot[u], {u, .4, 7, .2},
 AnimationDirection -> ForwardBackward, DefaultDuration -> 10]
```

Die eigentliche Aufgabenstellung lässt sich mit folgendem "Gewalt"−Ansatz lösen: Berechne den Flächeninhalt A[u]= $\int_u^{u+3} f[x]\,dx$ der zu untersuchenden Figur in Abhängigkeit von der Abszisse *u* der linken unteren Ecke, bestimme die Nullstellen der Ableitung von A[u] und überprüfe die üblichen Extremwert−Konditionen.

A[u_] = Assuming $\left[u > 0, \int_u^{u+3} f[x]\,dx\right]$

$$\frac{9}{16}+\frac{3u}{8}+2\,\mathrm{Log}\left[\frac{3+u}{u}\right]$$

Die Eingabe kann natürlich auch in der eindinemsionalen Input−Notation als **Integrate[f[x],{x,u,u+3}]** erfolgen. In *Mathematica* 6.0 muss die zusätzliche **Assuming**−Konstruktion verwendet werden, weil das CAS "zu schlau" ist und uns erzählt, dass für irgendwelche komplexe Werte von *u* Böses passieren kann.

```
Integrate[f[x], {x, u, u + 3}]
```

$$\mathrm{If}\left[\mathrm{Re}[u] > 0 \,||\, (\mathrm{Im}[u] \neq 0 \,\&\&\, \mathrm{Re}[u] == 0),\ \frac{1}{16}(9+6u-32\,\mathrm{Log}[u]+32\,\mathrm{Log}[3+u]),\ \mathrm{Integrate}\left[\frac{2}{x}+\frac{x}{8}, \{x, u, 3+u\}, \mathrm{Assumptions} \to\, !\left(\mathrm{Re}[u] > 0 \,||\, (\mathrm{Im}[u] \neq 0 \,\&\&\, \mathrm{Re}[u] == 0)\right)\right]\right]$$

```
dflaeche = D[A[u], u] // Together
```

$$\frac{3(-16+3u+u^2)}{8u(3+u)}$$

```
sol = Solve[dflaeche == 0]
```

$$\left\{\left\{u \to \frac{1}{2}\left(-3-\sqrt{73}\right)\right\}, \left\{u \to \frac{1}{2}\left(-3+\sqrt{73}\right)\right\}\right\}$$

```
sol // N
```

$\{\{u \to -5.772\}, \{u \to 2.772\}\}$

Nur die positive Lösung ist für die Aufgabenstellung relevant. Fassen wir zum Schluss noch alle zu berechnenden Stücke in einer Liste zusammen.

```
{u, A[u]} /. sol[[2]] // N
```

$\{2.772, 3.0689\}$

Eine einfachere mathematische Lösung (warum?) geht so:

```
sol = Solve[f[u + 3] - f[u] == 0, u]
```

$$\left\{\left\{u \to \frac{1}{2}\left(-3-\sqrt{73}\right)\right\}, \left\{u \to \frac{1}{2}\left(-3+\sqrt{73}\right)\right\}\right\}$$

Eine Eliminationsaufgabe

Aufgabe:
Bringen Sie die in Parameterform gegebene Funktion (x,y) = (2Cos[t],Cos[3t]) in explizite Form und zeichnen Sie den Graphen.
Bestimmen Sie den Inhalt der Flächenstücke, die zwischen der Kurve und der *x*−Achse eingeschlossen werden.

Auch hier erzeugen wir uns erst einmal ein Bild.

```
ClearAll["Global`*"]
ParametricPlot[{2 Cos[t], Cos[3 t]}, {t, 0, π}]
```



Zur Lösung einer Eliminationsaufgabe bietet sich das Kommando **Eliminate** an.
Wichtig: bei Eliminate muss == und nicht = verwendet werden; trotzdem ist das Ergebnis nicht das, was wir wollten.

```
Eliminate[{x == 2 Cos[t], y == Cos[3 t]}, t]
```

```
Eliminate::ifun :
 Inverse functions are being used by Eliminate, so some solutions
   may not be found; use Reduce for complete solution information.
```

$$\text{ArcCos}[y] == 3\,\text{ArcCos}\left[\frac{x}{2}\right]$$

Dabei hat *Mathematica* genau das gemacht, was wir gefordert haben: beide Gleichungen nach *t* aufgelöst und gleichgesetzt. Wir wollten aber eigentlich Cos[t] eliminiert haben.

```
Eliminate[{x == 2 Cos[t], y == Cos[3 t]}, Cos[t]]
```

```
Cos[3 t] == y
```

Das ging leider auch daneben: *Mathematica* weiß von sich aus nicht, dass Cos[3t] auch was mit Cos[t] zu tun hat.

```
Cos[3 t] // TrigExpand
```

$\mathrm{Cos}[t]^3 - 3\,\mathrm{Cos}[t]\,\mathrm{Sin}[t]^2$

```
% /. {Sin[x_]^2 -> 1 - Cos[x]^2}
```

$\mathrm{Cos}[t]^3 - 3\,\mathrm{Cos}[t]\left(1 - \mathrm{Cos}[t]^2\right)$

```
y0 = % // Expand
```

$-3\,\mathrm{Cos}[t] + 4\,\mathrm{Cos}[t]^3$

Mit der so umgeformten rechten Seite der Bestimmungsgleichung für y und Elimination nach Cos[t] statt t gelingt nun die Elimination.

```
gl = Eliminate[{x == 2 Cos[t], y == y0}, Cos[t]]
```

$2\,y == -3\,x + x^3$

Auch bei Elimination nach t erhalten wir dasselbe Ergebnis, weil die Gleichungen polynomial im einzigen Kern Cos[t] sind. Einzig eine zusätzliche Warnung wird ausgegeben.

```
Eliminate[{x == 2 Cos[t], y == y0}, t]
```

```
Eliminate::ifun :
 Inverse functions are being used by Eliminate, so some solutions
   may not be found; use Reduce for complete solution information.
```

$2\,y == -3\,x + x^3$

Reduce liefert die Lösung der Aufgabe, in einige zusätzliche Terme eingebettet, übrigens ohne vorbereitende Umformungen.

```
red = Reduce[{x == 2 Cos[t], y == Cos[3 t]}, t]
```

$$(C[1] \in \mathrm{Integers}\ \&\&\ x == -2\ \&\&\ y == -1\ \&\&\ t == \pi + 2\,\pi\,C[1])\ ||$$
$$\left(2 + x \neq 0\ \&\&\ y == \frac{1}{2}\left(-3\,x + x^3\right)\ \&\&\ C[1] \in \mathrm{Integers}\ \&\&\right.$$
$$\left.\left(t == -2\,\mathrm{ArcTan}\left[\frac{\sqrt{2-x}}{\sqrt{2+x}}\right] + 2\,\pi\,C[1]\ ||\ t == 2\,\mathrm{ArcTan}\left[\frac{\sqrt{2-x}}{\sqrt{2+x}}\right] + 2\,\pi\,C[1]\right)\right)$$

```
Reduce[{x == 2 Cos[t], y == y0}, t]
```

$$C[1] \in \mathrm{Integers}\ \&\&\ y == \frac{1}{2}\left(-3\,x + x^3\right)\ \&\&\ \left(t == -\mathrm{ArcCos}\left[\frac{x}{2}\right] + 2\,\pi\,C[1]\ ||\ t == \mathrm{ArcCos}\left[\frac{x}{2}\right] + 2\,\pi\,C[1]\right)$$

Reduce mit Cos[t] als Eliminationsziel stellt x in Abhängigkeit von y als **Root**−Ausdruck dar. y muss mit in die Liste der Variablen aufgenommen werden, nach denen aufgelöst wird, um y durch x auszudrücken und zu obigem Ergebnis zu gelangen.

```
Reduce[{x == 2 Cos[t], y == y0}, {y, Cos[t]}]
```

$y == \frac{1}{2}\left(-3x + x^3\right) \ \&\& \ \mathrm{Cos}[t] == \frac{x}{2}$

Nun verwandeln wir die Gleichung in einen expliziten Funktionszusammenhang y=y(x)

```
lsg = Solve[gl, y]
```

$\left\{\left\{y \to \frac{1}{2}\left(-3x + x^3\right)\right\}\right\}$

```
y[x_] = y /. lsg[[1]];
```

```
Plot[y[x], {x, -2, 2}]
```



Als Wertebereich für *x* wurde der gleiche Bereich wie oben im ersten Plot verwendet. Die Schnittpunkte mit den Achsen ergeben sich als

```
Solve[y[x] == 0, x]
```

$\left\{\{x \to 0\}, \left\{x \to -\sqrt{3}\right\}, \left\{x \to \sqrt{3}\right\}\right\}$

und die gesuchte Fläche (beachten Sie die Vorzeichenregeln) zu

$$2\int_{-\sqrt{3}}^{0} y[x]\, dx$$

$$\frac{9}{4}$$

Komplexe Wurzeln

Aufgabe:

Lösen Sie die Gleichung $z^3 = \frac{(3+4\,i)^5}{(4-3\,i)^2}$ im Bereich der komplexen Zahlen und stellen Sie die Lösungen in der Gaußschen Zahlenebene grafisch dar.

```
ClearAll["Global`*"]
lsg = Solve[z^3 == (3 + 4 I)^5/(4 - 3 I)^2, z]
```

$\{\{z\to(117-44\,i)^{1/3}\},\ \{z\to-(-1)^{1/3}\,(117-44\,i)^{1/3}\},\ \{z\to(-1)^{2/3}\,(117-44\,i)^{1/3}\}\}$

Das Ergebnis sollte in der Form $a + b^2\, i$ ausgegeben werden.

```
lsg // Simplify
```

$\{\{z\to(117-44\,i)^{1/3}\},\ \{z\to-3-4\,i\},\ \{z\to(-1)^{2/3}\,(117-44\,i)^{1/3}\}\}$

Auch **FullSimplify** führt zu keinem besseren Ergebnis.

```
lsg // RootReduce
```

$\{\{z\to\text{Root}[625-150\,\#1+11\,\#1^2-6\,\#1^3+\#1^4\,\&,\,3]\},$
$\{z\to-3-4\,i\},\ \{z\to\text{Root}[625-150\,\#1+11\,\#1^2-6\,\#1^3+\#1^4\,\&,\,2]\}\}$

Die beiden anderen Wurzeln sind algebraische Zahlen vom Grad 4. Mit der Root−Notation kann *Mathematica* besser rechnen.

```
u = z /. lsg[[1]] // RootReduce
```

$\text{Root}[625-150\,\#1+11\,\#1^2-6\,\#1^3+\#1^4\,\&,\,3]$

Wir extrahieren das Minimalpolynom und bestimmen dessen Nullstellen. Zwei davon sind die nicht identifizierten Elemente aus *lsg*.

```
f = u[[1]][z]
```

$625-150\,z+11\,z^2-6\,z^3+z^4$

```
lsg1 = Solve[f == 0, z] // FullSimplify
```

$\left\{\left\{z\to\left(\frac{3}{2}+2\,i\right)\left(1-i\sqrt{3}\right)\right\},\ \left\{z\to\left(2+\frac{3\,i}{2}\right)\left(-i+\sqrt{3}\right)\right\},\right.$
$\left.\left\{z\to\left(-2-\frac{3\,i}{2}\right)\left(i+\sqrt{3}\right)\right\},\ \left\{z\to\left(\frac{3}{2}+2\,i\right)\left(1+i\sqrt{3}\right)\right\}\right\}$

```
z^3 /. lsg1 // Expand
```

{117 − 44 ⅈ, 117 + 44 ⅈ, 117 + 44 ⅈ, 117 − 44 ⅈ}

Wir transformieren *lsg* in numerische Näherungswerte, um ein Bild zu erstellen.

```
nlsg = N[lsg]
```

{{z → 4.9641 − 0.598076 ⅈ}, {z → −3. − 4. ⅈ}, {z → −1.9641 + 4.59808 ⅈ}}

```
lst = {Re[z], Im[z]} /. nlsg
```

{{4.9641, −0.598076}, {−3., −4.}, {−1.9641, 4.59808}}

```
ListPlot[lst, PlotStyle → PointSize[0.04], PlotRange -> {-6, 6}]
```



■ Eine Extremwertaufgabe

Aufgabe:
Der Ellipse $9 x^2 + 16 y^2 == 144$ soll ein Rechteck maximaler Fläche einbeschrieben werden, dessen Seiten parallel zu den Koordinatenachsen sind.

Bestimmen Sie
a) die Abmessungen dieses Rechtecks und
b) das Volumen des Ellipsoids und des Zylinders, die bei Drehung dieser Figueren um die *x*−Achse entstehen.

Ein Bild der Situation erstellen. Die Ellipse:

```
ClearAll["Global`*"]
ellipse = 9 x^2 + 16 y^2 == 144
```

$9 x^2 + 16 y^2 == 144$

Die Ellipse kann nun mit ContourPlot gezeichnet werden. Die Standardeinstellungen (mit Frame, ohne Achsen, nicht maßstäblich) sind geändert.

```
e1 = ContourPlot[Evaluate[ellipse], {x, -6, 6}, {y, -4, 4},
  Frame → False, Axes → True, AspectRatio → Automatic]
```



Berechnung der Eckpunkte $(\pm x_0, \pm y_0)$ des einbeschriebenen Rechtecks bei vogegebenem x_0:

```
lsg = Solve[ellipse, y]
```

$$\left\{\left\{y \to -\frac{3}{4}\sqrt{16 - x^2}\right\}, \left\{y \to \frac{3\sqrt{16 - x^2}}{4}\right\}\right\}$$

```
y0 = y /. lsg[[2]] /. x → x0
```

$$\frac{3}{4}\sqrt{16 - x_0^2}$$

Beachten Sie, dass y_0 keine Funktion $y_0[x_0]$, sondern ein Ausdruck ist, welcher die Variable x_0 enthält. Diese kann durch ein Subsitutionskommando belegt werden.
Alles in einer gemeinsamen Grafik darstellen ...

```
rechteck = Line[{{x0, y0}, {-x0, y0}, {-x0, -y0}, {x0, -y0}, {x0, y0}}];
Show[e1, Graphics[rechteck /. x0 → 1.6], Graphics[rechteck /. x0 → 3.2]]
```



... oder eine Animation erstellen:

```
Animate[Show[e1, Graphics[rechteck /. x0 → u]], {u, 0.3, 3.8, 0.2},
 AnimationDirection → ForwardBackward]
```

Und nun die analytische Lösung der Aufgabe a).

```
flaeche = 4 x y /. lsg[[2]]
```

$3 x \sqrt{16 - x^2}$

```
extrema = Solve[D[flaeche, x] == 0, x]
```

$\left\{\left\{x \to -2\sqrt{2}\right\}, \left\{x \to 2\sqrt{2}\right\}\right\}$

```
flaeche /. extrema[[2]]
```

24

Jetzt beginnt der zweite Rechenabschnitt − die Berechnung der Volumens des Rotationsellipsoids und des Zylinders.
Die Begrenzungskurve

```
f1[x_] = y /. lsg[[2]]
```

$\frac{3\sqrt{16 - x^2}}{4}$

Die Grenzen

```
Solve[ellipse /. y → 0, x]
```

{{x → −4}, {x → 4}}

$$\text{ellipsoidVolumen} = \pi \int_{-4}^{4} f_1[x]^2 \, dx$$

48π

$$\text{zylinderVolumen} = 2\pi x y^2 \;/.\; \text{lsg}[[2]] \;/.\; \text{extrema}[[2]]$$

$18\sqrt{2}\,\pi$

■ Analytische Geometrie

Aufgabe:
Der Kreis $x^2 + y^2 + 6y - 91 == 0$ und die Kurve $y = a x^2 + b$ schneiden einander in $P(6; y)$, $y > 0$, unter einem Winkel von 90 Grad.

a) Berechnen Sie *a* und *b*.
b) Die Schnittfläche der beiden Kurven rotiert um die *y*−Achse.
Berechnen Sie das Volumen des entstehenden Rotationskörpers.

```
ClearAll["Global`*"]
lsg = Solve[kreis = x^2 + y^2 + 6 y - 91 == 0, y]
```

$\left\{\left\{y \to -3 - \sqrt{100 - x^2}\right\}, \left\{y \to -3 + \sqrt{100 - x^2}\right\}\right\}$

```
f1[x_] = y /. lsg[[2]]
```

$-3 + \sqrt{100 - x^2}$

Die Koordinaten des Punktes *P* ergeben sich nun durch Einsetzen.

```
{6, f1[6]}
```

{6, 5}

```
f[x_] = a x^2 + b
```

$b + a x^2$

Die beiden Parameter *a* und *b* ergeben sich aus den Bedingungen f1[6]== f[6] (beide Kurven gehen durch *P*) und f1'[6]·f'[6]==−1 (beide Kurven stehen in *P* aufeinander senkrecht).

```
sys = {f1[6] == f[6], f1'[6] * f'[6] == -1}
```

$\{5 == 36\,a + b,\ -9\,a == -1\}$

```
sol = Solve[sys, {a, b}]
```

$\left\{\left\{a \to \frac{1}{9},\ b \to 1\right\}\right\}$

```
f2[x_] = f[x] /. sol[[1]]
```

$1 + \frac{x^2}{9}$

```
Plot[{f1[x], f2[x]}, {x, -10, 10}, AspectRatio → Automatic]
```



Berechnung des Volumens des Körpers, der sich aus der Rotation der gemeinsamen Flächenanteile um die *y*−Achse ergibt (nach der Formel $V = \pi \int x[y]^2\, dy$). Die Integrationsgrenzen sind jeweils 5 und f1[0] bzw. f2[0] und 5.

Aus der Kreisgleichung wird dazu zunächst die Umkehrfunktion g1[y] extrahiert

```
sol1 = Solve[kreis, x]
```

$\left\{\left\{x \to -\sqrt{91 - 6\,y - y^2}\right\},\ \left\{x \to \sqrt{91 - 6\,y - y^2}\right\}\right\}$

```
g1[y_] = x /. sol1[[2]]
```

$\sqrt{91 - 6\,y - y^2}$

Der erste Teil des Volumenintegrals ergibt sich dann als Integral über g1 von y=5 bis y=f1[0], dem y−Wert des Schnittpunkts des Kreises mit der Ordinatenachse.

$$\text{vol1} = \pi \int_{5}^{\text{f1}[0]} \text{g1}[y]^2 \, dy$$

$$\frac{112\,\pi}{3}$$

Analog ergibt sich das zweite Teilvolumen und das Gesamtvolumen als Summe beider.

```
sol2 = Solve[y == f2[x], x]
```

$$\left\{\left\{x \to -3\sqrt{-1+y}\right\}, \left\{x \to 3\sqrt{-1+y}\right\}\right\}$$

```
g2[y_] = x /. sol2[[2]]
```

$$3\sqrt{-1+y}$$

$$\text{vol2} = \pi \int_{\text{f2}[0]}^{5} \text{g2}[y]^2 \, dy$$

$$72\,\pi$$

```
vol = vol1 + vol2
```

$$\frac{328\,\pi}{3}$$

```
l = Graphics3D[{Thick, Line[{{0, 0, -2}, {0, 0, 10}}]}];
r1 = RevolutionPlot3D[f1[t], {t, 0, 6}];
r2 = RevolutionPlot3D[f2[t], {t, 0, 6}];
Show[l, r1, r2, Boxed -> False, ViewAngle -> 20 °]
```