... Doing Mathematics by Computer

Grenzwerte und Reihen

```
ClearAll["Global`*"]
```

Grenzwertberechnung

$$\text{Limit}\left[\frac{\text{Sin}[x]}{x},\ x \to 0\right]$$

1

$$\text{Limit}\left[\left(\frac{4\,x}{2\,x+3}\right)^3,\ x \to \infty\right]$$

8

$$\text{Limit}\left[\left(1+\frac{1}{x}\right)^x,\ x \to \infty\right]$$

e

$$\text{Limit}\left[\frac{\text{Tan}[x]-\text{Sin}[x]}{x-\text{Sin}[x]},\ x \to 0\right]$$

3

Links− und rechtsseitiger Grenzwert

$$\text{Limit}\left[\text{Tan}[x],\ x \to \frac{\pi}{2}\right]$$

$-\infty$

$$\text{Limit}\left[\text{Tan}[x],\ x \to \frac{\pi}{2},\ \text{Direction} \to 1\right]$$

∞

```
Plot[Tan[x], {x, -3, 3}, PlotRange → {-10, 10}]
```



Grenzwert einer komplexen Funktion mit wesentlicher Singularität aus verschiedenen komplexen Richtungen

```
directions = Table[(1 + I)^n, {n, 8}]
```

{1 + ⅈ, 2 ⅈ, -2 + 2 ⅈ, -4, -4 - 4 ⅈ, -8 ⅈ, 8 - 8 ⅈ, 16}

$$\text{Limit}\left[\text{Exp}\left[\frac{1}{z}\right],\ z \to 0,\ \text{Direction} \to \#\right]\ \&\ /@\ \text{directions}$$

$\{0,\ e^{2\,i\,\text{Interval}[\{0,\pi\}]},\ \text{ComplexInfinity},\ \infty,\ \text{ComplexInfinity},\ e^{2\,i\,\text{Interval}[\{0,\pi\}]},\ 0,\ 0\}$

Grenzwert einer reellen Funktion, der nicht eindeutig bestimmt ist.

$$\text{Limit}\left[\text{Sin}\left[\frac{1}{x}\right],\ x \to 0\right]$$

```
Interval[{-1, 1}]
```

```
Plot[Sin[1/x], {x, -0.1, 0.1}]
```



Grenzwerte für Funktionen mit Parametern

$$\text{Limit}\left[\frac{-x^{-a}+x^{a}}{x-\frac{1}{x}},\ x\to 1\right]$$

a

$$\text{Limit}\left[a^{1/x},\ x\to\infty\right]$$

1

Mathematica's Fähigkeiten zur Grenzwertberechnung werden von Version zu Version besser. Diese Grenzwerte konnte es zum Beispiel in Version 4 noch nicht symbolisch bestimmen.

$$\text{Limit}\left[\frac{x^{n}}{E^{x}},\ x\to\infty\right]$$

0

$$\text{Limit}\left[\frac{x!}{x^{x}},\ x\to\infty\right]$$

0

$$\text{Limit}\left[\text{Tan}\left[\frac{\pi}{4}+x\right]^{\text{Cot}[2x]},\ x\to 0\right]$$

e

```
Limit[Zeta[s] - 1 / (s - 1), s → 1]
```

```
EulerGamma
```

Differentialquotient als Grenzwertberechnung bestimmen

$$\text{Limit}\left[\frac{\text{Sin}[x + dx] - \text{Sin}[x]}{dx}, dx \to 0\right]$$

```
Cos[x]
```

Grenzwerte mehrstelliger Funktionen sind nicht vorgesehen − sie sind auch mathematisch komplizierter.

$$f = \frac{\text{Sin}[x] + y - 1}{x} + \frac{\text{Cos}\left[y\,\frac{\pi}{2}\right] + x}{y};$$

```
Limit[f, {x, y} → {0, 1}]
```

```
General::ivar : {x, y} is not a valid variable.
```

$$\text{Limit}\left[\frac{x + \text{Cos}\left[\frac{\pi\,y}{2}\right]}{y} + \frac{-1 + y + \text{Sin}[x]}{x}, \{x, y\} \to \{0, 1\}\right]$$

```
Limit[f, x → 0, y → 1]
```

```
Limit::optx :
```

$$\text{Unknown option } y \text{ in Limit}\left[\frac{x + \text{Cos}\left[\frac{\pi\,y}{2}\right]}{y} + \frac{-1 + y + \text{Sin}[x]}{x}, x \to 0, y \to 1\right].$$

$$\text{Limit}\left[\frac{x + \text{Cos}\left[\frac{\pi\,y}{2}\right]}{y} + \frac{-1 + y + \text{Sin}[x]}{x}, x \to 0, y \to 1\right]$$

```
Limit[f, {x → 0, y → 1}]
```

$$\left\{\text{DirectedInfinity}[-1 + y], x + \frac{\text{Sin}[x]}{x}\right\}$$

Der Grenzwert kann als doppelter Grenzwert (erst nach x, dann nach y) berechnet werden. In diesem Beispiel kommt es allerdings auf die Reihenfolge an.

```
Limit[Limit[f, y → 1], x → 0]
```

```
1
```

```
Limit[Limit[f, x → 0], y → 1]
```

∞

Ein Bild schafft Aufklärung: Die zweistellige Funktion ist in der Nähe des Punktes (0,1) sehr unregelmäßig.

```
Plot3D[f, {x, -.01, .01}, {y, 0.99, 1.01}, Axes -> None]
```



Die Grenzwerte, die sich ergeben, wenn man sich dem Punkt aus unterschiedlichen Richtungen nähert, sind verschieden!

```
Limit[f /. {x → a ε, y → 1 + ε}, ε → 0]
```

$$1 + \frac{1}{a}$$

▪ Potenzreihen

Potenzreihen entwickeln

```
Series[Sin[x], {x, 0, 15}]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362\,880} - \frac{x^{11}}{39\,916\,800} + \frac{x^{13}}{6\,227\,020\,800} - \frac{x^{15}}{1\,307\,674\,368\,000} + O[x]^{16}$$

```
Series[Log[x], {x, 1, 4}]
```

$$(x-1) - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + O[x-1]^5$$

```
Series[1/(z^2 + 1), {z, I, 4}]
```

$$-\frac{i}{2\,(z-i)}+\frac{1}{4}+\frac{1}{8}\,i\,(z-i)-\frac{1}{16}\,(z-i)^2-\frac{1}{32}\,i\,(z-i)^3+\frac{1}{64}\,(z-i)^4+O[z-i]^5$$

```
f = Series[Sqrt[x], {x, 1, 4}]
```

$$1+\frac{x-1}{2}-\frac{1}{8}\,(x-1)^2+\frac{1}{16}\,(x-1)^3-\frac{5}{128}\,(x-1)^4+O[x-1]^5$$

```
f // InputForm
```

```
SeriesData[x, 1, {1, 1/2, −1/8, 1/16, −5/128}, 0, 5, 1]
```

```
{f⟦3, 5⟧, SeriesCoefficient[f, 4]}
```

$$\left\{-\frac{5}{128},\ -\frac{5}{128}\right\}$$

```
Series[Sin[x + y], {x, 0, 3}, {y, 0, 3}]
```

$$\left(y-\frac{y^3}{6}+O[y]^4\right)+\left(1-\frac{y^2}{2}+O[y]^4\right)x+\left(-\frac{y}{2}+\frac{y^3}{12}+O[y]^4\right)x^2+\left(-\frac{1}{6}+\frac{y^2}{12}+O[y]^4\right)x^3+O[x]^4$$

Grenzwertbestimmung für einen komplizierten Ausdruck.

```
f = Tan[Sin[x]] − Sin[Tan[x]]
```

```
−Sin[Tan[x]] + Tan[Sin[x]]
```

```
Limit[f/x^7, x → 0]
```

$$\frac{1}{30}$$

Der Grenzwert kann aus der zugehörigen Potenzreihe leicht abgelesen werden.

```
Series[f, {x, 0, 12}]
```

$$\frac{x^7}{30}+\frac{29\,x^9}{756}+\frac{1913\,x^{11}}{75\,600}+O[x]^{13}$$

Selbst bei so komplizierten Ausdrücken wie dem Folgenden wird die Reihenentwicklung ohne Verzögerung berechnet.

$$f = \left(\frac{x - \mathrm{Sin}[x]}{E^x - E^{\mathrm{Sin}[x]}}\right)^3;$$

```
Limit[f, x → 0]
```

```
1
```

```
Series[f, {x, 0, 5}]
```

$$1 - 3x + \frac{9x^2}{2} - \frac{17x^3}{4} + \frac{21x^4}{8} - \frac{73x^5}{80} + O[x]^6$$

In Reihenentwicklungen kann man keine Werte einsetzen. Das macht mathematisch ja auch wenig Sinn, denn eine Reihenentwicklung beschreibt ja das Aussehen einer Funktion in der Nähe eines Punkts.

```
f = Series[Sin[x], {x, 0, 6}]
f /. x → 1
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} + O[x]^7$$

```
SeriesData::ssdn : Attempt to evaluate
   a series at the number 1. Returning Indeterminate.
```

```
Indeterminate
```

Mit *Normal* kann man aber den polynomialen Teil der Reihe extrahieren und weiter verarbeiten.

```
f1 = Normal[f]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120}$$

```
f1 /. x → 1
```

$$\frac{101}{120}$$

```
Plot[{Sin[x], f1}, {x, -2 π, 2 π}, PlotStyle → {Automatic, Dashed}]
```



```
f2 = Normal[Series[Sin[x], {x, 0, 12}]]
```

$$x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040}+\frac{x^9}{362\,880}-\frac{x^{11}}{39\,916\,800}$$

```
Plot[{Sin[x], f1, f2}, {x, -2 π, 2 π}, PlotStyle → {Automatic, Dotted, Dashed}]
```



■ Mit Potenzreihen rechnen

```
Series[Cos[Tan[x]], {x, 0, 9}]
```

$$1-\frac{x^2}{2}-\frac{7\,x^4}{24}-\frac{97\,x^6}{720}-\frac{2063\,x^8}{40\,320}+O[x]^{10}$$

```
Cos[Tan[x]] + O[x, π]^8
```

$$1-\frac{1}{2}\,(x-\pi)^2-\frac{7}{24}\,(x-\pi)^4-\frac{97}{720}\,(x-\pi)^6+O[x-\pi]^8$$

```
x + 3 x^3 - 5 x^5 + x^6 + O[x, 1]^9
```

$$-9\,(x-1)-26\,(x-1)^2-27\,(x-1)^3-10\,(x-1)^4+(x-1)^5+(x-1)^6+O[x-1]^9$$

```
% // Normal
```

$$-9\,(-1+x)-26\,(-1+x)^2-27\,(-1+x)^3-10\,(-1+x)^4+(-1+x)^5+(-1+x)^6$$

```
% // Expand
```

$$x+3\,x^3-5\,x^5+x^6$$

```
ClearAll["Global`*"]
```

```
f = Sqrt[1 + 4 x];
s1 = Series[f, {x, 0, 8}]
```

$$1+2x-2x^2+4x^3-10x^4+28x^5-84x^6+264x^7-858x^8+O[x]^9$$

```
s2 = Series[Sin[Tan[x]], {x, 0, 4}]
```

$$x+\frac{x^3}{6}+O[x]^5$$

```
s1 + s2
```

$$1+3x-2x^2+\frac{25x^3}{6}-10x^4+O[x]^5$$

```
s1 s2
```

$$x+2x^2-\frac{11x^3}{6}+\frac{13x^4}{3}+O[x]^5$$

```
s = Series[Tan[Sin[x]] - Sin[Tan[x]], {x, 0, 9}]
```

$$\frac{x^7}{30}+\frac{29x^9}{756}+O[x]^{10}$$

s^2

$$\frac{x^{14}}{900}+\frac{29x^{16}}{11\,340}+O[x]^{17}$$

∂_x **s1**

$$2-4x+12x^2-40x^3+140x^4-504x^5+1848x^6-6864x^7+O[x]^8$$

∂_x **f**

$$\frac{2}{\sqrt{1+4x}}$$

```
Series[%, {x, 0, 8}]
```

$$2-4x+12x^2-40x^3+140x^4-504x^5+1848x^6-6864x^7+25\,740x^8+O[x]^9$$

$\int$ **s1** dx

$$x+x^2-\frac{2x^3}{3}+x^4-2x^5+\frac{14x^6}{3}-12x^7+33x^8-\frac{286x^9}{3}+O[x]^{10}$$

$\int$ **f** dx

$$\frac{1}{6}(1+4x)^{3/2}$$

```
Series[%, {x, 0, 9}]
```

$$\frac{1}{6}+x+x^2-\frac{2x^3}{3}+x^4-2x^5+\frac{14x^6}{3}-12x^7+33x^8-\frac{286x^9}{3}+O[x]^{10}$$

```
s1 + Sin[x]
```

$$1+3x-2x^2+\frac{23x^3}{6}-10x^4+\frac{3361x^5}{120}-84x^6+\frac{1\,330\,559\,x^7}{5040}-858x^8+O[x]^9$$

```
s1 - f
```

$$O[x]^9$$

$$\mathbf{s2}=\frac{\mathbf{1}}{\mathbf{s1}}$$

$$1-2x+6x^2-20x^3+70x^4-252x^5+924x^6-3432x^7+12\,870\,x^8+O[x]^9$$

```
s1 s2
```

$$1+O[x]^9$$

$$\mathbf{Series}\left[\frac{\mathbf{1}}{\mathbf{f}},\{\mathbf{x},\mathbf{0},\mathbf{8}\}\right]$$

$$1-2x+6x^2-20x^3+70x^4-252x^5+924x^6-3432x^7+12\,870\,x^8+O[x]^9$$

$$\mathbf{Series}\left[\frac{\mathbf{s1}}{\mathbf{Cos[x]}},\{\mathbf{x},\mathbf{0},\mathbf{4}\}\right]$$

$$1+2x-\frac{3x^2}{2}+5x^3-\frac{259x^4}{24}+O[x]^5$$

```
s1 = Series[Sin[x], {x, 0, 8}]
```

$$x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040}+O[x]^9$$

```
s2 = InverseSeries[s1, x]
```

$$x+\frac{x^3}{6}+\frac{3x^5}{40}+\frac{5x^7}{112}+O[x]^9$$

```
Series[ArcSin[x], {x, 0, 8}]
```

$$x+\frac{x^3}{6}+\frac{3x^5}{40}+\frac{5x^7}{112}+O[x]^9$$

ComposeSeries[f,g] funktioniert nur, wenn die beiden Reihen "verkettet" sind, d.h. wenn der konstante Term in g mit dem Expansionspunkt von f übereinstimmt.

```
f = Sin[x]; sf = Series[f, {x, 0, 8}]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + O[x]^9$$

```
g = Tan[x]; sg = Series[g, {x, 0, 8}]
```

$$x + \frac{x^3}{3} + \frac{2\,x^5}{15} + \frac{17\,x^7}{315} + O[x]^9$$

```
ComposeSeries[sf, sg]
```

$$x + \frac{x^3}{6} - \frac{x^5}{40} - \frac{55\,x^7}{1008} + O[x]^9$$

```
Series[Sin[Tan[x]], {x, 0, 8}]
```

$$x + \frac{x^3}{6} - \frac{x^5}{40} - \frac{55\,x^7}{1008} + O[x]^9$$

```
invsf = InverseSeries[sf, x]
```

$$x + \frac{x^3}{6} + \frac{3\,x^5}{40} + \frac{5\,x^7}{112} + O[x]^9$$

```
ComposeSeries[sf, invsf]
```

$$x + O[x]^9$$

Differentiation

Ableitungen einstelliger Funktionen

$\mathtt{D}[x^3 + x^2 + x,\ x]$

$$1 + 2\,x + 3\,x^2$$

$\mathtt{D}[x^3 + x^2 + x,\ \{x,\ 2\}]$

$$2 + 6\,x$$

Die Notation $\partial_x(A)$ wird intern sofort in die Notation **D[A,x]** umgesetzt.

$\partial_x\,(x^3 + x^2 + x)$

$$1 + 2\,x + 3\,x^2$$

$\partial_x \frac{x^2+1}{x^3-1}$ **// Simplify**

$-\frac{x\,(2+3\,x+x^3)}{(-1+x^3)^2}$

∂_x **Sin[Cos[x]]**

-Cos[Cos[x]] Sin[x]

Ist im Ausdruck **A** eine Funktion **f** enthalten, deren Ableitung nicht bekannt ist, so wird **D[A,x]** in einen Ausdruck umgeformt, der **f** und **f'** (FullForm: **Derivative[1][f]**) etc. enthält.

Clear[f, g]

∂_x **(f[x] g[x])**

g[x] f′[x] + f[x] g′[x]

$\partial_x \frac{f[x]}{g[x]}$ **// Together**

$\frac{g[x]\,f'[x] - f[x]\,g'[x]}{g[x]^2}$

∂_x **f[g[x]]**

f′[g[x]] g′[x]

Clear[f]

f[x_] := Sin[a x];

∂_x **(**∂_x **(**∂_x **(**∂_x **f[x])))**

a^4 Sin[a x]

$\partial_{x,x,x,x}$ **f[x]**

a^4 Sin[a x]

$\partial_{\{x,4\}}$ **f[x]**

a^4 Sin[a x]

f''''[x]

a^4 Sin[a x]

g' // FullForm

Derivative[1][g]

```
{Sin′, Sqrt′, Log′}
```

$\left\{\text{Cos}[\#1]\ \&,\ \frac{1}{2\sqrt{\#1}}\ \&,\ \frac{1}{\#1}\ \&\right\}$

```
Clear[f]
```

Achtung, die folgende Definition werden Sie nicht so schnell wieder los. Unklar bleibt, wo sie von *Mathematica* gespeichert wird.

$f'[x\_] := g[3\,x^2 + 5]$

$\partial_{y,y} f[y]$

$6\,y\,g'[5 + 3\,y^2]$

```
ClearAll["Global`*"]
```

```
D[f[y], y, y]
```

$6\,y\,g'[5 + 3\,y^2]$

```
D[f[y], {y, 2}]
```

$6\,y\,g'[5 + 3\,y^2]$

Ubrigens hat die Definition von f' keinen Einfluss auf f'', denn f' ist als **Derivative[1][f]** codiert, f'' dagegen als **Derivative[2][f]**. Das Muster der Definition passt also nicht. $\partial_{y,y} f[y]$ dagegen ist als **D[f[y],y,y]** codiert.

```
f''[x]
```

$f''[x]$

■ Ableitungen mehrstelliger Funktionen

$\left\{D\left[(x^2 + y^3)^4,\ x\right],\ \partial_x\,(x^2 + y^3)^4\right\}$

$\left\{8\,x\,(x^2 + y^3)^3,\ 8\,x\,(x^2 + y^3)^3\right\}$

$\partial_y\,(x^2 + y^3)^4$

$12\,y^2\,(x^2 + y^3)^3$

$\partial_{x,y}\,(x^2 + y^3)^4$

$72\,x\,y^2\,(x^2 + y^3)^2$

$\partial_{y,x}\,(x^2 + y^3)^4$

$72\,x\,y^2\,(x^2 + y^3)^2$

$\partial_x \left(x^2 + y[x]^3\right)^4$

$4 \left(x^2 + y[x]^3\right)^3 \left(2\,x + 3\,y[x]^2\,y'[x]\right)$

$\partial_{x,x}$ `f[x, y] // FullForm`

```
Derivative[2, 0][f][x, y]
```

```
ClearAll["Global`*"]
```

∂_x `f[x, y, z]`

$f^{(1,0,0)}[x, y, z]$

```
% // FullForm
```

```
Derivative[1, 0, 0][f][x, y, z]
```

$\partial_{x,y,y,z}$ `f[x, y, z]`

$f^{(1,2,1)}[x, y, z]$

```
% // FullForm
```

```
Derivative[1, 2, 1][f][x, y, z]
```

Totales Differential

$\mathrm{Dt}\left[\left(x^2 + y^3\right)^4\right]$

$4 \left(x^2 + y^3\right)^3 \left(2\,x\,\mathrm{Dt}[x] + 3\,y^2\,\mathrm{Dt}[y]\right)$

$\mathrm{Dt}\left[\left(x^2 + y^3\right)^4, x\right]$

$4 \left(x^2 + y^3\right)^3 \left(2\,x + 3\,y^2\,\mathrm{Dt}[y, x]\right)$

$\mathrm{Dt}\left[\left(x^2 + y^3\right)^4, y\right]$

$4 \left(x^2 + y^3\right)^3 \left(3\,y^2 + 2\,x\,\mathrm{Dt}[x, y]\right)$

$\mathrm{Dt}\left[\left(x^2 + y^3\right)^4, x, \mathrm{Constants} \to y\right]$

$8\,x \left(x^2 + y^3\right)^3$

$\mathrm{Dt}\left[x^2 + y^2 == a^2, x\right]$

$2\,x + 2\,y\,\mathrm{Dt}[y, x] == 2\,a\,\mathrm{Dt}[a, x]$

$\mathrm{Dt}\left[x^2 + y^2 == a^2, x, \mathrm{Constants} \to a\right]$

$2\,x + 2\,y\,\mathrm{Dt}[y, x, \mathrm{Constants} \to \{a\}] == 0$

```
a /: Dt[a, x] = 0;
```

```
Dt[x^2 + y^2 == a^2, x]
```

$2x + 2y\,\mathrm{Dt}[y, x] == 0$

```
ClearAll[a]
```

```
SetAttributes[a, Constant]
```

```
Dt[x^2 + y^2 == a^2, x]
```

$2x + 2y\,\mathrm{Dt}[y, x] == 0$

■ Ein Beispiel

```
y /: Dt[y, x] = k;
```

```
gl = x^2 + y^2 == 16;
```

```
gldt = Dt[gl, x]
```

$2x + 2ky == 0$

```
eqns = {gl, gldt}
```

$\{x^2 + y^2 == 16, 2x + 2ky == 0\}$

```
Solve[eqns, k]
```

$\left\{\left\{k \to -\frac{x}{y}\right\}\right\}$

```
Reduce[eqns, k]
```

$\left(x == -\sqrt{16 - y^2} \;||\; x == \sqrt{16 - y^2}\right) \&\& \; y \neq 0 \;\&\&\; k == -\frac{x}{y}$

```
lsg = Solve[eqns, k, y]
```

$\left\{\left\{k \to -\frac{x}{\sqrt{16 - x^2}}\right\}, \left\{k \to \frac{x}{\sqrt{16 - x^2}}\right\}\right\}$

```
p1 = Plot[Evaluate[k /. lsg[[1]]], {x, -4, 4}, PlotRange → {-6, 6}];
p2 = ContourPlot[Evaluate[gl], {x, -4, 4}, {y, -4, 4}];
Show[{p1, p2}, AspectRatio -> Automatic]
```



Integration

Einführung

$$\int (x^2 + x^3)\, dx$$

$$\frac{x^3}{3} + \frac{x^4}{4}$$

$$\int_2^3 \left(x^2 + x^3\right) dx$$

$$\frac{271}{12}$$

$$\int_a^b \left(x^2 + x^3\right) dx$$

$$-\frac{a^3}{3} - \frac{a^4}{4} + \frac{b^3}{3} + \frac{b^4}{4}$$

```
Integrate[x^2 + x^3, {x, a, b}]
```

$$-\frac{a^3}{3} - \frac{a^4}{4} + \frac{b^3}{3} + \frac{b^4}{4}$$

Integrale rationaler Funktionen

$$\int \frac{x^5 - 2x^4 + 2x^3 - x + 1}{x^4 + x^2} dx$$

$$-\frac{1}{x} - 2x + \frac{x^2}{2} + \mathrm{ArcTan}[x] - \mathrm{Log}[x] + \mathrm{Log}\left[1 + x^2\right]$$

$$\mathrm{i1} = \mathrm{Table}\left[\int \frac{1}{x^n - 2x + 1} dx, \{n, 2, 4\}\right]$$

$$\left\{-\frac{1}{-1 + x}, \frac{1}{10}\left(-5 - 3\sqrt{5}\right)\mathrm{Log}\left[-1 + \sqrt{5} - 2x\right] + \mathrm{Log}[-1 + x] + \frac{1}{10}\left(-5 + 3\sqrt{5}\right)\mathrm{Log}\left[1 + \sqrt{5} + 2x\right], \frac{1}{2}\mathrm{Log}[-1 + x] - \frac{1}{2}\mathrm{RootSum}\left[-1 + \#1 + \#1^2 + \#1^3 \&, \frac{3\,\mathrm{Log}[x - \#1] + 2\,\mathrm{Log}[x - \#1]\,\#1 + \mathrm{Log}[x - \#1]\,\#1^2}{1 + 2\,\#1 + 3\,\#1^2} \&\right]\right\}$$

```
i1 // Normal
```

$$\left\{-\frac{1}{-1+x},\ \frac{1}{10}\left(-5-3\sqrt{5}\right)\mathrm{Log}\left[-1+\sqrt{5}-2x\right]+\mathrm{Log}[-1+x]+\frac{1}{10}\left(-5+3\sqrt{5}\right)\mathrm{Log}\left[1+\sqrt{5}+2x\right],\right.$$

$$\frac{1}{2}\mathrm{Log}[-1+x]+\frac{1}{2}\left(-\left(3\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]]+2\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]+\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]^2\right)\Big/\left(1+2\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]+3\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,1]^2\right)-\right.$$

$$\left(3\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]]+2\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]+\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]^2\right)\Big/\left(1+2\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]+3\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,2]^2\right)-$$

$$\left.\left.\left(3\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]]+2\,\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]+\mathrm{Log}[x-\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]]\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]^2\right)\Big/\left(1+2\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]+3\,\mathrm{Root}[-1+\#1+\#1^2+\#1^3\&,3]^2\right)\right)\right\}$$

$$\mathbf{i2}=\int\frac{1+x^{1/3}}{x+\sqrt{x}}\,dx$$

$$3x^{1/3}-2\sqrt{3}\,\mathrm{ArcTan}\left[\frac{-1+2x^{1/6}}{\sqrt{3}}\right]+2\,\mathrm{Log}\left[1+x^{1/6}\right]-\mathrm{Log}\left[1-x^{1/6}+x^{1/3}\right]+2\,\mathrm{Log}\left[1+\sqrt{x}\right]$$

$$\mathbf{i3}=\int\frac{1}{x^n+1}\,dx$$

$$x\,\mathrm{Hypergeometric2F1}\left[\frac{1}{n},1,1+\frac{1}{n},-x^n\right]$$

$$\left\{\mathrm{FullSimplify}[\mathbf{i3}\ /.\ n\to 2],\ \mathrm{FullSimplify}[\mathbf{i3}\ /.\ n\to 3],\ \mathbf{i4}=\int\frac{1}{x^3+1}\,dx\right\}$$

$$\left\{\mathrm{ArcTan}[x],\ x\,\mathrm{Hypergeometric2F1}\left[\frac{1}{3},1,\frac{4}{3},-x^3\right],\ \frac{\mathrm{ArcTan}\left[\frac{-1+2x}{\sqrt{3}}\right]}{\sqrt{3}}+\frac{1}{3}\mathrm{Log}[1+x]-\frac{1}{6}\mathrm{Log}\left[1-x+x^2\right]\right\}$$

```
FunctionExpand[i3 /. n → 3] // Simplify
i5 = PowerExpand[%]
```

$$\frac{1}{3x^2}\left(x^3\right)^{2/3}\left(\mathrm{Log}\left[1+\left(x^3\right)^{1/3}\right]+(-1)^{2/3}\,\mathrm{Log}\left[1-(-1)^{1/3}\left(x^3\right)^{1/3}\right]-(-1)^{1/3}\,\mathrm{Log}\left[1+(-1)^{2/3}\left(x^3\right)^{1/3}\right]\right)$$

$$\frac{1}{3}\left(\mathrm{Log}[1+x]+(-1)^{2/3}\,\mathrm{Log}\left[1-(-1)^{1/3}x\right]-(-1)^{1/3}\,\mathrm{Log}\left[1+(-1)^{2/3}x\right]\right)$$

```
Plot[{i4, i5}, {x, 0, 5}]
```




```
FullSimplify[i4 - i5, x > 0]
```

$$\frac{1}{6}\left(2\sqrt{3}\,\mathrm{ArcTan}\left[\frac{-1+2\,x}{\sqrt{3}}\right]-2\,(-1)^{2/3}\,\mathrm{Log}\left[1-(-1)^{1/3}\,x\right]+2\,(-1)^{1/3}\,\mathrm{Log}\left[1+(-1)^{2/3}\,x\right]-\mathrm{Log}\left[1+(-1+x)\,x\right]\right)$$

Elliptische Integrale

```
ClearAll["Global`*"]
```

$$\int\frac{1}{\sqrt{1-n\,\mathrm{Sin}[\alpha]^2}}\,d\alpha$$

EllipticF[α, n]

$$\mathrm{i6}=\int\sqrt{a+b\,\mathrm{Sin}[\alpha]}\,d\alpha$$

$$-\frac{2\,\mathrm{EllipticE}\left[\frac{1}{4}(\pi-2\,\alpha),\frac{2\,b}{a+b}\right]\sqrt{a+b\,\mathrm{Sin}[\alpha]}}{\sqrt{\frac{a+b\,\mathrm{Sin}[\alpha]}{a+b}}}$$

D[i6, α]

$$\frac{b\,\text{Cos}[\alpha]\,\text{EllipticE}\left[\frac{1}{4}(\pi-2\alpha),\frac{2b}{a+b}\right]\sqrt{a+b\,\text{Sin}[\alpha]}}{(a+b)\left(\frac{a+b\,\text{Sin}[\alpha]}{a+b}\right)^{3/2}}-$$

$$\frac{b\,\text{Cos}[\alpha]\,\text{EllipticE}\left[\frac{1}{4}(\pi-2\alpha),\frac{2b}{a+b}\right]}{\sqrt{a+b\,\text{Sin}[\alpha]}\sqrt{\frac{a+b\,\text{Sin}[\alpha]}{a+b}}}+\frac{\sqrt{1-\frac{2b\,\text{Sin}\left[\frac{1}{4}(\pi-2\alpha)\right]^2}{a+b}}\sqrt{a+b\,\text{Sin}[\alpha]}}{\sqrt{\frac{a+b\,\text{Sin}[\alpha]}{a+b}}}$$

% // Simplify

$$\sqrt{a+b\,\text{Sin}[\alpha]}$$

i7 = $\int\sqrt{\frac{x}{x^3+x^2+x+1}}\,dx$

$$(1-i)\,x\sqrt{\frac{(1+i)(i+x)}{x}}\sqrt{\frac{(-i+x)(1+x)}{x^2}}$$

$$\sqrt{\frac{x}{1+x+x^2+x^3}}\,\text{EllipticPi}\left[1-i,\text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)(1+x)}{x}}}{\sqrt{2}}\right],-i\right]$$

D[i7, x]

$$\frac{\left(\frac{1}{2}-\frac{i}{2}\right) x \sqrt{\frac{(1+i)(i+x)}{x}} \sqrt{\frac{(-i+x)(1+x)}{x^2}} \sqrt{\frac{x}{1+x+x^2+x^3}} \left(\frac{1+i}{x}-\frac{(1+i)(1+x)}{x^2}\right)}{\sqrt{2} \sqrt{\frac{(1+i)(1+x)}{x}} \left(1-\frac{1+x}{x}\right) \sqrt{1-\frac{\left(\frac{1}{2}+\frac{i}{2}\right)(1+x)}{x}} \sqrt{1-\frac{\left(\frac{1}{2}-\frac{i}{2}\right)(1+x)}{x}}}+$$

$$(1-i) \sqrt{\frac{(1+i)(i+x)}{x}} \sqrt{\frac{(-i+x)(1+x)}{x^2}} \sqrt{\frac{x}{1+x+x^2+x^3}}$$

$$\text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)(1+x)}{x}}}{\sqrt{2}}\right], -i\right] + \frac{1}{\sqrt{\frac{(1+i)(i+x)}{x}}} \left(\frac{1}{2}-\frac{i}{2}\right) x \sqrt{\frac{(-i+x)(1+x)}{x^2}}$$

$$\sqrt{\frac{x}{1+x+x^2+x^3}} \left(\frac{1+i}{x}-\frac{(1+i)(i+x)}{x^2}\right) \text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)(1+x)}{x}}}{\sqrt{2}}\right], -i\right] +$$

$$\frac{1}{\sqrt{\frac{(-i+x)(1+x)}{x^2}}} \left(\frac{1}{2}-\frac{i}{2}\right) x \sqrt{\frac{(1+i)(i+x)}{x}} \sqrt{\frac{x}{1+x+x^2+x^3}}$$

$$\left(\frac{-i+x}{x^2}+\frac{1+x}{x^2}-\frac{2(-i+x)(1+x)}{x^3}\right) \text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)(1+x)}{x}}}{\sqrt{2}}\right], -i\right] +$$

$$\frac{1}{\sqrt{\frac{x}{1+x+x^2+x^3}}} \left(\frac{1}{2}-\frac{i}{2}\right) x \sqrt{\frac{(1+i)(i+x)}{x}} \sqrt{\frac{(-i+x)(1+x)}{x^2}}$$

$$\left(-\frac{x(1+2x+3x^2)}{(1+x+x^2+x^3)^2}+\frac{1}{1+x+x^2+x^3}\right) \text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)(1+x)}{x}}}{\sqrt{2}}\right], -i\right]$$

% // Simplify

$$\sqrt{\frac{x}{1+x+x^2+x^3}}$$

Das Ergebnis i7 stellt Sie sicher nicht zufrieden, da die verschiedenen Wurzeln zusammengefasst werden können. Aus Gründen der mathematischen Exaktheit lässt sich das mit den üblichen Simplifikationskommandos nicht bewerkstelligen.

Das nächste Kommando extrahiert alle Terme des Produkts bis auf den letzten. Dazu wird das Produkt in eine Liste verwandelt, **DeleteCases** angewendet und danach das Ergebnis wieder in eine Liste verwandelt. Natürlich können Sie alternativ auch die zu untersuchenden Teilausdrücke mit der Maus markieren und in eine neue Zelle kopieren.

u = Times @@ DeleteCases[List @@ i7, _EllipticPi]
u^2 // Together

$$(1-i)\,x\,\sqrt{\frac{(1+i)\,(i+x)}{x}}\,\sqrt{\frac{(-i+x)\,(1+x)}{x^2}}\,\sqrt{\frac{x}{1+x+x^2+x^3}}$$

$$2-2\,i$$

Auch hätte es nicht des Umwegs über eine Liste bedurft, da **DeleteCases** auch auf allgemeinere Ausdrücke angewendet werden kann.

DeleteCases[i7, _EllipticPi]

$$(1-i)\,x\,\sqrt{\frac{(1+i)\,(i+x)}{x}}\,\sqrt{\frac{(-i+x)\,(1+x)}{x^2}}\,\sqrt{\frac{x}{1+x+x^2+x^3}}$$

Das Integral i7 sollte also zum folgenden Ausdruck äquivalent sein.

i8 = $\sqrt{2-2\,I}$ EllipticPi$\left[1-I, \text{ArcSin}\left[\sqrt{\frac{(1+I)\,(1+x)}{2\,x}}\right], -I\right]$

$$\sqrt{2-2\,i}\ \text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)\,(1+x)}{x}}}{\sqrt{2}}\right], -i\right]$$

Alternativ hätten Sie natürlich auch einfach `i7`^{2} berechnen können:

i7^2 // Together

$$(2-2\,i)\ \text{EllipticPi}\left[1-i, \text{ArcSin}\left[\frac{\sqrt{\frac{(1+i)\,(1+x)}{x}}}{\sqrt{2}}\right], -i\right]^2$$

Versuchen wir die Probe, indem wieder die Ableitung der Stammfunktion i6 berechnet wird.

a8 = D[i8, x]

$$\frac{\sqrt{1-i}\,\left(\frac{1+i}{x}-\frac{(1+i)\,(1+x)}{x^2}\right)}{2\,\sqrt{\frac{(1+i)\,(1+x)}{x}}\,\left(1-\frac{1+x}{x}\right)\,\sqrt{1-\frac{\left(\frac{1}{2}+\frac{i}{2}\right)\,(1+x)}{x}}\,\sqrt{1-\frac{\left(\frac{1}{2}-\frac{i}{2}\right)\,(1+x)}{x}}}$$

```
a8 // Simplify
```

$$\frac{1}{\sqrt{\frac{1}{2}-\frac{i}{2}}\,x\,\sqrt{\frac{(1+i)\,(i+x)}{x}}\,\sqrt{\frac{(-i+x)\,(1+x)}{x^2}}}$$

Leider dasselbe Problem wie oben mit den Wurzeln. Wir quadrieren einfach wieder. i6 ist also bis auf das Vorzeichen die richtige Stammfunktion und mehr können wir auch nicht erwarten, da für komplexe x der Radikand selbst nur bis aufs Vorzeichen eindeutig bestimmt ist.

```
a8^2 // Together
a8^2 // Simplify
```

$$\frac{x}{(-i+x)\,(i+x)\,(1+x)}$$

$$\frac{x}{1+x+x^2+x^3}$$

Mehrfachintegrale

$$\int_0^2\int_0^3 1\,dy\,dx$$

6

Bei vielen bestimmten Mehrfachintegralen ist die Reihenfolge des Abintegrierens relevant, da die Grenzen des inneren Integrals von der äußeren Integrationsvariablen abhängen können! Die beiden folgenden Integrale sind gleichwertig, einmal in der Formelschreibweise, einmal als konventionelles Kommando; beachten Sie beim zweiten Beispiel, dass das innere Integral durch den äußeren Parameter angegeben wird (während bei der Formelschreibweise das innere Integral natürlich innen steht)

$$\int_0^1\int_0^{\sqrt{1-x^2}} 1\,dy\,dx$$

$$\frac{\pi}{4}$$

```
Integrate[1, {x, 0, 1}, {y, 0, Sqrt[1 - x^2]}]
```

$$\frac{\pi}{4}$$

$$4\int_0^1\int_0^{\sqrt{1-x^2}} E^{-(x^2+y^2)}\,dy\,dx$$

$$4\int_0^1 \frac{1}{2}\,e^{-x^2}\sqrt{\pi}\,\mathrm{Erf}\left[\sqrt{1-x^2}\right]dx$$

```
% // N
```

1.98587

$$4\int_0^{\pi/2}\int_0^1 E^{-r^2}\, r\, dr\, d\phi$$

$$\frac{(-1+e)\,\pi}{e}$$

```
% // N
```

1.98587

```
p1 = RevolutionPlot3D[E^(-r^2), {r, 0, 1}, {φ, 0, 2 π}];
p2 = ParametricPlot3D[{Cos[φ], Sin[φ], z},
       {φ, 0, 2 π}, {z, 0, 1 / E}];
Show[{p1, p2}, Boxed → False, Axes → None, PlotRange -> All]
```



Numerische Integration

Für eine Reihe von Integralen, für die keine geschlossenen symbolischen Ausdrücke existieren, kann numerische Integration versucht werden.

$$\int_1^3 x^{1/x}\, dx$$

$$\int_1^3 x^{\frac{1}{x}}\, dx$$

```
% // N
```

2.71508

NIntegrate$\left[x^{1/x}, \{x, 1, 3\}\right]$

2.71508

Limit$\left[x^{1/x}, x \to 0\right]$
Limit$\left[x^{1/x}, x \to \infty\right]$

0

1

Plot$\left[x^{1/x}, \{x, 1, 20\}, \text{AxesOrigin} \to \{0, 1\}\right]$




```
Integrate[x^(1/x) - 1, {x, 1, ∞}]
```

Integrate::idiv : Integral of $-1 + x^{\frac{1}{x}}$ does not converge on $\{1, \infty\}$.

$$\int_1^{\infty} \left(-1 + x^{\frac{1}{x}}\right) dx$$

```
NIntegrate[x^(1/x) - 1, {x, 1, ∞}]
```

```
NIntegrate::slwcon :
 Numerical integration converging too slowly; suspect one of the
   following: singularity, value of the integration is 0,
   highly oscillatory integrand, or WorkingPrecision too small.

NIntegrate::ncvb :
 NIntegrate failed to converge to prescribed accuracy after
   9 recursive bisections in x near {x} = {6.13224×10^28}.
   NIntegrate obtained 1.7541166899586654`*^8 and
   1.7070432577814826`*^8 for the integral and error estimates.
```

1.75412×10^{8}

```
Clear[f];
f[u_] := NIntegrate[x^(1/x) - 1, {x, 1, u}]
```

```
Grid[Table[{n, PaddedForm[f[10^n], {5, 2}]}, {n, 2, 10}], Alignment → Right]
```

2	11.51
3	24.89
4	43.48
5	67.34
6	96.50
7	130.96
8	170.73
9	215.79
10	266.16

```
Plot[f[u], {u, 10, 1000}]
```



```
Plot[{f[u], Log[u]^2/2}, {u, 3, 10^3}, PlotStyle -> {Dashed, Dotted}]
```



Uneigentliche Integrale

```
f = 1/x^2; {Integrate[f, x], Integrate[f, {x, 1, ∞}]}
```

$\left\{-\frac{1}{x}, 1\right\}$

```
f = 1/(1 - x^2)^(1/2); {Integrate[f, x], Integrate[f, {x, 0, 1}]}
```

$\left\{\text{ArcSin}[x], \frac{\pi}{2}\right\}$

```
f = E^(-x^2); {Integrate[f, x], Integrate[f, {x, 0, ∞}]}
```

$\left\{\frac{1}{2}\sqrt{\pi}\,\text{Erf}[x], \frac{\sqrt{\pi}}{2}\right\}$

```
f = 1/(1 - x^2)^(1/3); {Integrate[f, x], s = Integrate[f, {x, 0, 1}]}
```

$\left\{x\,\text{Hypergeometric2F1}\left[\frac{1}{2}, \frac{1}{3}, \frac{3}{2}, x^2\right], \frac{3\sqrt{\pi}\,\text{Gamma}\left[\frac{2}{3}\right]}{\text{Gamma}\left[\frac{1}{6}\right]}\right\}$

```
N[s, 30]
```

1.29355477961489526747675751257

```
f = 1/x; {Integrate[f, x], Integrate[f, {x, -1, 1}]}
```

Integrate::idiv : Integral of $\frac{1}{x}$ does not converge on {-1, 1}.

$\left\{\text{Log}[x], \int_{-1}^{1} \frac{1}{x}\, dx\right\}$

```
Plot[f, {x, -3, 3}, PlotRange → {-5, 5}]
```




```
Integrate[f, {x, -1, 1}, PrincipalValue → True]
```

0

```
f = 1/x^2; {Integrate[f, x], Integrate[f, {x, -1, 1}]}
```

Integrate::idiv : Integral of $\frac{1}{x^2}$ does not converge on {-1, 1}.

$\left\{-\frac{1}{x}, \int_{-1}^{1} \frac{1}{x^2}\, dx\right\}$

```
Plot[f, {x, -3, 3}, PlotRange → {0, 5}]
```




```
Integrate[f, {x, -1, -a}] + Integrate[f, {x, b, 1}]
```

$$\text{If}\left[\text{Re}[a] \geq 0 \,||\, \text{Im}[a] \neq 0,\ -1+\frac{1}{a},\ \text{Integrate}\left[\frac{1}{x^2},\ \{x, -1, -a\},\ \text{Assumptions} \to\ !\left(\text{Re}[a] \geq 0 \,||\, \text{Im}[a] \neq 0\right)\right]\right] + \text{If}\left[\left(\text{Re}\left[\frac{b}{1-b}\right] \geq 0\ \&\&\ \frac{b}{-1+b} \neq 0\right) \,||\, \text{Re}\left[\frac{b}{1-b}\right] \leq -1 \,||\, \text{Im}\left[\frac{b}{1-b}\right] \neq 0,\ -1+\frac{1}{b},\ \text{Integrate}\left[\frac{1}{x^2},\ \{x, b, 1\},\ \text{Assumptions} \to\ !\left(\left(\text{Re}\left[\frac{b}{1-b}\right] \geq 0\ \&\&\ \frac{b}{-1+b} \neq 0\right) \,||\, \text{Re}\left[\frac{b}{1-b}\right] \leq -1 \,||\, \text{Im}\left[\frac{b}{1-b}\right] \neq 0\right)\right]\right]$$

```
Integrate[f, {x, -1, -a}, Assumptions → Re[a] > 0] +
 Integrate[f, {x, b, 1}, Assumptions → Re[b] > 0]
```

$$-2+\frac{1}{a}+\frac{1}{b}$$

```
f = (1 - Cos[x])^(-1/2);
Plot[f, {x, -π/2, π/2}, PlotRange → {0, 5}]
```



$$\int_0^{\frac{\pi}{2}} f\, dx$$

Integrate::idiv : Integral of $\frac{1}{\sqrt{1 - \mathrm{Cos}[x]}}$ does not converge on $\left\{0, \frac{\pi}{2}\right\}$.

$$\int_0^{\frac{\pi}{2}} \frac{1}{\sqrt{1 - \mathrm{Cos}[x]}}\, dx$$

```
Integrate[f, {x, -π/2, π/2}]
```

Integrate::idiv : Integral of $\frac{1}{\sqrt{1 - \mathrm{Cos}[x]}}$ does not converge on $\left\{-\frac{\pi}{2}, \frac{\pi}{2}\right\}$.

$$\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \frac{1}{\sqrt{1 - \mathrm{Cos}[x]}}\, dx$$

```
f0 = Integrate[f, x]
```

$$\frac{2\, \mathrm{Log}\left[\mathrm{Tan}\left[\frac{x}{4}\right]\right] \mathrm{Sin}\left[\frac{x}{2}\right]}{\sqrt{1 - \mathrm{Cos}[x]}}$$

```
FullSimplify[f0, Assumptions → {-π/2 < x < π/2}]
```

$\sqrt{1 - \text{Cos}[x]}\ \text{Csc}\left[\frac{x}{2}\right] \text{Log}\left[\text{Tan}\left[\frac{x}{4}\right]\right]$

```
FullSimplify[f0, Assumptions → {0 < x < π/2}]
```

$\sqrt{2}\ \text{Log}\left[\text{Tan}\left[\frac{x}{4}\right]\right]$

```
s = Series[f, {x, 0, 3}]
```

$\frac{\sqrt{2}}{x} + \frac{x}{12\sqrt{2}} + \frac{7x^3}{2880\sqrt{2}} + O[x]^4$

```
u = Normal[s];
Plot[{f, u}, {x, -0.1, 0.1}, PlotRange → {-100, 100}, PlotStyle -> {Dashed, Dotted}]
```




```
i1 = Integrate[f - √2/x, {x, 0, π/2}]
```

$\sqrt{2}\ \text{Log}\left[\frac{8\,\text{Tan}\left[\frac{\pi}{8}\right]}{\pi}\right]$

```
i2 = Integrate[f - √2/x, {x, -π/2, π/2}]
```

$-i\sqrt{2}\ \text{MeijerG}\left[\left\{\left\{\frac{1}{2}\right\}, \left\{\frac{1}{4}, \frac{3}{4}\right\}\right\}, \left\{\left\{0, 0, \frac{1}{2}\right\}, \{\}\right\}, -1, 2\right]$

```
i2 // N
```

$-1.97392 - 3.10148\, i$

$$i3 = \mathrm{Integrate}\left[f - \frac{\sqrt{2}}{\mathrm{Abs}[x]}, \left\{x, -\frac{\pi}{2}, \frac{\pi}{2}\right\}\right]$$

$$2\sqrt{2}\ \mathrm{Log}\left[\frac{8\ \mathrm{Tan}\left[\frac{\pi}{8}\right]}{\pi}\right]$$

Integrale mit Parametern

$$\int_0^\infty E^{-a x^2}\, dx$$

$$\mathrm{If}\left[\mathrm{Re}[a] > 0, \frac{\sqrt{\pi}}{2\sqrt{a}}, \mathrm{Integrate}\left[e^{-a x^2}, \{x, 0, \infty\}, \mathrm{Assumptions} \to \mathrm{Re}[a] \le 0\right]\right]$$

$$\mathrm{Integrate}\left[E^{-a x^2}, \{x, 0, \infty\}, \mathrm{Assumptions} \to \mathrm{Re}[a] > 0\right]$$

$$\frac{\sqrt{\pi}}{2\sqrt{a}}$$

$$\mathrm{Integrate}\left[E^{-a x^2}, \{x, 0, \infty\}, \mathrm{Assumptions} \to \mathrm{Re}[a] < 0\right]$$

Integrate::idiv : Integral of $e^{-a x^2}$ does not converge on $\{0, \infty\}$.

$$\mathrm{Integrate}\left[e^{-a x^2}, \{x, 0, \infty\}, \mathrm{Assumptions} \to \mathrm{Re}[a] < 0\right]$$

Pfadintegrale im Komplexen

Für genügend glatte Funktionen gilt der Hauptsatz der Integralrechnung auch im Komplexen. Deshalb können die folgenden bestimmten Integrale über Einsetzen der Integrationsgrenzen in die Stammfunktion ermittelt werden.

$$\int_0^{1+I} z^2\, dz$$

$$-\frac{2}{3} + \frac{2\,i}{3}$$

$$\int_0^{\pi I} z\ \mathrm{Cos}\left[z^2\right] dz$$

$$-\frac{1}{2}\ \mathrm{Sin}\left[\pi^2\right]$$

Hat die Funktion dagegen Singularitäten wie hier $\frac{1}{z}$ eine Polstelle bei z=0, so ist der Wert des Integrals vom Integrationsweg abhängig.

```
Integrate[1/z, {z, -1, 1}]
```

Integrate::idiv : Integral of $\frac{1}{z}$ does not converge on {-1, 1}.

$$\int_{-1}^{1} \frac{1}{z} \, dz$$

Würde der Hauptsatz der Integralrechnung zur Anwendung kommen, so wäre das Ergebnis gleich $-i\pi$.

```
F[z_] = ∫ 1/z dz
F[1] - F[-1]
```

Log[z]

$-i\pi$

Diese Rechnung zeigt, dass *Mathematica* (nun) weiß, dass der Wert des Integrals vom Integrationspfad abhängt.

Integrate verwendet als Pfad immer die direkte geradlinige Verbindung zwischen den Integrationsgrenzen.

```
Assuming[a > 0, ∫_{-1}^{1+I a} 1/z dz]
```

$-i\pi + \text{Log}[1 + i a]$

```
Assuming[a < 0, ∫_{-1}^{1+I a} 1/z dz]
```

$\text{Log}[-1 - i a]$

```
F[1 + a] - F[-1]
```

$-i\pi + \text{Log}[1 + a]$

Im zweiten Parameter des **Integrate**−Kommandos können mehrere Stützstellen angegeben werden. Das zugehörige Integral wird als Summe der Teilintegrale berechnet.

```
Integrate[1/z, {z, 1, I, -1, -I, 1}]
```

$2 i\pi$

```
Integrate[1/Sin[z], {z, 1, I, -1, -I, 1}]
```

$$2 i\pi + 2\,\text{Log}[1 + e] + 2\,\text{Log}\left[\frac{\text{Cot}\left[\frac{1}{2}\right]}{1 + e}\right] + 2\,\text{Log}\left[\text{Tan}\left[\frac{1}{2}\right]\right]$$

```
% // Simplify
```

2 ⅈ π

Kann *Mathematica* die Stammfunktion nicht berechnen, so wird das Integral unausgewertet zurückgegeben. In diesen Fällen kann eine numerische Auswertung mit **% // N** oder **NIntegrate** noch zu brauchbaren numerischen Werten führen. Die symbolische Ausertung dauert hier ziemlich lange.

```
Integrate[1/Sin[Sin[z]], {z, 1, I, -1, -I, 1}]
```

∫Csc[Sin[z]] ⅆ{z, 1, ⅈ, -1, -ⅈ, 1}

```
% // N
```

0. + 6.28319 ⅈ

Viele Warnungen ergeben sich, wenn der Wert des Integrals gleich Null ist. Hier hilft oft eine Zerlegung des Integrationspfades in mehrere Teile.

```
NIntegrate[1/Sin[z^2], {z, 1, I, -1, -I, 1}]
```

```
NIntegrate::ncvb :
 NIntegrate failed to converge to prescribed accuracy
   after 9 recursive bisections in z near {z} =
   {2.76990133638922052432986798580204914618864284497153983530204398084×10^-15 - <<83>> ⅈ}. NIntegrate obtained
   3.07046×10^-16 - 3.05311×10^-16 ⅈ and 3.309051717675891‘*^-13
   for the integral and error estimates.
```

$0.\times10^{-13} + 0.\times10^{-13}$ ⅈ

```
NIntegrate[1/Sin[z^2], {z, 1, I, -1}] + NIntegrate[1/Sin[z^2], {z, -1, -I, 1}]
```

0. + 0. ⅈ

```
NIntegrate[1/Sin[Sin[z]], {z, 1, I, -1, -I, 1}]
```

0. + 6.28319 ⅈ

Eine (etwas komplexere) Analyse der Residuen zeigt, dass das richtige Ergebnis 2 *i* π lautet.

```
Solve[Sin[Sin[z]] == 0, z]
```

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may
   not be found; use Reduce for complete solution information.
```

{{z → 0}}

```
Reduce[Sin[Sin[z]] == 0, z]
```

```
(C[1] | C[2]) ∈ Integers &&
 (z == π - ArcSin[2 π C[1]] + 2 π C[2] || z == ArcSin[2 π C[1]] + 2 π C[2] ||
   z == π - ArcSin[π + 2 π C[1]] + 2 π C[2] || z == ArcSin[π + 2 π C[1]] + 2 π C[2])
```

Die Lösung kann auch kürzer als
z == ArcSin[π C[1]]+ π C[2], C[1],C[2] ∈ Integers
geschrieben werden. Von diesen liegt genau z=0 im Gebiet, welches vom Integrationspfad umschlossen wird.

```
Residue[1/Sin[Sin[z]], {z, 0}]
```

```
1
```

Nach dem Resiudensatz ist also das Integral gleich 2 *i* π .
Die folgende Aufstellung zeigt, dass alle anderen *z* bereits einen zu großen Realteil haben.

```
Table[ArcSin[k π] // N, {k, -10, 10}]
```

```
{-1.5708 + 4.14021 i, -1.5708 + 4.03479 i, -1.5708 + 3.91692 i, -1.5708 + 3.78327 i,
 -1.5708 + 3.62893 i, -1.5708 + 3.4463 i, -1.5708 + 3.22258 i, -1.5708 + 2.93366 i,
 -1.5708 + 2.52463 i, -1.5708 + 1.81153 i, 0., 1.5708 - 1.81153 i, 1.5708 - 2.52463 i,
 1.5708 - 2.93366 i, 1.5708 - 3.22258 i, 1.5708 - 3.4463 i, 1.5708 - 3.62893 i,
 1.5708 - 3.78327 i, 1.5708 - 3.91692 i, 1.5708 - 4.03479 i, 1.5708 - 4.14021 i}
```

Pfadintegrale lassen sich auch direkt berechnen.

```
pfadIntegral[f_, z_] :=
 Integrate[f[z[t]] z'[t], {t, 0, 1}]
```

Für obiges Beispiel und geradlinigen Pfad ergibt sich eine geschlossene Formel, die für a→ +0und a→ −0verschiedene Werte ergibt.

```
Clear[f];
f[z_] := 1 / z
γ[t_] = t (-1) + (1 - t) (1 + I a);
Assuming[a ∈ Reals && a ≠ 0, pfadIntegral[f, γ]]
```

```
Log[i/(-i + a)]
```

```
Limit[%, a → 0, Direction → #] & /@ {-1, 1}
```

```
{i π, -i π}
```

Auch hier darf nicht einfach der Hauptsatz der Integralrechnung zur Anwendung kommen.

```
u = Integrate[f[γ[t]] γ'[t], t]
```

```
Log[i + a (-1 + t) - 2 i t]
```

```
(u /. t → 1) - (u /. t → 0)
```

$-\frac{i\pi}{2} - \text{Log}[i - a]$

```
Limit[%, a → 0, Direction → #] & /@ {-1, 1}
```

```
{-i π, -i π}
```

Pfadintegrale lassen sich mit **pfadIntegral** auch über kompliziertere Pfade berechnen.

```
γ[t_] = Cos[2 π t] + I Sin[2 π t];
pfadIntegral[f, γ]
```

```
2 i π
```

Das dauert manchmal länger, denn auch hier gilt, dass nicht einfach der Hauptsatz der Integralrechnung angewendet werden kann. Die Stammfunktion selbst ist schnell berechnet.

```
f[z_] = 1 / Sin[z];
γ[t_] = Cos[2 π t] + I Sin[2 π t];
pfadIntegral[f, γ] // Timing
```

```
{94.66, 2 i π}
```

```
f[γ[t]] γ'[t]
```

```
Csc[Cos[2 π t] + i Sin[2 π t]] (2 i π Cos[2 π t] - 2 π Sin[2 π t])
```

```
u = Integrate[%, t]
```

$\text{Log}\left[\text{Tan}\left[\frac{1}{2} (\text{Cos}[2\pi t] + i\,\text{Sin}[2\pi t])\right]\right]$

```
(u /. t → 1) - (u /. t → 0)
```

```
0
```

Residuen

```
Residue[1 / Sin[z], {z, 0}]
```

```
1
```

$\text{Residue}\left[1 / \text{Sin}\left[z^2\right], \{z, 0\}\right]$

```
0
```

```
Series[1/Sin[z^2], {z, 0, 3}]
```

$$\frac{1}{z^2} + \frac{z^2}{6} + O[z]^4$$

Als Beispiel berechnen wir das komplexe
Wegintegral auf dem Kreis um den Mittelpunkt z=0 und mit dem Radius $\frac{3}{2}$ für die Funktion
f[z] = $\frac{-z^2-22\,z+8}{z^3-5\,z^2+4\,z}$.

```
Clear[f, z];
f[z_] := (-z^2 - 22 z + 8)/(z^3 - 5 z^2 + 4 z);
```

Bestimmen der Singularitäten.

```
Solve[Denominator[f[z]] == 0, z]
```

{{z → 0}, {z → 1}, {z → 4}}

Summe über die Residuen innerhalb des Integrationsbereichs.

```
2 π I (Residue[f[z], {z, 0}] + Residue[f[z], {z, 1}])
```

$14\,i\,\pi$

pfadIntegral liefert dasselbe Ergebnis.

```
γ[t_] = 3 / 2 (Cos[2 π t] + I Sin[2 π t]);
pfadIntegral[f, γ]
```

$14\,i\,\pi$