Differentialgleichungen analysieren

Analyse mit NDSolve

Die folgende Gleichung hat wirklich nur die Lösung y[x]=Sin[x], wenn man eine auf ganz **R** definierte sucht. Das wird deutlich, wenn man die numerischen Lösungen mit **NDSolve** und verschiedenen Anfangswerten berechnet. Die nähern sich alle der Standardlösung an.

Das sah in der Version 4.1 noch deutlich komplizierter aus.

```
eqn = y'[x] == Cos[x] + y[x] Sin[x] - y[x]^2;
sol = DSolve[eqn, y, x] // Simplify;
y[x] /. sol // Simplify
```

```
{Sin[x]}
```

Wir wollen das mit **NDSolve** überprüfen.

```
NDSolve[eqn, y, {x, 0, 20}]
```

```
NDSolve::ndnco :
 The number of constraints (0) (initial conditions) is not
   equal to the total differential order of the system (1).
```

```
NDSolve[y'[x] == Cos[x] + Sin[x] y[x] - y[x]^2, y, {x, 0, 20}]
```

```
nsol = Join @@ Table[NDSolve[eqn && y[0] == i, y, {x, 0, 20}], {i, 0, 2, .3}]
```

```
{{y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]},
 {y → InterpolatingFunction[{{0., 20.}}, <>]}}
```

```
eqn /. nsol
```

```
{InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2,
 InterpolatingFunction[{{0., 20.}}, <>][x] ==
  Cos[x] + Sin[x] InterpolatingFunction[{{0., 20.}}, <>][x] -
   InterpolatingFunction[{{0., 20.}}, <>][x]^2}
```

```
eqn /. nsol /. x → 0.1
```

```
{False, False, False, False, False, False, False}
```

Probe mit den Näherungswerten

```
eqn[[1]] - eqn[[2]] /. nsol /. x → 0.1
```

$\{3.10781\times10^{-7}, -1.20047\times10^{-7}, 5.39979\times10^{-8}, 2.08514\times10^{-7}, -2.56219\times10^{-6}, 3.99217\times10^{-6}, -2.43506\times10^{-6}\}$

```
Table[Chop[#, 10^-4] & /@ (eqn[[1]] - eqn[[2]] /. nsol /. x → i), {i, 0, 20}]
```

```
{{0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}}
```

```
Plot[y[x] /. nsol, {x, 0, 20}]
```



Lösungen mit einem Fourierreihenansatz suchen:

```
n = 7;
ansatz = (a_0 + Sum[a_i Cos[i x] + b_i Sin[i x], {i, 1, n}]);
vars = {a_0, Table[{a_i, b_i}, {i, 1, n}]} // Flatten;
eqn1 = eqn[[1]] - eqn[[2]] /. {y[x] → ansatz, y'[x] → D[ansatz, x]}
```

$$-\text{Cos}[x] - \text{Sin}[x]\, a_1 - 2\,\text{Sin}[2\,x]\, a_2 - 3\,\text{Sin}[3\,x]\, a_3 - 4\,\text{Sin}[4\,x]\, a_4 - 5\,\text{Sin}[5\,x]\, a_5 - 6\,\text{Sin}[6\,x]\, a_6 - 7\,\text{Sin}[7\,x]\, a_7 + \text{Cos}[x]\, b_1 + 2\,\text{Cos}[2\,x]\, b_2 + 3\,\text{Cos}[3\,x]\, b_3 + 4\,\text{Cos}[4\,x]\, b_4 + 5\,\text{Cos}[5\,x]\, b_5 + 6\,\text{Cos}[6\,x]\, b_6 + 7\,\text{Cos}[7\,x]\, b_7 - \text{Sin}[x]\,(a_0 + \text{Cos}[x]\, a_1 + \text{Cos}[2\,x]\, a_2 + \text{Cos}[3\,x]\, a_3 + \text{Cos}[4\,x]\, a_4 + \text{Cos}[5\,x]\, a_5 + \text{Cos}[6\,x]\, a_6 + \text{Cos}[7\,x]\, a_7 + \text{Sin}[x]\, b_1 + \text{Sin}[2\,x]\, b_2 + \text{Sin}[3\,x]\, b_3 + \text{Sin}[4\,x]\, b_4 + \text{Sin}[5\,x]\, b_5 + \text{Sin}[6\,x]\, b_6 + \text{Sin}[7\,x]\, b_7) + (a_0 + \text{Cos}[x]\, a_1 + \text{Cos}[2\,x]\, a_2 + \text{Cos}[3\,x]\, a_3 + \text{Cos}[4\,x]\, a_4 + \text{Cos}[5\,x]\, a_5 + \text{Cos}[6\,x]\, a_6 + \text{Cos}[7\,x]\, a_7 + \text{Sin}[x]\, b_1 + \text{Sin}[2\,x]\, b_2 + \text{Sin}[3\,x]\, b_3 + \text{Sin}[4\,x]\, b_4 + \text{Sin}[5\,x]\, b_5 + \text{Sin}[6\,x]\, b_6 + \text{Sin}[7\,x]\, b_7)^2$$

```
sys = CoefficientList[eqn1, Flatten[Table[{Sin[i x], Cos[i x]}, {i, 1, n}]]] // Flatten //
  Union
```

$$\{0,\ a_0^2,\ a_1^2,\ 2\,a_1 a_2,\ a_2^2,\ 2\,a_1 a_3,\ 2\,a_2 a_3,\ a_3^2,\ 2\,a_1 a_4,\ 2\,a_2 a_4,\ 2\,a_3 a_4,\ a_4^2,\ 2\,a_1 a_5,\ 2\,a_2 a_5,\ 2\,a_3 a_5,\ 2\,a_4 a_5,\ a_5^2,\ 2\,a_1 a_6,\ 2\,a_2 a_6,\ 2\,a_3 a_6,\ 2\,a_4 a_6,\ 2\,a_5 a_6,\ a_6^2,\ 2\,a_1 a_7,\ 2\,a_2 a_7,\ 2\,a_3 a_7,\ 2\,a_4 a_7,\ 2\,a_5 a_7,\ 2\,a_6 a_7,\ a_7^2,\ -1 + 2\,a_0 a_1 + b_1,\ -a_0 - a_1 + 2\,a_0 b_1,\ -a_1 + 2\,a_1 b_1,\ -a_2 + 2\,a_2 b_1,\ -a_3 + 2\,a_3 b_1,\ -a_4 + 2\,a_4 b_1,\ -a_5 + 2\,a_5 b_1,\ -a_6 + 2\,a_6 b_1,\ -a_7 + 2\,a_7 b_1,\ -b_1 + b_1^2,\ 2\,a_1 b_2,\ 2\,a_2 b_2,\ 2\,a_3 b_2,\ 2\,a_4 b_2,\ 2\,a_5 b_2,\ 2\,a_6 b_2,\ 2\,a_7 b_2,\ b_2^2,\ 2\,a_0 a_2 + 2\,b_2,\ -2\,a_2 + 2\,a_0 b_2,\ -b_2 + 2\,b_1 b_2,\ 2\,a_1 b_3,\ 2\,a_2 b_3,\ 2\,a_3 b_3,\ 2\,a_4 b_3,\ 2\,a_5 b_3,\ 2\,a_6 b_3,\ 2\,a_7 b_3,\ 2\,b_2 b_3,\ b_3^2,\ 2\,a_0 a_3 + 3\,b_3,\ -3\,a_3 + 2\,a_0 b_3,\ -b_3 + 2\,b_1 b_3,\ 2\,a_1 b_4,\ 2\,a_2 b_4,\ 2\,a_3 b_4,\ 2\,a_4 b_4,\ 2\,a_5 b_4,\ 2\,a_6 b_4,\ 2\,a_7 b_4,\ 2\,b_2 b_4,\ 2\,b_3 b_4,\ b_4^2,\ 2\,a_0 a_4 + 4\,b_4,\ -4\,a_4 + 2\,a_0 b_4,\ -b_4 + 2\,b_1 b_4,\ 2\,a_1 b_5,\ 2\,a_2 b_5,\ 2\,a_3 b_5,\ 2\,a_4 b_5,\ 2\,a_5 b_5,\ 2\,a_6 b_5,\ 2\,a_7 b_5,\ 2\,b_2 b_5,\ 2\,b_3 b_5,\ 2\,b_4 b_5,\ b_5^2,\ 2\,a_0 a_5 + 5\,b_5,\ -5\,a_5 + 2\,a_0 b_5,\ -b_5 + 2\,b_1 b_5,\ 2\,a_1 b_6,\ 2\,a_2 b_6,\ 2\,a_3 b_6,\ 2\,a_4 b_6,\ 2\,a_5 b_6,\ 2\,a_6 b_6,\ 2\,a_7 b_6,\ 2\,b_2 b_6,\ 2\,b_3 b_6,\ 2\,b_4 b_6,\ 2\,b_5 b_6,\ b_6^2,\ 2\,a_0 a_6 + 6\,b_6,\ -6\,a_6 + 2\,a_0 b_6,\ -b_6 + 2\,b_1 b_6,\ 2\,a_1 b_7,\ 2\,a_2 b_7,\ 2\,a_3 b_7,\ 2\,a_4 b_7,\ 2\,a_5 b_7,\ 2\,a_6 b_7,\ 2\,a_7 b_7,\ 2\,b_2 b_7,\ 2\,b_3 b_7,\ 2\,b_4 b_7,\ 2\,b_5 b_7,\ 2\,b_6 b_7,\ b_7^2,\ 2\,a_0 a_7 + 7\,b_7,\ -7\,a_7 + 2\,a_0 b_7,\ -b_7 + 2\,b_1 b_7\}$$

```
sol = Reduce[sys == 0, vars] // Solve
```

```
{{a0 → 0, a1 → 0, a2 → 0, a3 → 0, a4 → 0, a5 → 0,
  a6 → 0, a7 → 0, b1 → 1, b2 → 0, b3 → 0, b4 → 0, b5 → 0, b6 → 0, b7 → 0}}
```

```
ansatz /. sol
```

```
{Sin[x]}
```

Ein komplexes Beispiel

```
eqn = x y[x] (y[x] y'[x] + x) + x^2 - y[x]^2 y'[x] == 0;
y0[x_] = First[y[x] /. DSolve[eqn, y, x]] // Simplify
```

```
Solve::tdep :
 The equations appear to involve the variables to be solved
   for in an essentially non-algebraic way.
```

$$\mathrm{InverseFunction}\left[\mathrm{Log}[1+\#1]-2\,(1+\#1)+\frac{1}{2}\,(1+\#1)^2\,\&\right]\left[\frac{1}{2}\left(3-2\,x-x^2+2\,C[1]-2\,\mathrm{Log}[-1+x]\right)\right]$$

Eine symbolische Lösung mit der Randbedingung y[0]=4 kann *Mathematica* nicht finden.

```
DSolve[eqn && y[0] == 4, y, x]
```

```
Solve::tdep :
 The equations appear to involve the variables to be solved
   for in an essentially non-algebraic way.

DSolve::bvnul :
 For some branches of the general solution, the given boundary
   conditions lead to an empty solution.

{}
```

Das Vektorfeld, welches zu dieser Gleichung gehört

eqn lässt sich einfach nach y'[x] auflösen, womit sich das Vektorfeld der Dgl. berechnen lässt.

```
sol = Solve[eqn, y'[x]];
vfy = First[y'[x] /. sol /. y[x] → y]
```

$$\frac{-x^2 - x^2\,y}{(-1+x)\,y^2}$$

```
frameTicks = {{-1., -.5, 0, .5, 1.}, Automatic, None, None};
myPlotOptions = {Frame -> True, Axes -> None, FrameTicks -> frameTicks};
Needs["VectorFieldPlots`"]

vf[a_, b_, opts___?OptionQ] := VectorFieldPlot[{1, vfy}, {x, -1, .9}, {y, a, b}, opts];
```

Graphics−Optionen haben leider überhaupt keine Auswirkungen, obwohl angeblich doch.

```
Show[vf[-3, 3, PlotPoints -> 20], myPlotOptions, AspectRatio -> 1 ]
```



Lösung mit NDSolve

Die bisherige Analyse hat gezeigt, dass in der Nähe von x=1 Instabilitäten zu erwarten sind, wahrscheinlich eine Polstelle. Deshalb wollen wir Lösungen im Bereich −1<x<0.9 für verschiedene Anfangswerte y[0]=c studieren.

```
NDSolve[eqn && y[0] == 4, y, {x, -1, .9}]
```

```
{{y → InterpolatingFunction[{{-1., 0.9}}, <>]}}
```

Für Anfangswerte c≥ 2hat **NDSolve** keine Probleme, eine Lösung über den ganzen uns interessierenden Bereich zu generieren. Diese passt auch gut in das Vektorfeld der Dgl.

```
nsol1 = Table[NDSolve[eqn && y[0] == c, y, {x, -1, .9}], {c, 2, 4, .5}]
```

```
{{{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}}}
```

Eine Art Probe.

```
Plot[First[eqn] /. nsol1[[1]], {x, -1, .9}]
```



Und hier die Lösungen selbst.

```
p1 = Plot[y[x] /. nsol1, {x, -1, .9}];
Show[p1, vf[2, 4.5], myPlotOptions]
```



Für Anfangswerte in der Nähe von c=0 hat **NDSolve** Probleme, weil die Lösung nicht über das ganze Gebiet ausgedehnt werden kann. Das ist mit Blick auf das Vektorfeld auch nicht weiter verwunderlich. Über die Stelle x_0 mit $y[x_0] = 0$ lässt sich die Lösung nicht sinnvoll fortsetzen.

```
nsol2 = Table[NDSolve[eqn && y[0] == c, y, {x, -.3, .9}], {c, 0.06, 0.2, .02}]
```

```
NDSolve::ndsz : At x == -0.0599997, step size is
   effectively zero; singularity or stiff system suspected.

NDSolve::ndsz : At x == -0.0799997, step size is
   effectively zero; singularity or stiff system suspected.

NDSolve::ndsz : At x == -0.0999997, step size is
   effectively zero; singularity or stiff system suspected.

General::stop : Further output of
   NDSolve::ndsz will be suppressed during this calculation.
```

```
{{{y → InterpolatingFunction[{{-0.0599997, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.0799997, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.0999997, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.12, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.14, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.16, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.18, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.2, 0.9}}, <>]}}}
```

```
p2 = Plot[y[x] /. nsol2, {x, -.3, .9}];
Show[p2, vf[-1, 2.2], myPlotOptions, PlotRange → {-1, 2}]
```

```
InterpolatingFunction::dmval :
 Input value {-0.299975} lies outside the range of data in
   the interpolating function. Extrapolation will be used.

InterpolatingFunction::dmval :
 Input value {-0.299975} lies outside the range of data in
   the interpolating function. Extrapolation will be used.

InterpolatingFunction::dmval :
 Input value {-0.299975} lies outside the range of data in
   the interpolating function. Extrapolation will be used.

General::stop :
 Further output of InterpolatingFunction::dmval will be
   suppressed during this calculation.
```



Für Anfangswerte c<−2 werden ebenfalls plausible Lösungen im ganzen uns interessierenden Bereich gefunden.

```
nsol3 = Table[NDSolve[eqn && y[0] == c, y, {x, -1, .9}], {c, -4, -2, .3}]
```

```
{{{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}}}
```

```
p3 = Plot[y[x] /. nsol3, {x, -1, 1}];
Show[p3, vf[-5, -2], myPlotOptions]
```



Im Bereich −2.2<c<0.5 sehen wir noch einmal deutlich, dass **NDSolve** eine Lösung generell nicht über die Stelle y=0 hinweg verlängern kann, ansonsten aber plausible Lösungen liefert.

```
nsol4 = Table[NDSolve[eqn && y[0] == c, y, {x, -1, .9}], {c, -2.2, 0.5, .3}]
```

```
NDSolve::ndsz :
 At x == 0.6999996454469152‘, step size is effectively zero;
   singularity or stiff system suspected.

NDSolve::ndsz :
 At x == 0.39999972754492774‘, step size is effectively
   zero; singularity or stiff system suspected.

NDSolve::ndsz :
 At x == 0.09999968741677946‘, step size is effectively
   zero; singularity or stiff system suspected.

General::stop : Further output of
   NDSolve::ndsz will be suppressed during this calculation.
```

```
{{{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.7}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.4}}, <>]}},
 {{y → InterpolatingFunction[{{-1., 0.0999997}}, <>]}},
 {{y → InterpolatingFunction[{{-0.2, 0.9}}, <>]}},
 {{y → InterpolatingFunction[{{-0.5, 0.9}}, <>]}}}
```

```
p4 = Plot[y[x] /. nsol4, {x, -1, 1}];
Show[p4, vf[-2, 1], PlotRange → {-2, 1}, myPlotOptions]
```

```
InterpolatingFunction::dmval :
 Input value {-0.999959} lies outside the range of data in
   the interpolating function. Extrapolation will be used.

InterpolatingFunction::dmval :
 Input value {-0.999959} lies outside the range of data in
   the interpolating function. Extrapolation will be used.
```



Noch einmal die symbolische Lösung

Auch in der symbolischen Lösungsschar gibt es Lösungen, welche durch y[0]=c gehen. Wir bestimmen zunächst den Zusammenhang zwischen c und C[1], siehe auch Notebook solve.nb

```
Clear[c]
u = y0[0]
csol = Solve[u[[0, 1]][c] == u[[1]], C[1]]
```

$$\text{InverseFunction}\left[\text{Log}[1+\#1]-2\,(1+\#1)+\frac{1}{2}\,(1+\#1)^2\,\&\right]\left[\frac{1}{2}\,(3-2\,i\,\pi+2\,C[1])\right]$$

$$\left\{\left\{C[1]\to\frac{1}{2}\left(-6-2\,c+c^2+2\,i\,\pi+2\,\text{Log}[1+c]\right)\right\}\right\}$$

Die numerische Auswertung wird von *Mathematica* verweigert, da jeder solche Funktionswert einen Aufruf von **FindRoot** bedeutet. Das kann selbst "nachgerüstet" werden.
Da **InverseFunction** schreibgeschützt ist, ersetzen wir diesen Funktionsnamen durch einen anderen und definieren zugleich eine Funktionenschar $y_1[c][x]$, wobei jede Funktion $y_1[c]$ eine Lösung von **eqn** mit $y_1[c][0] = c$ ist.

```
Clear[y1]
y1[c_][x_] = y0[x] /. csol[[1]] /. InverseFunction → InvFct
```

$$\text{InvFct}\left[\text{Log}[1+\#1]-2\,(1+\#1)+\frac{1}{2}\,(1+\#1)^2\,\&\right]\left[\frac{1}{2}\left(-3-2\,c+c^2+2\,i\,\pi-2\,x-x^2+2\,\text{Log}[1+c]-2\,\text{Log}[-1+x]\right)\right]$$

Und nun definieren wir eine Regel, die InvFct[f] mit numerischen Argumenten wirklich numerisch auswertet.

```
InvFct[f_Function][x_?NumericQ] := Module[{y},
  y /. FindRoot[f[y] == N[x], {y, 1.}]]
```

Die Wertetabelle sieht plausibel aus.

```
Grid[Table[{x, y1[4][x]}, {x, -2.4, 0.9, 0.2}] // Chop, Alignment -> "."]
```

−2.4	3.41672
−2.2	3.53549
−2.	3.63696
−1.8	3.723
−1.6	3.79511
−1.4	3.85449
−1.2	3.90217
−1.	3.93909
−0.8	3.96614
−0.6	3.98434
−0.4	3.99485
−0.2	3.99927
0	4.
0.2	4.00098
0.4	4.00962
0.6	4.04232
0.8	4.14959

Achtung: Für x=1 wertet das Argument B von InvFct[A][B] zu ∞ aus und die InvFct−Regel wird nicht angewendet.

```
y1[4][1]
```

$$\text{InvFct}\left[\text{Log}[1+\#1]-2\,(1+\#1)+\frac{1}{2}\,(1+\#1)^2\,\&\right][\infty]$$

Selbst eine grafische Darstellung kann man aus diesem Weg generieren.

```
Plot[y1[4][x], {x, -1, 1.2}]
```



Statt einzelner Graphen kann auch eine ganze Funktionenschar untersucht werden.

Im Bereich 1.3 ≤ c ≤ 4 zeigen die so gewonnenen Lösungskurven ein reguläres Verhalten ...

```
p5 = Plot[Evaluate[Table[y1[c][x], {c, 1.3, 4, .3}]], {x, -1, 1}];
Show[p5, vf[1, 5], myPlotOptions, PlotRange → {1, 5}]
```



Für c<1 wird das Geschehen insbesondere unterhalb der y−Achse chaotisch; **FindRoot** kann sich nicht für eine stetig variierende Nullstelle entscheiden.

```
p6 = Plot[Evaluate[Table[y1[c][x], {c, -1, 1, .21}]], {x, -1, 1}];
Show[p6, vf[-3, 3], myPlotOptions, PlotRange → {-3, 3}]
```

```
FindRoot::nlnum :
 The function value {Indeterminate} is not a list of numbers
   with dimensions {1} at {y$106762} = {-1.}.

FindRoot::nlnum :
 The function value {Indeterminate} is not a list of numbers
   with dimensions {1} at {y$106775} = {-1.}.

FindRoot::nlnum :
 The function value {Indeterminate} is not a list of numbers
   with dimensions {1} at {y$107147} = {-1.}.

General::stop : Further output of FindRoot::nlnum
   will be suppressed during this calculation.

FindRoot::lstol :
 The line search decreased the step size to within tolerance
   specified by AccuracyGoal and PrecisionGoal but was
   unable to find a sufficient decrease in the merit
   function.  You may need more than MachinePrecision
   digits of working precision to meet these tolerances.

FindRoot::lstol :
 The line search decreased the step size to within tolerance
   specified by AccuracyGoal and PrecisionGoal but was
   unable to find a sufficient decrease in the merit
   function.  You may need more than MachinePrecision
   digits of working precision to meet these tolerances.

FindRoot::lstol :
 The line search decreased the step size to within tolerance
   specified by AccuracyGoal and PrecisionGoal but was
   unable to find a sufficient decrease in the merit
   function.  You may need more than MachinePrecision
   digits of working precision to meet these tolerances.

General::stop : Further output of FindRoot::lstol
   will be suppressed during this calculation.
```

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

```
General::stop : Further output of
   Power::infy will be suppressed during this calculation.
```


3
2
1
0
−1
−2
−3
−1.
−0.5
0
0.5
1.

```
p7 = Plot[Evaluate[Table[y1[c][x], {c, -.9, -.1, .1}]], {x, -1, 1}, PlotStyle -> Thick];
Show[p7, vf[-3, 3], myPlotOptions, PlotRange → {-3, 3}]
```

```
FindRoot::nlnum :
 The function value {Indeterminate} is not a list of numbers
   with dimensions {1} at {y$173351} = {-1.}.
```

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

Power::infy : Infinite expression $\frac{1}{0^2}$ encountered.

```
General::stop : Further output of
   Power::infy will be suppressed during this calculation.
```



Mit c=−1 kann unser **InvFct** gar nichts anfangen, da −∞ nicht die Eigenschaft **NumericQ** (und auch nicht **NumberQ**) hat.

Für c<−1 findet **FindRoot** Nullstellen mit imaginärem Anteil, geht also auf ein anderes Blatt der **Log**−Funktion. Auch $y_3[c][0]$ ergibt nicht mehr c.

```
y1[-1][0]
y1[-1.2][0]
Table[y1[-1.2][x], {x, -3, 1, 0.3}] // N
```

$$\text{InvFct}\left[\text{Log}[1+\#1]-2\,(1+\#1)+\frac{1}{2}\,(1+\#1)^2\;\&\right][-\infty]$$

$2.47112 + 1.80895\,i$

$\{1.8579 + 2.76559\,i,\ 1.95156 + 2.55225\,i,\ 2.0499 + 2.36421\,i,\ 2.14771 + 2.20488\,i,\ 2.23905 + 2.07556\,i,\ 2.31852 + 1.97543\,i,\ 2.38224 + 1.90218\,i,\ 2.42821 + 1.85275\,i,\ 2.45639 + 1.82374\,i,\ 2.46894 + 1.81112\,i,\ 2.47112 + 1.80895\,i,\ 2.47457 + 1.80551\,i,\ 2.51184 + 1.7693\,i,\ 2.78425 + 1.54366\,i\}$

Lösung mit Potenzreihenansatz

```
n = 7;
ansatz = Sum[a_i x^i, {i, 0, n}] + O[x, 0]^(n+1)
eqn1 = eqn /. {y[x] → ansatz, y'[x] → D[ansatz, x]}
```

$$a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 + O[x]^8$$

$$\begin{aligned}
&-a_0^2 a_1 + \left(a_0^2 a_1 - 2 a_0 a_1^2 - 2 a_0^2 a_2\right) x + \\
&\left(1 + a_0 a_1^2 - 4 a_0 a_1 a_2 - a_1 \left(a_1^2 + 2 a_0 a_2\right) + a_0 \left(1 + a_1^2 + 2 a_0 a_2\right) - 3 a_0^2 a_3\right) x^2 + \\
&\Big(a_0 a_1 a_2 - 2 a_2 \left(a_1^2 + 2 a_0 a_2\right) + a_1 \left(1 + a_1^2 + 2 a_0 a_2\right) - \\
&\quad 6 a_0 a_1 a_3 - a_1 (2 a_1 a_2 + 2 a_0 a_3) + a_0 (3 a_1 a_2 + 3 a_0 a_3) - 4 a_0^2 a_4\Big) x^3 + \\
&\Big(a_2 \left(1 + a_1^2 + 2 a_0 a_2\right) + a_0 a_1 a_3 - 3 \left(a_1^2 + 2 a_0 a_2\right) a_3 - 2 a_2 (2 a_1 a_2 + 2 a_0 a_3) + a_1 (3 a_1 a_2 + 3 a_0 a_3) - \\
&\quad 8 a_0 a_1 a_4 - a_1 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_0 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) - 5 a_0^2 a_5\Big) x^4 + \\
&\Big(\left(1 + a_1^2 + 2 a_0 a_2\right) a_3 - 3 a_3 (2 a_1 a_2 + 2 a_0 a_3) + a_2 (3 a_1 a_2 + 3 a_0 a_3) + a_0 a_1 a_4 - \\
&\quad 4 \left(a_1^2 + 2 a_0 a_2\right) a_4 - 2 a_2 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_1 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) - \\
&\quad 10 a_0 a_1 a_5 - a_1 (2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5) + a_0 (5 a_2 a_3 + 5 a_1 a_4 + 5 a_0 a_5) - 6 a_0^2 a_6\Big) x^5 + \\
&\Big(a_3 (3 a_1 a_2 + 3 a_0 a_3) + \left(1 + a_1^2 + 2 a_0 a_2\right) a_4 - 4 (2 a_1 a_2 + 2 a_0 a_3) a_4 - \\
&\quad 3 a_3 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_2 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) + a_0 a_1 a_5 - 5 \left(a_1^2 + 2 a_0 a_2\right) a_5 - \\
&\quad 2 a_2 (2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5) + a_1 (5 a_2 a_3 + 5 a_1 a_4 + 5 a_0 a_5) - 12 a_0 a_1 a_6 - \\
&\quad a_1 \left(a_3^2 + 2 a_2 a_4 + 2 a_1 a_5 + 2 a_0 a_6\right) + a_0 \left(3 a_3^2 + 6 a_2 a_4 + 6 a_1 a_5 + 6 a_0 a_6\right) - 7 a_0^2 a_7\Big) x^6 + O[x]^7 == 0
\end{aligned}$$

```
eqn2 = CoefficientList[Normal[First[eqn1]], x]
```

$$\begin{aligned}
\Big\{&-a_0^2 a_1,\ a_0^2 a_1 - 2 a_0 a_1^2 - 2 a_0^2 a_2, \\
&1 + a_0 a_1^2 - 4 a_0 a_1 a_2 - a_1 \left(a_1^2 + 2 a_0 a_2\right) + a_0 \left(1 + a_1^2 + 2 a_0 a_2\right) - 3 a_0^2 a_3,\ a_0 a_1 a_2 - 2 a_2 \left(a_1^2 + 2 a_0 a_2\right) + \\
&\quad a_1 \left(1 + a_1^2 + 2 a_0 a_2\right) - 6 a_0 a_1 a_3 - a_1 (2 a_1 a_2 + 2 a_0 a_3) + a_0 (3 a_1 a_2 + 3 a_0 a_3) - 4 a_0^2 a_4, \\
&a_2 \left(1 + a_1^2 + 2 a_0 a_2\right) + a_0 a_1 a_3 - 3 \left(a_1^2 + 2 a_0 a_2\right) a_3 - 2 a_2 (2 a_1 a_2 + 2 a_0 a_3) + a_1 (3 a_1 a_2 + 3 a_0 a_3) - \\
&\quad 8 a_0 a_1 a_4 - a_1 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_0 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) - 5 a_0^2 a_5, \\
&\left(1 + a_1^2 + 2 a_0 a_2\right) a_3 - 3 a_3 (2 a_1 a_2 + 2 a_0 a_3) + a_2 (3 a_1 a_2 + 3 a_0 a_3) + a_0 a_1 a_4 - \\
&\quad 4 \left(a_1^2 + 2 a_0 a_2\right) a_4 - 2 a_2 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_1 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) - \\
&\quad 10 a_0 a_1 a_5 - a_1 (2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5) + a_0 (5 a_2 a_3 + 5 a_1 a_4 + 5 a_0 a_5) - 6 a_0^2 a_6, \\
&a_3 (3 a_1 a_2 + 3 a_0 a_3) + \left(1 + a_1^2 + 2 a_0 a_2\right) a_4 - 4 (2 a_1 a_2 + 2 a_0 a_3) a_4 - \\
&\quad 3 a_3 \left(a_2^2 + 2 a_1 a_3 + 2 a_0 a_4\right) + a_2 \left(2 a_2^2 + 4 a_1 a_3 + 4 a_0 a_4\right) + a_0 a_1 a_5 - 5 \left(a_1^2 + 2 a_0 a_2\right) a_5 - \\
&\quad 2 a_2 (2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5) + a_1 (5 a_2 a_3 + 5 a_1 a_4 + 5 a_0 a_5) - 12 a_0 a_1 a_6 - \\
&\quad a_1 \left(a_3^2 + 2 a_2 a_4 + 2 a_1 a_5 + 2 a_0 a_6\right) + a_0 \left(3 a_3^2 + 6 a_2 a_4 + 6 a_1 a_5 + 6 a_0 a_6\right) - 7 a_0^2 a_7\Big\}
\end{aligned}$$

```
sol2 = Solve[eqn2 == 0 ∧ a0 ≠ 0]
```

Solve::svars : Equations may not give solutions for all "solve" variables.

$$\left\{\left\{a_7 \to \frac{-14 - 21 a_0 - 7 a_0^2 + 12 a_0^3 + 12 a_0^4}{84 a_0^5}, a_6 \to \frac{-2 - 3 a_0 - a_0^2 + 3 a_0^3 + 3 a_0^4}{18 a_0^5}, a_5 \to \frac{1 + a_0}{5 a_0^2}, a_4 \to \frac{1 + a_0}{4 a_0^2}, a_3 \to \frac{1 + a_0}{3 a_0^2}, a_2 \to 0, a_1 \to 0\right\}\right\}$$

```
s = ansatz + O[x, 0]^n /. sol2
```

$$\left\{a_0 + \frac{(1 + a_0) x^3}{3 a_0^2} + \frac{(1 + a_0) x^4}{4 a_0^2} + \frac{(1 + a_0) x^5}{5 a_0^2} + \frac{\left(-2 - 3 a_0 - a_0^2 + 3 a_0^3 + 3 a_0^4\right) x^6}{18 a_0^5} + O[x]^7\right\}$$

Hier also die gefundene Potenzreihenlösung ohne den O−Term, wieder in der Notation als Funktionenschar.

```
Clear[y2]
y2[c_][x_] = (s[[1]] /. a0 -> c) // Normal
```

$$c + \frac{(1 + c) x^3}{3 c^2} + \frac{(1 + c) x^4}{4 c^2} + \frac{(1 + c) x^5}{5 c^2} + \frac{\left(-2 - 3 c - c^2 + 3 c^3 + 3 c^4\right) x^6}{18 c^5}$$

In diesem Fall ist auch eine symbolische Probe möglich.

```
eqn /. {y[x] -> s, y'[x] -> D[s, x]}
```

$$\{O[x]^6\} == 0$$

```
p8 = Plot[Evaluate[Table[y2[c][x], {c, 1.3, 5, .3}]], {x, -1, 1}];
Show[p8, vf[1, 5], myPlotOptions]
```




```
p9 = Plot[Evaluate[Table[y2[c][x], {c, -1, 1, .12}]], {x, -1, 1}];
Show[p9, vf[-3, 5], myPlotOptions]
```



```
p10 = Plot[Evaluate[Table[y2[c][x], {c, -4, -1, .3}]], {x, -1, 1}];
Show[p10, vf[-5, -1], myPlotOptions]
```



Noch ein Beispiel

```
eqn = x^3 y''[x] + 3 x^2 y'[x] + x y[x] == 6 Log[x]
```

$x\, y[x] + 3\, x^2\, y'[x] + x^3\, y''[x] == 6\, \mathrm{Log}[x]$

Die exakte Lösung

```
sol = DSolve[eqn, y, x]
```

$$\left\{\left\{y \to \mathrm{Function}\left[\{x\}, \frac{C[1]}{x} + \frac{C[2]\,\mathrm{Log}[x]}{x} + \frac{\mathrm{Log}[x]^3}{x}\right]\right\}\right\}$$

```
Plot[Evaluate[Table[y[x] /. sol[[1]] /. {C[1] → 3, C[2] → c}, {c, 0, 3, .7}]], {x, 0, 2}]
```



Entwicklung der exakten Lösungen als Potenzreihe in der Umgebung von x=1 für y[1]=3.

```
Clear[y1]
y1[c_][x_] = Series[y[x] /. First[sol], {x, 1, 5}] /. {C[1] → 3, C[2] → c + 3} // Simplify
```

$$3 + c\,(x-1) - \frac{3}{2}(1+c)\,(x-1)^2 + \left(\frac{7}{2} + \frac{11\,c}{6}\right)(x-1)^3 + \frac{1}{12}(-69 - 25\,c)\,(x-1)^4 + \frac{1}{60}(486 + 137\,c)\,(x-1)^5 + O[x-1]^6$$

Dieselbe Lösungsschar näherungsweise über einen Potenzreihenansatz.

```
n = 5;
ansatz = 3 + Sum[a_i (x - 1)^i, {i, 1, n}] + O[x, 1]^(n+1)
```

$$3 + a_1\,(x-1) + a_2\,(x-1)^2 + a_3\,(x-1)^3 + a_4\,(x-1)^4 + a_5\,(x-1)^5 + O[x-1]^6$$

```
eqn1 = (eqn[[1]] /. {y[x] → ansatz, y'[x] → D[ansatz, x], y''[x] → D[ansatz, x, x]}) -
  Series[eqn[[2]], {x, 1, n - 2}]
```

$$(3 + 3\,a_1 + 2\,a_2) + (-3 + 7\,a_1 + 12\,a_2 + 6\,a_3)\,(x-1) + (3 + 4\,a_1 + 19\,a_2 + 27\,a_3 + 12\,a_4)\,(x-1)^2 + (-2 + 9\,a_2 + 37\,a_3 + 48\,a_4 + 20\,a_5)\,(x-1)^3 + O[x-1]^4$$

```
sys = CoefficientList[Normal[eqn1], x]
```

$$\{11 - 16\,a_3 - 36\,a_4 - 20\,a_5,\; -3 + 7\,a_1 + 12\,a_2 + 6\,a_3 - 2\,(3 + 4\,a_1 + 19\,a_2 + 27\,a_3 + 12\,a_4) + 3\,(-2 + 9\,a_2 + 37\,a_3 + 48\,a_4 + 20\,a_5),\; 3 + 4\,a_1 + 19\,a_2 + 27\,a_3 + 12\,a_4 - 3\,(-2 + 9\,a_2 + 37\,a_3 + 48\,a_4 + 20\,a_5),\; -2 + 9\,a_2 + 37\,a_3 + 48\,a_4 + 20\,a_5\}$$

```
lsol = Solve[sys == 0, Table[a_i, {i, 2, n}]]
```

$$\left\{\left\{a_2 \to -\frac{3}{2}(1+a_1),\ a_3 \to \frac{1}{6}(21+11\,a_1),\ a_4 \to \frac{1}{12}(-69-25\,a_1),\ a_5 \to \frac{1}{60}(486+137\,a_1)\right\}\right\}$$

```
y2[c_][x_] = ansatz /. First[lsol] /. a_1 → c
```

$$3 + c\,(x-1) - \frac{3}{2}(1+c)(x-1)^2 + \frac{1}{6}(21+11\,c)(x-1)^3 + \frac{1}{12}(-69-25\,c)(x-1)^4 + \frac{1}{60}(486+137\,c)(x-1)^5 + O[x-1]^6$$

Beides sind dieselben Lösungen.

```
y1[c][x] - y2[c][x]
```

$$O[x-1]^6$$