Funktionen und Auswertung

■ Einführung

```
Set // Attributes
```

{HoldFirst, Protected, SequenceHold}

```
SetDelayed // Attributes
```

{HoldAll, Protected, SequenceHold}

Ein und dieselbe Notation auf der rechten Seite kann für unterschiedliche Objekte stehen.

f1 ist ein Ausdruck mit zwei Symbolvariablen
f2 ist eine Funktion mit formalem Parameter x, in der eine weitere Symbolvariable vorkommt.
f3 ist eine zweistelllige Funktion.
f4 ist eine Schar einstelliger Funktionen mit Scharparameter t und Funktionsparameter x.

$$\begin{aligned}&\mathbf{f1} := x^2 + t\,x + 1\\ &\mathbf{f2}[x\_] := x^2 + t\,x + 1\\ &\mathbf{f3}[x\_, t\_] := x^2 + t\,x + 1\\ &\mathbf{f4}[t\_][x\_] := x^2 + t\,x + 1\end{aligned}$$

Das erste Listenelement ist der Ausdruck f1.
Das zweite Listenelement ist der Funktionswert von f2 an der (symbolischen) Stelle x für den speziellen Wert t== 3.
Das dritte Listenelement ist der Funktionswert von f3 an der Stelle (x,t)=(2,x). Das erste Argument (für x) ist eine Zahl, das zweite (für t) das Symbol x.
Das vierte Listenelement ist der Funktionswert der "allgemeinen" Funktion f4[t] der Schar (der Unterschied ist wie der zwischen f2 und f2[x] im zweiten Fall) an der Stelle x== 3.

$$\{\mathbf{f1}, \mathbf{f2}[x] \mathbin{/.} t \to 3, \mathbf{f3}[2, x], \mathbf{f4}[t][3]\}$$

$$\left\{1 + t\,x + x^2, 1 + 3\,x + x^2, 5 + 2\,x, 10 + 3\,t\right\}$$

Und hier werden schließlich fünf der Funktionen aus der Schar f4 grafisch dargestellt. Ganz genau: Dargestellt werden nicht die Funktionen f4[t], sondern die Ausdrücke f4[t][x].

```
u = Table[f4[t][x], {t, 1, 5}];
v = Plot[u, {x, -1, 1} , Ticks → {Join[Range[-1, 1, .5], Table[{i, ""}, {i, -1, 1, .1}]]}]
```



Auswertung von Ausdrücken

Attribute mit Transformationswirkung

```
ClearAll["Global`*"]
```

```
Plus[Plus[x1, x3], x2] // FullForm
```

```
Plus[x1, x2, x3]
```

$\text{Sin}\left[\left\{x, 2, \frac{\pi}{4}, 3.5\right\}\right]$

$\left\{\text{Sin}[x], \text{Sin}[2], \frac{1}{\sqrt{2}}, -0.350783\right\}$

```
Plus[a]
```

```
a
```

Auswertung von Plot−Ausdrücken

```
Attributes[Plot]
```

```
{HoldAll, Protected}
```

```
Plot[∂x Sin[x^2], {x, -3, 3}]
```

```
General::ivar : -2.99988 is not a valid variable.

General::ivar : -2.87743 is not a valid variable.

General::ivar : -2.75498 is not a valid variable.

General::stop : Further output of
   General::ivar will be suppressed during this calculation.
```



```
Plot[Evaluate[∂x Sin[x^2]], {x, -3, 3}]
```



Hiermit gab es gelegentlich Schwierigkeiten, wenn beide Kommandos auf einer Zeile stehen.

```
u = ∂x Sin[x^2]; Plot[u, {x, -3, 3}]
```




```
∂x Sin[x^2];
Plot[%, {x, -3, 3}]
```



Auswertung boolescher Ausdrücke

Die oft zu beobachtende "Faulheit" von *Mathematica* bei der Auswertung boolescher Ausdrücke hat verschiedene Ursachen.

Die erste liegt in der Natur der Ausdrücke selbst: Sie können *Aussagen* (boolesche Ausdrücke ohne freie Variablen, denen sich ein Wahrheitswert zuordnen lässt) oder *Aussageformen* (boolesche Ausdrücke mit freien Variablen) sein.

Beim Anschreiben eines Gleichungssystems etwa werden Aussageformen mit == (**Equal**) verwendet. Hier soll nicht untersucht werden, ob dies für alle (x,y) gilt, sondern es wird das *Problem* formuliert, die Lösungspaare (x,y) zu finden.

```
sys = {x^2 + y - 2 == 0, 3 x - y^2 - 2 == 0};
sys // FullForm
```

```
List[Equal[Plus[-2, Power[x, 2], y], 0],
 Equal[Plus[-2, Times[3, x], Times[-1, Power[y, 2]]], 0]]
```

```
sol = Solve[sys, {x, y}]
```

$$\left\{\{x\to 1,\ y\to 1\},\ \{x\to 2,\ y\to -2\},\ \left\{x\to \frac{1}{2}\left(-3-i\sqrt{3}\right),\ y\to \frac{1}{2}\left(1-3\,i\sqrt{3}\right)\right\},\ \left\{x\to \frac{1}{2}\left(-3+i\sqrt{3}\right),\ y\to \frac{1}{2}\left(1+3\,i\sqrt{3}\right)\right\}\right\}$$

An der Stelle ist die Probe einfach − wenigstens mit Vorversionen von *Mathematica* 6. Alle Vereinfachungen wurden automatisch vorgenommen und die entsprechenden booleschen Ausdrücke zu **True** vereinfacht. Nun nicht mehr (und von einem systematischen Standpunkt aus sollte auch ein **Expand** benötigt werden ...

```
sys /. sol
```

$$\left\{\{\text{True},\ \text{True}\},\ \{\text{True},\ \text{True}\},\ \left\{-2+\frac{1}{4}\left(-3-i\sqrt{3}\right)^2+\frac{1}{2}\left(1-3\,i\sqrt{3}\right)==0,\ \text{True}\right\},\ \left\{-2+\frac{1}{4}\left(-3+i\sqrt{3}\right)^2+\frac{1}{2}\left(1+3\,i\sqrt{3}\right)==0,\ \text{True}\right\}\right\}$$

```
% // Expand
```

```
{{True, True}, {True, True}, {True, True}, {True, True}}
```

... wie es der Fall ist, wenn die rechten Seiten durch einen Parameter *a* ersetzt werden.

```
sys = {x^2 + 2 y == a, y^2 + 2 x == a};
sol = Solve[sys, {x, y}]
sys /. sol
```

$$\left\{\left\{x\to 1-\sqrt{-3+a},\ y\to 1+\sqrt{-3+a}\right\},\ \left\{x\to 1+\sqrt{-3+a},\ y\to 1-\sqrt{-3+a}\right\},\ \left\{x\to -1-\sqrt{1+a},\ y\to -1-\sqrt{1+a}\right\},\ \left\{x\to -1+\sqrt{1+a},\ y\to -1+\sqrt{1+a}\right\}\right\}$$

$$\left\{\left\{\left(1-\sqrt{-3+a}\right)^2+2\left(1+\sqrt{-3+a}\right)==a,\ 2\left(1-\sqrt{-3+a}\right)+\left(1+\sqrt{-3+a}\right)^2==a\right\},\ \left\{2\left(1-\sqrt{-3+a}\right)+\left(1+\sqrt{-3+a}\right)^2==a,\ \left(1-\sqrt{-3+a}\right)^2+2\left(1+\sqrt{-3+a}\right)==a\right\},\ \left\{2\left(-1-\sqrt{1+a}\right)+\left(-1-\sqrt{1+a}\right)^2==a,\ 2\left(-1-\sqrt{1+a}\right)+\left(-1-\sqrt{1+a}\right)^2==a\right\},\ \left\{2\left(-1+\sqrt{1+a}\right)+\left(-1+\sqrt{1+a}\right)^2==a,\ 2\left(-1+\sqrt{1+a}\right)+\left(-1+\sqrt{1+a}\right)^2==a\right\}\right\}$$

Mathematica braucht wenigstens den Hinweis, dass die linken Seiten zu expandieren sind.

$$\left(1-\sqrt{-3+a}\right)^2+2\left(1+\sqrt{-3+a}\right)\ \texttt{// Expand}$$

a

```
sys /. sol // Expand
```

{{True, True}, {True, True}, {True, True}, {True, True}}

Dann kann es sein, dass *Mathematica* zwar Aussagen vorgelegt bekommt, aber mit den eingesetzten Mitteln deren Wahrheitswert nicht entscheiden kann.

Die folgende Identität erkennt *Mathematica* Version 6 inzwischen automatisch.

$$\sqrt{11-6\sqrt{2}}+\sqrt{11+6\sqrt{2}}==6$$

True

Aber schon beim nächsten Beispiel muss ein so mächtiges Werkzeug wie **RootReduce** oder **FullSimplify** herangezogen werden.

$$\texttt{expr1} = \sqrt{3-\sqrt{5}}+\sqrt{3+\sqrt{5}}==\sqrt{10}$$

$$\sqrt{3-\sqrt{5}}+\sqrt{3+\sqrt{5}}==\sqrt{10}$$

```
expr1 // FullSimplify
```

True

Beide Beispiele gehören zu einer Familie von Identitäten der folgenden Bauart, die für a>b>0 gelten, was aber nicht einmal **FullSimplify** in der Lage ist nachzuprüfen.

```
expr2 = (Sqrt[a - Sqrt[a^2 - b^2]] + Sqrt[a + Sqrt[a^2 - b^2]] == Sqrt[2 (a + b)])
```

$$\sqrt{a-\sqrt{a^2-b^2}}+\sqrt{a+\sqrt{a^2-b^2}}=\sqrt{2}\sqrt{a+b}$$

```
FullSimplify[expr2, a > b > 0]
```

$$\sqrt{2}\sqrt{a+b}=\sqrt{a-\sqrt{a^2-b^2}}+\sqrt{a+\sqrt{a^2-b^2}}$$

Dabei müssen nur beide Seiten quadriert und vereinfacht werden.

```
Simplify[{expr2[[1]], expr2[[2]]}^2, a > b > 0]
```

{2 (a + b), 2 (a + b)}

Beispiele dieser Art können beliebig kompliziert sein. In den 1960er Jahren ist im Zusammenhang mit dem 10. Hilbertschen Problem bewiesen worden, dass die Frage der algorithmischen Bestimmung des Wahrheitswerts eines Booleschen Ausdrucks bereits in nicht sehr umfangreichen Klassen nicht entscheidbar ist.
Mit folgendem Beispiel kommt *Mathematica* 6 nun jedoch in beiden Varianten nach einiger Arbeit zurecht.

```
expr3 = Cos[π/7] - Cos[2 π/7] + Cos[3 π/7]
```

$$\mathrm{Cos}\left[\frac{\pi}{7}\right]+\mathrm{Sin}\left[\frac{\pi}{14}\right]-\mathrm{Sin}\left[\frac{3\pi}{14}\right]$$

```
expr3 // FullSimplify
```

$$\frac{1}{2}$$

```
expr4 = expr3 // TrigToExp
```

$$\frac{1}{2}\,i\,e^{-\frac{i\pi}{14}}-\frac{1}{2}\,i\,e^{\frac{i\pi}{14}}+\frac{1}{2}\,e^{-\frac{i\pi}{7}}+\frac{1}{2}\,e^{\frac{i\pi}{7}}-\frac{1}{2}\,i\,e^{-\frac{3i\pi}{14}}+\frac{1}{2}\,i\,e^{\frac{3i\pi}{14}}$$

```
expr4 // N
```

0.5 + 0. i

```
expr4 // FullSimplify
```

$$\frac{1}{2}$$

Zur Untersuchung der allgemeingültigen syntaktischen Gleichwertigkeit Boolescher Ausdrücke mit Symbolvariablen dient der Operator === (**SameQ**). Beachten Sie aber, dass natürlich auch hier vor dem Vergleich die Argumente von **SameQ** ausgewertet werden.

```
x + y === y + x
```

```
True
```

```
x + x === 2 x
```

```
True
```

x^2 `+ x === x (x + 1)`

```
False
```

=== (**SameQ**) und == (**Equal**) reagieren in diesem Beispiel unterschiedlich.

x^2 `+ x === x (x + 1) // Expand`

```
False
```

x^2 `+ x == x (x + 1) // Expand`

```
True
```

Auch in Steuerstrukturen kann es geschehen, dass Boolesche Ausdrücke nicht vollständig ausgewertet werden können.

Verzweigungen werden dann in symbolischer Form als Funktionsausdruck zurückgegeben.

```
u = If[x == y, 1, 2]
```

```
If[x == y, 1, 2]
```

```
u /. x → y
```

```
1
```

```
u /. x → y + 1
```

```
If[1 + y == y, 1, 2]
```

```
% // Simplify
```

```
2
```

Schleifenkörper werden dagegen nur dann betreten, wenn die Testbedingung explizit zu **True** auswertet.

```
Clear[n];
i = 0; While[i ≤ n, i++; Print[i]]; i
```

```
0
```

Dies kann zu falschen Ergebnissen führen, wenn eine Aussage als Testbedingung zwar den Wahrheitswert **True** hat, dies aber von *Mathematica* nicht erkannt wird.

$$\sqrt{2+\sqrt{3}}+\sqrt{2-\sqrt{3}}==\sqrt{6}$$

$$\sqrt{2-\sqrt{3}}+\sqrt{2+\sqrt{3}}==\sqrt{6}$$

i = 5; While$\left[\sqrt{i}\leq\sqrt{2+\sqrt{3}}+\sqrt{2-\sqrt{3}}\text{, i++}\right]$**; i**

N::meprec :
 Internal precision limit \$MaxExtraPrecision = 50.‘ reached
 while evaluating $\sqrt{6}-\sqrt{2-\sqrt{3}}-\sqrt{2+\sqrt{3}}$.

6

s = $\sqrt{2+\sqrt{3}}+\sqrt{2-\sqrt{3}}\leq\sqrt{6}$

N::meprec :
 Internal precision limit \$MaxExtraPrecision = 50.‘ reached
 while evaluating $-\sqrt{6}+\sqrt{2-\sqrt{3}}+\sqrt{2+\sqrt{3}}$.

$$\sqrt{2-\sqrt{3}}+\sqrt{2+\sqrt{3}}\leq\sqrt{6}$$

Interessanterweise wird als Ergebnis der numerischen Näherung **True** zurückgegeben, obwohl dies auf Grund von Rundungsungenauigkeiten numerisch niemals genau festgestellt werden kann.

```
s // N
```

True

```
s[[1]] - s[[2]] // N
```

4.44089×10^{-16}

Mehr noch, wir bekommen damit zwei Ausdrücke, von denen *Mathematica* behauptet, dass sie numerisch gleich seien, ihre Differenz aber nicht numerisch gleich Null.

```
s[[1]] == s[[2]] // N
```

True

```
s[[1]] - s[[2]] == 0 // N
```

False

```
(s[[1]] - s[[2]] // N // Chop) == 0
```

True

Mit zusätzlichem Aufwand kann *Mathematica* den Ausdruck zu **True** vereinfachen und liefert dann auch das korrekte Ergebnis.

```
s // RootReduce
```

True

$$i = 5;\ \mathrm{While}\left[\sqrt{i} \le \mathrm{RootReduce}\left[\sqrt{2+\sqrt{3}} + \sqrt{2-\sqrt{3}}\right],\ i{+}{+};\ \mathrm{Print}[i]\right];\ i$$

6

7

7

Funktionen definieren

Transformationen, Regeln, Muster

```
ClearAll["Global`*"]
```

```
i1[f[x_]] := f[x + 1]
```

$$\mathrm{Map}\left[i1, \left\{f[x], f[2, 3], g[y], f\left[\frac{1}{x}\right]\right\}\right]$$

$$\left\{f[1 + x], i1[f[2, 3]], i1[g[y]], f\left[1 + \frac{1}{x}\right]\right\}$$

```
?? i1
```

Global`i1

i1[f[x_]] := f[x + 1]

```
i2[_[y_]] := y / x
```

$$\mathrm{Map}\left[i2, \left\{f[x], f[2, 3], g[y], g\left[\frac{1}{x}\right]\right\}\right]$$

$$\left\{1, i2[f[2, 3]], \frac{y}{x}, \frac{1}{x^2}\right\}$$

```
?? i2
```

Global`i2

$i2[\_[y\_]] := \frac{y}{x}$

Partiell definierte Funktionen

```
l = {7, -3, √12, 1.7, x};

Clear[ourBinomial];
ourBinomial[n_, k_] := n! / (k! (n - k)!)

Map[ourBinomial[#, 2] &, l]
```

∞::indet : Indeterminate expression 0 ComplexInfinity encountered.

$$\left\{21,\ \text{Indeterminate},\ \frac{\left(2\sqrt{3}\right)!}{2\left(-2+2\sqrt{3}\right)!},\ 0.595,\ \frac{x!}{2\,(-2+x)!}\right\}$$

```
Map[Binomial[#, 2] &, l]
```

$$\left\{21,\ 6,\ \sqrt{3}\left(-1+2\sqrt{3}\right),\ 0.595,\ \frac{1}{2}(-1+x)\,x\right\}$$

Bedingte Funktionsvorschriften

```
Clear[ourBinomial];
ourBinomial[n_, k_] := n! / (k! (n - k)!) /; IntegerQ[n] ∧ IntegerQ[k] ∧ n ≥ 0 ∧ k ≥ 0

Map[ourBinomial[#, 2] &, l]
```

$$\left\{21,\ \text{ourBinomial}[-3,2],\ \text{ourBinomial}\left[2\sqrt{3},2\right],\ \text{ourBinomial}[1.7,2],\ \text{ourBinomial}[x,2]\right\}$$

Bedingte Muster

```
Clear[ourBinomial];
ourBinomial[n_?NonNegative, k_?NonNegative] := Product[(n - i + 1)/i, {i, 1, k}]

Map[NonNegative, l]
```

{True, False, True, True, NonNegative[x]}

```
Map[ourBinomial[#, 2] &, l]
```

$$\left\{21,\ \text{ourBinomial}[-3,2],\ \sqrt{3}\left(-1+2\sqrt{3}\right),\ 0.595,\ \text{ourBinomial}[x,2]\right\}$$

```
l1 = {1.5, ∞, √12};
NonNegative /@ l1
NumberQ /@ l1
NumericQ /@ l1
```

{True, True, True}

{True, False, False}

{True, False, True}

```
ourBinomial[15, #] & /@ {3.3, 3}
```

{455, 455}

```
Binomial[15, #] & /@ {3.3, 3}
```

{655.375, 455}

```
Clear[ourBinomial];
ourBinomial[n_, k_?(IntegerQ[#] ∧ NonNegative[#] &)] := Product[(n - i + 1)/i, {i, 1, k}]
```

```
Map[ourBinomial[#, 2] &, l]
```

$\left\{21,\ 6,\ \sqrt{3}\ \left(-1+2\ \sqrt{3}\right),\ 0.595,\ \frac{1}{2}\ (-1+x)\ x\right\}$

Typmuster

```
Clear[ourBinomial];
ourBinomial[n_, k_Integer?NonNegative] := Product[(n - i + 1)/i, {i, 1, k}]
```

```
Map[ourBinomial[#, 2] &, l]
```

$\left\{21,\ 6,\ \sqrt{3}\ \left(-1+2\ \sqrt{3}\right),\ 0.595,\ \frac{1}{2}\ (-1+x)\ x\right\}$

```
Clear[f]
f[x_Integer] := x
f /@ l
```

$\left\{7,\ -3,\ f\left[2\ \sqrt{3}\right],\ f[1.7],\ f[x]\right\}$

```
Clear[f]
f[x_Real] := x
f /@ l
```

$\left\{f[7],\ f[-3],\ f\left[2\ \sqrt{3}\right],\ 1.7,\ f[x]\right\}$

```
f[x_Integer] := x
f /@ l
```

$\left\{7,\ -3,\ f\left[2\ \sqrt{3}\right],\ 1.7,\ f[x]\right\}$

```
?f
```

```
Global`f

f[x_Real] := x

f[x_Integer] := x
```

$\sqrt{2}$ `// InputForm`

```
Sqrt[2]
```

```
% // Head
```

```
Power
```

$\sqrt{2}$ `// FullForm`

```
Power[2, Rational[1, 2]]
```

```
f[x_?NumericQ] := f[N[x]]
f /@ l
```

```
{7, -3, 3.4641, 1.7, f[x]}
```

Funktionen mit variabler Parameterzahl

`relSum[a_, b_] :=` $\frac{a+b}{1+a\,b}$

```
relSum[a_, b_, c__] := relSum[a, relSum[b, c]] // Together
```

```
relSum[x, y, z]
```

$$\frac{x+y+z+x\,y\,z}{1+x\,y+x\,z+y\,z}$$

```
relSum[0.4, 0.4, 0.4, 0.4]
```

```
0.93473
```

```
relSum[a, 1, b]
```

```
1
```

Muster, Regeln, Substitutionen und lokale Zuweisungen

$a^2 + 2\,a\,b^2$ `/.` $\{a \to b^2 + 1,\ b \to 2\}$

$$8\,(1+b^2) + (1+b^2)^2$$

```
a^2 + 2 a b^2 //. {a → b^2 + 1, b → 2}
```

```
65
```

■ Stückweise zusammengesetzte Funktionen

■ Realisierung als bedingte Funktionsdefinitions

```
ClearAll["Global`*"]
f[x_] := x /; 0 ≤ x < 1
f[x_] := 1 /; 1 ≤ x < 2
f[x_] := 3 - x /; 2 ≤ x < 3
f[x_] := 0
```

```
?f
```

```
Global`f
```

```
f[x_] := x /; 0 ≤ x < 1

f[x_] := 1 /; 1 ≤ x < 2

f[x_] := 3 - x /; 2 ≤ x < 3

f[x_] := 0
```

```
Plot[f[x], {x, -1, 4}, AxesOrigin -> {-1, -0.2}]
```




```
{f[1], f[x] /. x → 1}
```

```
{1, 0}
```

```
f[x] /. x → 1 // Hold // FullForm
```

```
Hold[ReplaceAll[f[x], Rule[x, 1]]]
```

```
Clear[f]
f[x_] := x /; 0 ≤ x < 1
f[x_] := 1 /; 1 ≤ x < 2
f[x_] := 3 - x /; 2 ≤ x < 3
f[x_Real] := 0
```

```
{f[1], f[x] /. x → 1}
```

```
{1, 1}
```

```
Plot[f[x], {x, -1, 4}, AxesOrigin → {-1, -0.2}]
```




```
{f'[x], f'[0], f'[0.1], f'[0.5]}
```

```
{f'[x], f'[0], 0.959457, 1.}
```

```
{f'[x], f'[0], f'[0.1], f'[0.5]} // N
```

```
{f'[x], 0.5, 0.959457, 1.}
```

```
Plot[f'[x],  {x, -1, 4}, AxesOrigin → {-1, -0.2}]
```



Funktionsdefinition mit If oder Which

```
Clear[f];
f[x_] := Which[0 ≤ x < 1, x, 1 ≤ x < 2, 1, 2 ≤ x < 3, 3 - x, True, 0]

Plot[{f[x], f'[x] + 0.1}, {x, -1, 4},
 PlotStyle → {Automatic, Dashed}, AxesOrigin → {-1, -0.2}]
```




```
{f'[x], f'[0], f'[-0.1], f'[0.1]}
```

```
{Which[0 ≤ x < 1, 1, 1 ≤ x < 2, 0, 2 ≤ x < 3, -1, True, 0], 1, 0, 1}
```

Realisierung mit Piecewise

```
Clear[f];
f[x_] := Piecewise[{{x, 0 ≤ x < 1},
{1, 1 ≤ x < 2}, {3 - x, 2 ≤ x < 3}}, 0]
```

$$f[x\_] := \begin{cases} 1 & 1 \le x < 2 \\ 3 - x & 2 \le x < 3 \\ x & 0 \le x < 1 \end{cases}$$

```
?f
```

Global`f

$$f[x\_] := \begin{cases} 1 & 1 \le x < 2 \\ 3 - x & 2 \le x < 3 \\ x & 0 \le x < 1 \end{cases}$$

```
Plot[{f[x], f'[x] + 0.1}, {x, -1, 4},
 PlotStyle → {Automatic, Dashed}, AxesOrigin → {-1, -0.2}]
```




```
f'[x]
```

$$\begin{cases} 0 & x < 0 \\ 1 & 0 < x < 1 \\ 0 & 1 < x < 2 \\ -1 & 2 < x < 3 \\ 0 & x > 3 \\ \text{Indeterminate} & \text{True} \end{cases}$$

```
Limit[f'[x], x → 0, Direction → #] & /@ {-1, 1}
```

{1, 0}

```
g[x_] := { x           1 < x
         { 2 Sqrt[x] - 1   x ≤ 1
```

$$g[x\_] := \begin{cases} x & 1 < x \\ 2\sqrt{x} - 1 & x \leq 1 \end{cases}$$

```
Plot[g[x], {x, 0, 2}]
```




```
g'[x]
```

$$\begin{cases} \frac{1}{\sqrt{x}} & x < 1 \\ 1 & \text{True} \end{cases}$$

```
g''[x]
```

$$\begin{cases} -\frac{1}{2 x^{3/2}} & x < 1 \\ 0 & x > 1 \\ \text{Indeterminate} & \text{True} \end{cases}$$

```
PiecewiseExpand[Max[x, y] * UnitStep[x * y]]
```

$$\begin{cases} x & x y \geq 0 \,\&\&\, x - y \geq 0 \\ y & x y \geq 0 \,\&\&\, x - y < 0 \end{cases}$$

```
PiecewiseExpand[Round[x^2 * Floor[y]], 1 < x < 2 && 2 < y < 4]
```

$$\begin{cases} 2 & y < 3 \,\&\&\, x \le \frac{\sqrt{5}}{2} \\ 3 & \left(\frac{\sqrt{5}}{2} < x < \frac{\sqrt{7}}{2} \,\&\&\, y < 3\right) \,||\, \left(x < \sqrt{\frac{7}{6}} \,\&\&\, y \ge 3\right) \\ 4 & \left(y < 3 \,\&\&\, \frac{\sqrt{7}}{2} \le x \le \frac{3}{2}\right) \,||\, \left(y \ge 3 \,\&\&\, \sqrt{\frac{7}{6}} \le x \le \sqrt{\frac{3}{2}}\right) \\ 5 & \left(\frac{3}{2} < x < \frac{\sqrt{11}}{2} \,\&\&\, y < 3\right) \,||\, \left(\sqrt{\frac{3}{2}} < x < \sqrt{\frac{11}{6}} \,\&\&\, y \ge 3\right) \\ 6 & \left(y < 3 \,\&\&\, \frac{\sqrt{11}}{2} \le x \le \frac{\sqrt{13}}{2}\right) \,||\, \left(y \ge 3 \,\&\&\, \sqrt{\frac{11}{6}} \le x \le \sqrt{\frac{13}{6}}\right) \\ 7 & \left(\frac{\sqrt{13}}{2} < x < \frac{\sqrt{15}}{2} \,\&\&\, y < 3\right) \,||\, \left(\sqrt{\frac{13}{6}} < x < \sqrt{\frac{5}{2}} \,\&\&\, y \ge 3\right) \\ 8 & \left(y < 3 \,\&\&\, x \ge \frac{\sqrt{15}}{2}\right) \,||\, \left(y \ge 3 \,\&\&\, \sqrt{\frac{5}{2}} \le x \le \sqrt{\frac{17}{6}}\right) \\ 9 & \sqrt{\frac{17}{6}} < x < \sqrt{\frac{19}{6}} \,\&\&\, y \ge 3 \\ 10 & y \ge 3 \,\&\&\, \sqrt{\frac{19}{6}} \le x \le \sqrt{\frac{7}{2}} \\ 11 & \sqrt{\frac{7}{2}} < x < \sqrt{\frac{23}{6}} \,\&\&\, y \ge 3 \\ 12 & \text{True} \end{cases}$$

```
h[x_] = Which[0 ≤ x < 1, x, 1 ≤ x < 2, 1, 2 ≤ x < 3, 3 - x] // PiecewiseExpand
```

$$\begin{cases} 1 & 1 \le x < 2 \\ 3 - x & 2 \le x < 3 \\ x & 0 \le x < 1 \end{cases}$$

```
h'[x]
```

$$\begin{cases} 0 & x < 0 \\ 1 & 0 < x < 1 \\ 0 & 1 < x < 2 \\ -1 & 2 < x < 3 \\ 0 & x > 3 \\ \text{Indeterminate} & \text{True} \end{cases}$$

■ Funktionen von Funktionen

```
ClearAll["Global`*"]
```

$\partial_x \mathrm{Sin}[x^2]$

$2\, x \,\mathrm{Cos}[x^2]$

```
∂x f[x^2]
```

$2\,x\,f'[x^2]$

```
% // FullForm
```

```
Times[2, x, Derivative[1][f][Power[x, 2]]]
```

```
f' // FullForm
```

```
Derivative[1][f]
```

```
Sin'
```

```
Cos[#1] &
```

```
%[x]
```

```
Cos[x]
```

Namenlose Funktionen

```
Log'
```

$\frac{1}{\#1}$ &

```
ArcTan'
```

$\frac{1}{1+\#1^2}$ &

```
% // FullForm
```

```
Function[Power[Plus[1, Power[Slot[1], 2]], -1]]
```

```
dlog = Log'
```

$\frac{1}{\#1}$ &

```
dlog[a]
```

$\frac{1}{a}$

```
elog = Function[{x}, 1 / x]; elog[b]
```

$\frac{1}{b}$

Solche namenlosen Funktionen treten überall dort auf, wo *Mathematica* selbst Funktionen erzeugt; etwa in der Ausgabe der Lösung von Differenzialgleichungen.

```
lsg = DSolve[{y'[x] + y[x] == 0, y[0] == 1}, y, x]
```

```
{{y → Function[{x}, e^-x]}}
```

```
{y[x], y'[x]} /. lsg[[1]]
```

```
{e^-x, -e^-x}
```

Interpolationsfunktionen

```
Options[Interpolation]
```

```
{InterpolationOrder → 3, PeriodicInterpolation → False}
```

```
data = {0.53, 0.66, 0.24, 0.58, 0.92, 0.65, 0.19, 0.67, 0.52, 0.30};
```

```
f = Interpolation[data]
```

```
InterpolatingFunction[{{1., 10.}}, <>]
```

```
p1 = ListPlot[data, PlotStyle → PointSize[.02]];
p2 = Plot[{f[x], f'[x] - 1, f''[x] - 2.5}, {x, 1, 10}];
Show[{p1, p2}, PlotRange → {{1, 10}, {-4, 2}}, Axes → None]
```



```
f'[#] & /@ (3 + Range[-0.01, +0.01, 0.002])
```

```
{-0.265868, -0.264371, -0.26287, -0.261363, -0.259851,
 -0.258333, 0.0881851, 0.0897006, 0.091213, 0.0927223, 0.0942287}
```

Funktionen mit Gedächtnis

```
ClearAll["Global`*"]
```

```
fib[n_] := fib[n] =
  Switch[n, 0, 0, 1, 1, _, fib[n - 1] + fib[n - 2]]
```

```
? fib
```

```
Global`fib
```

```
fib[n_] := fib[n] = Switch[n,
   0, 0,
   1, 1,
   _, fib[n - 1] + fib[n - 2]]
```

```
fib[6]
```

```
8
```

```
? fib
```

```
Global`fib
```

```
fib[0] = 0

fib[1] = 1

fib[2] = 1

fib[3] = 2

fib[4] = 3

fib[5] = 5

fib[6] = 8

fib[n_] := fib[n] = Switch[n,
   0, 0,
   1, 1,
   _, fib[n - 1] + fib[n - 2]]
```

```
Clear[fib];
fib[0] = 0;
fib[1] = 1;
fib[n_] := fib[n] = fib[n - 1] + fib[n - 2]
```

```
? fib
```

```
Global`fib
```

```
fib[0] = 0

fib[1] = 1

fib[n_] := fib[n] = fib[n - 1] + fib[n - 2]
```

```
fib[6]
```

8

```
? fib
```

Global`fib

fib[0] = 0

fib[1] = 1

fib[2] = 1

fib[3] = 2

fib[4] = 3

fib[5] = 5

fib[6] = 8

fib[n_] := fib[n] = fib[n - 1] + fib[n - 2]

```
fib[1000]
```

$RecursionLimit::reclim : Recursion depth of 256 exceeded.

$RecursionLimit::reclim : Recursion depth of 256 exceeded.

```
54 122 222 371 037 658 776 676 579 571 233 761 483 351 206 693 809 497
  Hold[fib[745 - 1] + fib[745 - 2]] +
 87 571 595 343 018 854 458 033 386 304 178 158 174 356 588 264 390 370
  Hold[fib[746 - 1] + fib[746 - 2]]
```

Numerische Funktionen

```
Clear[f]
f[x_] := Normal[Series[Sin[x^2], {x, 0, 20}]];
```

```
f[x]
```

$$x^2 - \frac{x^6}{6} + \frac{x^{10}}{120} - \frac{x^{14}}{5040} + \frac{x^{18}}{362\,880}$$

```
f[2] // N
```

General::ivar : 2 is not a valid variable.

General::ivar : 2 is not a valid variable.

-0.756802

```
f[x] /. x -> 2 // N
```

-0.661728

```
?f
```

Global`f

$f[x\_] := \text{Normal}[\text{Series}[\text{Sin}[x^2], \{x, 0, 20\}]]$

$f[x\_] = \text{Normal}[\text{Series}[\text{Sin}[x^2], \{x, 0, 20\}]];$

```
f[2] // N
```

-0.661728

```
?f
```

Global`f

$f[x\_] = x^2 - \frac{x^6}{6} + \frac{x^{10}}{120} - \frac{x^{14}}{5040} + \frac{x^{18}}{362880}$

$\text{Plot}[\{\text{Sin}[x^2], f[x]\}, \{x, -\pi, \pi\}, \text{PlotRange} \to \{1.1, -2\}]$




```
f[2]
```

$-\frac{268}{405}$

```
f[a]
```

$a^2 - \frac{a^6}{6} + \frac{a^{10}}{120} - \frac{a^{14}}{5040} + \frac{a^{18}}{362880}$

```
Clear[f]
```

$N[f[x\_]] = \text{Normal}[\text{Series}[\text{Sin}[x^2], \{x, 0, 40\}]];$

```
?f
```

Global`f

$$f/: N[f[x\_], \{MachinePrecision, MachinePrecision\}] = x^2 - \frac{x^6}{6} + \frac{x^{10}}{120} - \frac{x^{14}}{5040} + \frac{x^{18}}{362\,880} - \frac{x^{22}}{39\,916\,800} + \frac{x^{26}}{6\,227\,020\,800} - \frac{x^{30}}{1\,307\,674\,368\,000} + \frac{x^{34}}{355\,687\,428\,096\,000} - \frac{x^{38}}{121\,645\,100\,408\,832\,000}$$

```
f[2]
```

f[2]

```
f[a]
```

f[a]

```
f[2] // N
```

-0.756803

```
f[a] // N
```

$$a^2 - 0.166667\, a^6 + 0.00833333\, a^{10} - 0.000198413\, a^{14} + 2.75573\times 10^{-6}\, a^{18} - 2.50521\times 10^{-8}\, a^{22} + 1.6059\times 10^{-10}\, a^{26} - 7.64716\times 10^{-13}\, a^{30} + 2.81146\times 10^{-15}\, a^{34} - 8.22064\times 10^{-18}\, a^{38}$$

```
N[f[2], 20]
```

f[2.0000000000000000000]

Zusatzdefinition, damit f in beliebiger Genauigkeit berechnet werden kann

```
N[f[x_], _] = Normal[Series[Sin[x^2], {x, 0, 40}]];
```

```
?f
```

Global`f

$$f/: N[f[x\_], \{MachinePrecision, MachinePrecision\}] = x^2 - \frac{x^6}{6} + \frac{x^{10}}{120} - \frac{x^{14}}{5040} + \frac{x^{18}}{362\,880} - \frac{x^{22}}{39\,916\,800} + \frac{x^{26}}{6\,227\,020\,800} - \frac{x^{30}}{1\,307\,674\,368\,000} + \frac{x^{34}}{355\,687\,428\,096\,000} - \frac{x^{38}}{121\,645\,100\,408\,832\,000}$$

$$N[f[x\_], \_] = x^2 - \frac{x^6}{6} + \frac{x^{10}}{120} - \frac{x^{14}}{5040} + \frac{x^{18}}{362\,880} - \frac{x^{22}}{39\,916\,800} + \frac{x^{26}}{6\,227\,020\,800} - \frac{x^{30}}{1\,307\,674\,368\,000} + \frac{x^{34}}{355\,687\,428\,096\,000} - \frac{x^{38}}{121\,645\,100\,408\,832\,000}$$

```
N[f[2], 20]
```

-0.75680257873961450259

```
Plot[{Sin[x^2], f[x]}, {x, -π, π}, PlotRange → {1, -2}]
```