Listen

Listen im alltäglichen Umgang mit *Mathematica*

```
ClearAll["Global`*"]
```

```
Solve[x^2 == 1, x]
```

```
{{x → -1}, {x → 1}}
```

```
Solve[{2 x + 4 y == 6, x - y == 0}, {x, y}]
```

```
{{x → 1, y → 1}}
```

```
Plot[{Sin[x], Sin[3 x]/3, Sin[5 x]/5}, {x, 0, π}]
```



Iteratoren

```
Table[i^2, {3}]
```

```
{i^2, i^2, i^2}
```

```
Table[i^2, {i, 3}]
```

```
{1, 4, 9}
```

```
Table[i^2, {i, 2, 4}]
```

```
{4, 9, 16}
```

```
Table[i^2, {i, 2, 4, 1/2}]
```

$$\left\{4, \frac{25}{4}, 9, \frac{49}{4}, 16\right\}$$

```
Table[i^2, {i, 4, 2, -1}]
```

```
{16, 9, 4}
```

```
Range[5]
```

```
{1, 2, 3, 4, 5}
```

```
Range[3, 5]
```

```
{3, 4, 5}
```

```
Range[3, 5, 0.3]
```

```
{3., 3.3, 3.6, 3.9, 4.2, 4.5, 4.8}
```

Listen erzeugen

```
l = {1, 2, 3, 4, 5}
```

```
{1, 2, 3, 4, 5}
```

```
l = Table[x^2, {x, 0, 4}]
```

```
{0, 1, 4, 9, 16}
```

```
Map[#^2 &, Range[0, 4]]
```

```
{0, 1, 4, 9, 16}
```

```
l = Table[x^2, {x, 0, 50, 10}]
```

```
{0, 100, 400, 900, 1600, 2500}
```

```
l = Table[x^y, {x, 0, 3}, {y, 2, 4}]
```

```
{{0, 0, 0}, {1, 1, 1}, {4, 8, 16}, {9, 27, 81}}
```

```
l = Table[Table[x^y, {y, 2, 4}], {x, 0, 3}]
```

```
{{0, 0, 0}, {1, 1, 1}, {4, 8, 16}, {9, 27, 81}}
```

```
l // Grid
```

```
0  0   0
1  1   1
4  8  16
9 27 81
```

```
Table[n^m, {n, 1, 3}, {m, 1, n}]
```

```
{{1}, {2, 4}, {3, 9, 27}}
```

```
Table[Table[n^m, {m, 1, n}], {n, 1, 3}]
```

```
{{1}, {2, 4}, {3, 9, 27}}
```

```
l := Table[f[x^n], {n, 1, 5}];
l
```

$$\{f[x], f[x^2], f[x^3], f[x^4], f[x^5]\}$$

```
f[x_] := x^3;
l
```

$$\{x^3, x^6, x^9, x^{12}, x^{15}\}$$

```
Table[x[i], {i, 1, 5}]
```

```
{x[1], x[2], x[3], x[4], x[5]}
```

```
Array[x, 5]
```

```
{x[1], x[2], x[3], x[4], x[5]}
```

```
l = {{1, 2}, {3, 4, 5}, {6, 7, 8, 9}};
```

```
Length[l]
```

```
3
```

```
Length[l[[3]]]
```

```
4
```

```
Length[a + b + c]
```

```
3
```

```
Depth[{a, b, c}]
```

```
2
```

```
Depth[l]
```

```
3
```

```
{Depth[a + b c], Depth[a + b / c]}
```

```
{3, 4}
```

```
Depth[{a, b, c + d}]
```

```
3
```

Auf Listenelemente zugreifen

```
l = {a, b, c, d};
```

```
l[[2]]
```

```
b
```

```
l[[-1]]
```

```
d
```

```
l[[0]]
```

```
List
```

```
(a + b)[[0]]
```

```
Plus
```

```
l[[{2, 4}]]
```

```
{b, d}
```

```
Part[l, 2]
```

```
b
```

```
Part[l, {2, 4}]
```

```
{b, d}
```

```
Part[l, Range[2, 4]]
```

```
{b, c, d}
```

```
l[[2 ;; 4]]
```

```
{b, c, d}
```

```
l = {{a, b, c}, {d, e, f}, {g, h, i}};
```

```
l // Grid
```

```
a b c
d e f
g h i
```

```
l[[1]]
```

```
{a, b, c}
```

```
l[[1, 3]]
```

```
c
```

```
Part[l, 1, 3]
```

```
c
```

```
l[[{2, 3}, {1, 2}]]
```

```
{{d, e}, {g, h}}
```

```
Part[l, {2, 3}, {1, 2}]
```

```
{{d, e}, {g, h}}
```

```
l = {1, a, b^c, √3};
```

```
{Head[l], First[l], Last[l]}
```

```
{List, 1, √3}
```

```
Head[a + b]
```

```
Plus
```

Funktionen und Listen

```
ClearAll["Global`*"]
```

```
Map[f, {a, b, c}]
```

```
{f[a], f[b], f[c]}
```

```
f /@ {a, b, c}
```

```
{f[a], f[b], f[c]}
```

```
5 {a, b, c}
```

```
{5 a, 5 b, 5 c}
```

```
Attributes[Plus]
```

```
{Flat, Listable, NumericFunction, OneIdentity, Orderless, Protected}
```

```
{a, b, c} + {d, e, f}
```

```
{a + d, b + e, c + f}
```

```
l = Table[i/(i + 1), {i, 5}]
```

$$\left\{\frac{1}{2}, \frac{2}{3}, \frac{3}{4}, \frac{4}{5}, \frac{5}{6}\right\}$$

```
l // N
```

```
{0.5, 0.666667, 0.75, 0.8, 0.833333}
```

```
x^{1,2,3,4,5}
```

$$\{x, x^2, x^3, x^4, x^5\}$$

```
Apply[f, {a, b, c}]
```

```
f[a, b, c]
```

```
f @@ {a, b, c}
```

```
f[a, b, c]
```

```
Map[f, a + b + c]
```

```
f[a] + f[b] + f[c]
```

```
f /@ g[a, b, c]
```

```
g[f[a], f[b], f[c]]
```

```
Apply[f, a + b + c]
```

```
f[a, b, c]
```

```
f @@ g[a, b, c]
```

```
f[a, b, c]
```

▪ Listen manipulieren

```
ClearAll["Global`*"]
```

```
l = {1, 2, 3, 4};
```

```
Append[l, 7]
```

```
{1, 2, 3, 4, 7}
```

```
l
```

```
{1, 2, 3, 4}
```

```
Prepend[l, 7]
```

```
{7, 1, 2, 3, 4}
```

```
{First[l], Rest[l]}
```

```
{1, {2, 3, 4}}
```

```
Join[{1, 2, 3, 4}, {6, 4, 5, 3}]
```

```
{1, 2, 3, 4, 6, 4, 5, 3}
```

```
l = {{{a, b}, {c, d}}, {{e, f}, {g, h}}};
```

```
Flatten[l]
```

```
{a, b, c, d, e, f, g, h}
```

```
Flatten[l, 1]
```

```
{{a, b}, {c, d}, {e, f}, {g, h}}
```

```
l = Table[Random[Integer, 100], {50}];
Partition[l, 5]
```

```
{{84, 80, 9, 32, 99}, {18, 29, 75, 14, 84}, {78, 8, 47, 81, 34},
 {0, 66, 95, 83, 69}, {22, 47, 11, 39, 59}, {62, 42, 91, 80, 66},
 {18, 16, 100, 66, 66}, {93, 48, 89, 27, 49}, {99, 6, 90, 31, 50}, {6, 85, 7, 15, 59}}
```

```
l[[#]] & /@ Table[5 i + j, {i, 0, 9}, {j, 1, 5}]
```

```
{{84, 80, 9, 32, 99}, {18, 29, 75, 14, 84}, {78, 8, 47, 81, 34},
 {0, 66, 95, 83, 69}, {22, 47, 11, 39, 59}, {62, 42, 91, 80, 66},
 {18, 16, 100, 66, 66}, {93, 48, 89, 27, 49}, {99, 6, 90, 31, 50}, {6, 85, 7, 15, 59}}
```

```
l = {{a, b, c}, {d, e, f}}
```

```
{{a, b, c}, {d, e, f}}
```

```
Thread[l]
```

```
{{a, d}, {b, e}, {c, f}}
```

```
Transpose[l]
```

```
{{a, d}, {b, e}, {c, f}}
```

```
l[[{1, 2}, #]] & /@ Range[3]
```

```
{{a, d}, {b, e}, {c, f}}
```

```
{1, 2, 3} + {a, b, c}
```

```
{1 + a, 2 + b, 3 + c}
```

```
{1, 2, 3} == {a, b, c}
```

```
{1, 2, 3} == {a, b, c}
```

```
{1, 2, 3} == {a, b, c} // Thread
```

```
{1 == a, 2 == b, 3 == c}
```

```
glsys = {x + y, y - z, z + 2 x} == {1, 2, 3};
glsys // Thread
```

$$\{x + y == 1,\ y - z == 2,\ 2\,x + z == 3\}$$

```
Solve[glsys, {x, y, z}]
```

$$\{\{x \to 4,\ y \to -3,\ z \to -5\}\}$$

■ Substitutionslisten

```
ClearAll["Global`*"]
```

```
gls = {x^2 + y == 3, y^2 + x == 3}
```

$$\{x^2 + y == 3,\ x + y^2 == 3\}$$

```
sol = Solve[gls, {x, y}]
```

$$\left\{\{x \to -1,\ y \to 2\},\ \{x \to 2,\ y \to -1\},\ \left\{x \to \frac{1}{2}\left(-1 - \sqrt{13}\right),\ y \to \frac{1}{2}\left(-1 - \sqrt{13}\right)\right\},\ \left\{x \to -\frac{1}{2} + \frac{\sqrt{13}}{2},\ y \to \frac{1}{2}\left(-1 + \sqrt{13}\right)\right\}\right\}$$

```
Map[{x, y} /. # &, sol]
```

$$\left\{\{-1, 2\},\ \{2, -1\},\ \left\{\frac{1}{2}\left(-1 - \sqrt{13}\right),\ \frac{1}{2}\left(-1 - \sqrt{13}\right)\right\},\ \left\{-\frac{1}{2} + \frac{\sqrt{13}}{2},\ \frac{1}{2}\left(-1 + \sqrt{13}\right)\right\}\right\}$$

```
{x, y} /. sol
```

$$\left\{\{-1, 2\},\ \{2, -1\},\ \left\{\frac{1}{2}\left(-1 - \sqrt{13}\right),\ \frac{1}{2}\left(-1 - \sqrt{13}\right)\right\},\ \left\{-\frac{1}{2} + \frac{\sqrt{13}}{2},\ \frac{1}{2}\left(-1 + \sqrt{13}\right)\right\}\right\}$$

```
Map[Table[x^i + y^i, {i, 1, 4}] /. # &, sol] // Expand
```

$$\left\{\{1, 5, 7, 17\},\ \{1, 5, 7, 17\},\ \left\{-1 - \sqrt{13},\ 7 + \sqrt{13},\ -10 - 4\sqrt{13},\ 31 + 7\sqrt{13}\right\},\ \left\{-1 + \sqrt{13},\ 7 - \sqrt{13},\ -10 + 4\sqrt{13},\ 31 - 7\sqrt{13}\right\}\right\}$$

```
Table[x^i + y^i, {i, 1, 4}] /. sol // Expand
```

$$\left\{\{1, 5, 7, 17\},\ \{1, 5, 7, 17\},\ \left\{-1 - \sqrt{13},\ 7 + \sqrt{13},\ -10 - 4\sqrt{13},\ 31 + 7\sqrt{13}\right\},\ \left\{-1 + \sqrt{13},\ 7 - \sqrt{13},\ -10 + 4\sqrt{13},\ 31 - 7\sqrt{13}\right\}\right\}$$

■ Beispiel: Nullstellen eines Polynoms über Z_p

```
ClearAll["Global`*"]
```

```
primeList = Select[Range[50], PrimeQ]
```

```
{2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47}
```

```
werteTafel = Table[{x, Mod[x^3 + x + 1, 31]}, {x, 0, 30}]
```

```
{{0, 1}, {1, 3}, {2, 11}, {3, 0}, {4, 7}, {5, 7}, {6, 6}, {7, 10},
 {8, 25}, {9, 26}, {10, 19}, {11, 10}, {12, 5}, {13, 10}, {14, 0}, {15, 12},
 {16, 21}, {17, 2}, {18, 23}, {19, 28}, {20, 23}, {21, 14}, {22, 7}, {23, 8},
 {24, 23}, {25, 27}, {26, 26}, {27, 26}, {28, 2}, {29, 22}, {30, 30}}
```

```
Select[werteTafel, #[[2]] == 0 &]
```

```
{{3, 0}, {14, 0}}
```

```
Map[#[[1]] &, %]
```

```
{3, 14}
```

```
Select[Range[0, 30], Mod[#^3 + # + 1, 31] == 0 &]
```

```
{3, 14}
```

```
sol[p_?PrimeQ] := Select[Range[0, p - 1], Mod[#^3 + # + 1, p] == 0 &]
```

```
?sol
```

```
Global`sol
```

```
sol[p_?PrimeQ] := Select[Range[0, p - 1], Mod[#1^3 + #1 + 1, p] == 0 &]
```

```
sol[13]
```

```
{7}
```

```
{#, sol[#]} & /@ primeList
```

```
{{2, {}}, {3, {1}}, {5, {}}, {7, {}}, {11, {2}}, {13, {7}}, {17, {11}}, {19, {}}, {23, {4}},
 {29, {26}}, {31, {3, 14}}, {37, {25}}, {41, {}}, {43, {38}}, {47, {25, 34, 35}}}
```

```
{#, Factor[x^3 + x + 1, Modulus → #]} & /@ primeList
```

$$\{\{2, 1+x+x^3\}, \{3, (2+x)(2+x+x^2)\}, \{5, 1+x+x^3\},$$
$$\{7, 1+x+x^3\}, \{11, (9+x)(5+2x+x^2)\}, \{13, (6+x)(11+7x+x^2)\},$$
$$\{17, (6+x)(3+11x+x^2)\}, \{19, 1+x+x^3\}, \{23, (19+x)(17+4x+x^2)\},$$
$$\{29, (3+x)(10+26x+x^2)\}, \{31, (17+x)^2(28+x)\}, \{37, (12+x)(34+25x+x^2)\},$$
$$\{41, 1+x+x^3\}, \{43, (5+x)(26+38x+x^2)\}, \{47, (12+x)(13+x)(22+x)\}\}$$

■ Beispiel: Punkte auf einer elliptischen Kurve über Z_p

```
ClearAll["Global`*"]
```

```
l1 = Table[{x, y}, {x, 0, 10}, {y, 0, 10}];
Short[l1, 4]
```

```
{{{0, 0}, {0, 1}, {0, 2}, {0, 3}, {0, 4}, {0, 5}, {0, 6}, {0, 7}, {0, 8}, {0, 9}, {0, 10}},
 <<9>>, {{10, 0}, {10, 1}, {10, 2}, {10, 3}, {10, 4},
  {10, 5}, {10, 6}, {10, 7}, {10, 8}, {10, 9}, {10, 10}}}
```

```
l2 = Join @@ l1;
Short[l2, 5]
```

```
{{0, 0}, {0, 1}, {0, 2}, {0, 3}, {0, 4}, {0, 5}, {0, 6}, {0, 7}, {0, 8},
 {0, 9}, {0, 10}, {1, 0}, {1, 1}, {1, 2}, {1, 3}, {1, 4}, {1, 5}, <<87>>,
 {9, 5}, {9, 6}, {9, 7}, {9, 8}, {9, 9}, {9, 10}, {10, 0}, {10, 1}, {10, 2},
 {10, 3}, {10, 4}, {10, 5}, {10, 6}, {10, 7}, {10, 8}, {10, 9}, {10, 10}}
```

```
ec11 = Select[l2, Mod[#[[2]]^2 - #[[1]]^3 - 1, 11] == 0 &]
```

```
{{0, 1}, {0, 10}, {2, 3}, {2, 8}, {5, 4}, {5, 7}, {7, 5}, {7, 6}, {9, 2}, {9, 9}, {10, 0}}
```

```
ec11 = Select[Join @@ Table[{x, y}, {x, -5, 5}, {y, -5, 5}], Mod[#[[2]]^2 - #[[1]]^3 - 1, 11] == 0 &]
```

```
{{-4, -5}, {-4, 5}, {-2, -2}, {-2, 2},
 {-1, 0}, {0, -1}, {0, 1}, {2, -3}, {2, 3}, {5, -4}, {5, 4}}
```

```
ListPlot[ec11, PlotStyle → PointSize[0.02]]
```




```
ec[p_?PrimeQ] :=
 Select[Join @@ Table[{x, y}, {x, p}, {y, p}], Mod[#[[2]]^2 - #[[1]]^3 - 1, p] == 0 &]
```

```
primeList = Select[Range[50], PrimeQ]; Map[{#, Length[ec[#]]} &, primeList]
```

```
{{2, 2}, {3, 3}, {5, 5}, {7, 11}, {11, 11}, {13, 11}, {17, 17}, {19, 11},
 {23, 23}, {29, 29}, {31, 35}, {37, 47}, {41, 41}, {43, 35}, {47, 47}}
```

Teillisten und Muster

Select, MatchQ und Cases

```
l = {1, a, b^c, d^3, e + f, √3};
```

```
Cases[l, _-]
```

$\{b^c, d^3, \sqrt{3}\}$

```
Select[l, MatchQ[#, _-] &]
```

$\{b^c, d^3, \sqrt{3}\}$

```
Cases[l, x_^y_Integer]
```

$\{d^3\}$

```
Select[l, MatchQ[#, _-^Integer] &]
```

$\{d^3\}$

```
Cases[l, x_^y_Integer → x^2 y]
```

$\{d^6\}$

```
Select[l, MatchQ[#, _-^Integer] &] /. x_^y_Integer → x^2 y
```

$\{d^6\}$

```
DeleteCases[l, _-]
```

$\{1, a, e + f\}$

```
Select[l, ! MatchQ[#, _-] &]
```

$\{1, a, e + f\}$

```
u = 12 (1 + x)^5 // Expand
```

$12 + 60\,x + 120\,x^2 + 120\,x^3 + 60\,x^4 + 12\,x^5$

```
Cases[u, 12 _-]
```

$\{12\,x^5\}$

```
Cases[u, __ _]
```

$\{60\,x, 120\,x^2, 120\,x^3, 60\,x^4, 12\,x^5\}$

```
Select[u, MatchQ[#, _ __] &]
```

$$60\,x + 120\,x^2 + 120\,x^3 + 60\,x^4 + 12\,x^5$$

```
List @@ %
```

$$\{60\,x,\ 120\,x^2,\ 120\,x^3,\ 60\,x^4,\ 12\,x^5\}$$

```
l
```

▪ Weitere Listenkommandos

```
Cases[l, _Symbol]
```

$$\{a\}$$

```
Count[l, _Symbol]
```

```
1
```

```
MemberQ[l, _Times]
```

```
False
```

```
FreeQ[l, _Power]
```

```
False
```

```
FreeQ[l, _Times]
```

```
True
```

```
Count[u, _Times]
```

```
5
```

```
MemberQ[u, _Power]
```

```
False
```

```
FreeQ[u, _Power]
```

```
False
```

▪ Zusammengesetzte Muster und Defaultmuster

```
l = {5, x^2, a^(b+c), (a + b)^(c+d)};
```

```
Clear[f];
f[x_-] := x + 3;
Map[f, l]
```

$$\{f[5],\ 3 + x,\ 3 + a,\ 3 + a + b\}$$

```
Clear[f];
f[x : _-] := x + 3;
Map[f, l]
```

$$\{f[5],\ 3 + x^2,\ 3 + a^{b+c},\ 3 + (a + b)^{c+d}\}$$

```
l = {1, x, x^2, a, a x, a x^2};
```

```
Cases[l, x-]
```

$$\{x^2\}$$

```
Cases[l, x-·]
```

$$\{x,\ x^2\}$$

```
Cases[l, _ x-·]
```

$$\{a\,x,\ a\,x^2\}$$

```
Cases[l, _. x-·]
```

$$\{x,\ x^2,\ a\,x,\ a\,x^2\}$$

■ Spezielle listenbasierte Datenstrukturen

▪ Mengen

```
ClearAll["Global‘*"]
```

```
l1 = {1, 2, 3, 4}; l2 = {6, 5, 4, 3}; l3 = {1, 3, 2, 1, 2};
```

```
l1 ⋃ l2
```

```
{1, 2, 3, 4, 5, 6}
```

```
Union[l3]
```

```
{1, 2, 3}
```

```
l3 ⋃ {}
```

```
{1, 2, 3}
```

```
l1 ⋂ l2
```

```
{3, 4}
```

```
Complement[l1, l2]
```

```
{1, 2}
```

```
MemberQ[l1, 1]
```

```
True
```

Vektoren und Matrizen

```
ClearAll["Global`*"]
```

```
VectorQ[v = {1, 2, 3}]
```

```
True
```

```
VectorQ[{{1}, {2, 3}}]
```

```
False
```

```
MatrixQ[{{1, 2}, {3, 4}}]
```

```
True
```

```
MatrixQ[{{1, 2}, {3, 4, 5}}]
```

```
False
```

```
m = {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}, {10, 11, 12}};
m // MatrixForm
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \\ 10 & 11 & 12 \end{pmatrix}$$

```
m.v // MatrixForm
```

$$\begin{pmatrix} 14 \\ 32 \\ 50 \\ 68 \end{pmatrix}$$

```
v.v
```

```
14
```

```
Dimensions[m]
```

```
{4, 3}
```

```
(ml = Transpose[m]) // MatrixForm
```

$$\begin{pmatrix} 1 & 4 & 7 & 10 \\ 2 & 5 & 8 & 11 \\ 3 & 6 & 9 & 12 \end{pmatrix}$$

```
ml // Dimensions
```

```
{3, 4}
```

Weitere Beispiele

Komplexe Zahlen in Real− und Imaginärteil aufspalten

```
ClearAll["Global`*"]
```

```
l = Table[Random[Complex], {5}]
```

```
{0.0330212 + 0.465473 i, 0.155084 + 0.88999 i,
 0.821288 + 0.221346 i, 0.144917 + 0.0909742 i, 0.223424 + 0.8069 i}
```

```
Table[{Re[l[[i]]], Im[l[[i]]]}, {i, Length[l]}]
```

```
{{0.0330212, 0.465473}, {0.155084, 0.88999},
 {0.821288, 0.221346}, {0.144917, 0.0909742}, {0.223424, 0.8069}}
```

```
Map[ {Re[#], Im[#]} &, l]
```

```
{{0.0330212, 0.465473}, {0.155084, 0.88999},
 {0.821288, 0.221346}, {0.144917, 0.0909742}, {0.223424, 0.8069}}
```

```
{Re[#], Im[#]} & /@ l
```

```
{{0.0330212, 0.465473}, {0.155084, 0.88999},
 {0.821288, 0.221346}, {0.144917, 0.0909742}, {0.223424, 0.8069}}
```

```
{Re[l], Im[l]} // Thread
```

```
{{0.0330212, 0.465473}, {0.155084, 0.88999},
 {0.821288, 0.221346}, {0.144917, 0.0909742}, {0.223424, 0.8069}}
```

Komplexe Zahlen aus zwei Listen konstruieren

```
ClearAll["Global`*"]
```

```
l = {Table[Random[], {5}], Table[Random[], {5}]}
```

```
{{0.523902, 0.616014, 0.421656, 0.194263, 0.85307},
 {0.231122, 0.0517396, 0.461006, 0.957623, 0.667823}}
```

```
Table[l[[1, i]] + I l[[2, i]], {i, Length[l[[1]]]}]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
#[[1]] + I #[[2]] & /@ Thread[l]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
Map[Apply[#1 + I #2 &, #] &, Transpose[l]]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
Thread[{l[[1]], l[[2]]}] /. {x_, y_} -> x + I y
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
Apply[(#1 + I #2) &, l]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
Thread[(#1 + I #2) &[l[[1]], l[[2]]]]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
(#1 + I #2) &[l[[1]], l[[2]]]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

```
l[[1]] + I l[[2]]
```

```
{0.523902 + 0.231122 i, 0.616014 + 0.0517396 i,
 0.421656 + 0.461006 i, 0.194263 + 0.957623 i, 0.85307 + 0.667823 i}
```

Gleitender Durchschnitt

```
ClearAll["Global`*"]
```

```
l = Table[Random[Integer, 20], {10}]
```

```
{17, 11, 3, 0, 1, 10, 11, 6, 11, 12}
```

```
sublists = Table[l[[i + Range[4]]], {i, 0, 6}]
```

```
{{17, 11, 3, 0}, {11, 3, 0, 1}, {3, 0, 1, 10},
 {0, 1, 10, 11}, {1, 10, 11, 6}, {10, 11, 6, 11}, {11, 6, 11, 12}}
```

```
Partition[l, 4, 1]
```

```
{{17, 11, 3, 0}, {11, 3, 0, 1}, {3, 0, 1, 10},
 {0, 1, 10, 11}, {1, 10, 11, 6}, {10, 11, 6, 11}, {11, 6, 11, 12}}
```

```
sums = Apply[Plus, #] & /@ sublists
```

```
{31, 15, 14, 22, 28, 38, 40}
```

```
Apply[Plus, sublists, {1}]
```

```
{31, 15, 14, 22, 28, 38, 40}
```

```
sums / 4
```

$$\left\{\frac{31}{4}, \frac{15}{4}, \frac{7}{2}, \frac{11}{2}, 7, \frac{19}{2}, 10\right\}$$

```
gleitenderDurchschnitt[l_List, n_Integer?(# > 0 &)] :=
 (Apply[Plus, #] & /@ Table[l[[i + Range[n]]], {i, 0, Length[l] - n}]) / n
```

```
l1 = Table[{x, 2 Sin[x] + Random[]}, {x, 0, π, 0.01}];

l2 = gleitenderDurchschnitt[l1, 20];

p1 = ListLinePlot[l1, AxesOrigin → {0, 0}];
p2 = ListLinePlot[l2, AxesOrigin → {0, 0}] ;
GraphicsGrid[{{p1, p2}}]
```