Summen und Integrale

Folgen und Partialsummen

```
Table[Sum[i^k, {i, 1, n}], {k, 1, 6}]
```

$$\left\{\frac{1}{2} n (1+n), \frac{1}{6} n (1+n) (1+2 n), \frac{1}{4} n^2 (1+n)^2, \frac{1}{30} n (1+n) (1+2 n) \left(-1+3 n+3 n^2\right), \frac{1}{12} n^2 (1+n)^2 \left(-1+2 n+2 n^2\right), \frac{1}{42} n (1+n) (1+2 n) \left(1-3 n+6 n^3+3 n^4\right)\right\}$$

```
Sum[Binomial[n, k] x^k y^(n-k), {k, 0, n}]
```

$$y^n \left(\frac{x+y}{y}\right)^n$$

```
% // PowerExpand
```

$$(x+y)^n$$

```
Sum[Binomial[n, k], {k, 0, n}]
```

$$2^n$$

```
Sum[(-1)^k Binomial[n, k], {k, 0, n}]
```

KroneckerDelta[n]

Hypergeometrische Summen

```
res = Sum[Binomial[n, k]^2, {k, 0, n}]
```

$$\frac{4^n \text{Gamma}\left[\frac{1}{2}+n\right]}{\sqrt{\pi} \text{ Gamma}[1+n]}$$

```
n! - Gamma[n + 1] // Simplify
```

n! - Gamma[1 + n]

```
n! - Gamma[n + 1] // FullSimplify
```

0

```
res - Binomial[2 n, n] // Simplify
```

$$-\text{Binomial}[2 n, n]+\frac{4^n \text{Gamma}\left[\frac{1}{2}+n\right]}{\sqrt{\pi} \text{ Gamma}[1+n]}$$

```
% // FullSimplify
```

0

```
f[n_, k_] := Binomial[n, k]^2
```

Die Formel lässt sich über lineare Rekursionen und doppeltes Summieren beweisen.
f[n,k] etwa erfüllt die folgende zweistellige lineare Rekursionsbeziehung.

```
eqn1 = n f[n, k] - (2 n - 1) (f[n - 1, k] + f[n - 1, k - 1]) +
   (n - 1) (f[n - 2, k] - 2 f[n - 2, k - 1] + f[n - 2, k - 2])
```

```
(-1 + n) (Binomial[-2 + n, -2 + k]^2 - 2 Binomial[-2 + n, -1 + k]^2 + Binomial[-2 + n, k]^2) -
  (-1 + 2 n) (Binomial[-1 + n, -1 + k]^2 + Binomial[-1 + n, k]^2) + n Binomial[n, k]^2
```

Das kann mit *Mathematica* bis k=14 mit **FullSImplify** überprüft werden.

```
Table[eqn1, {k, 2, 14}] // FullSimplify
```

{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Bei größeren Werten für k muss noch **FunctionExpand** vorangestellt werden.

```
(eqn1 /. k → 15) // FullSimplify
% // FunctionExpand // FullSimplify
```

```
(-1 + n) (Binomial[-2 + n, 13]^2 - 2 Binomial[-2 + n, 14]^2 + Binomial[-2 + n, 15]^2) -
  (-1 + 2 n) (Binomial[-1 + n, 14]^2 + Binomial[-1 + n, 15]^2) + n Binomial[n, 15]^2
```

0

Interessanterweise hat **FullSimplify** mit demselben Ausdruck für allgemeine k keine Probleme.

```
eqn1 // FullSimplify
```

0

Summiert man diese Relation über alle k, so ergibt sich wegen $\binom{n}{k}$=0 für k<0 und k>n für

s[n] = $\sum_k f[n, k]$

die Beziehung

```
eqn2 = n s[n] == 2 (2 n - 1) s[n - 1]
```

n s[n] == 2 (-1 + 2 n) s[-1 + n]

RSolve liefert dann die eingangs von *Mathematica* gefundene Antwort.

```
sol = RSolve[{eqn2, s[0] == 1}, s[n], n]
```

$$\left\{\left\{s[n] \to \frac{2^{2n}\,\text{Gamma}\left[\frac{1}{2}+n\right]}{\sqrt{\pi}\,\text{Gamma}[1+n]}\right\}\right\}$$

Das bisher untersuchte Beispiel ist ein Spezialfall der folgenden allgemeineren Summationsaufgabe.

```
res = Sum[Binomial[n, k]^2 (1 + x)^k (1 - x)^(n-k), {k, 0, n}]
```

$$(1-x)^n\,\text{Hypergeometric2F1}\left[-n, -n, 1, \frac{1+x}{1-x}\right]$$

Das Ergebnis kann auch in Termini der (*Mathematica* ebenfalls bekannten) Legendre−Polynome **LegendreP** angegeben werden, wie ein Vergleich der Ergebnisse für einige Werte von n zeigt.

```
Table[res, {n, 1, 5}] // Together
Table[(2 x)^n LegendreP[n, 1 / x], {n, 1, 5}] // Together
```

$$\left\{2, -2\left(-3+x^2\right), -4\left(-5+3x^2\right), 2\left(35-30x^2+3x^4\right), 4\left(63-70x^2+15x^4\right)\right\}$$

$$\left\{2, -2\left(-3+x^2\right), -4\left(-5+3x^2\right), 2\left(35-30x^2+3x^4\right), 4\left(63-70x^2+15x^4\right)\right\}$$

Mathematica hat aber selbst mit **FullSimplify** Schwierigkeiten, dies in symbolischer Form zu erkennen.

```
res - (2 x)^n LegendreP[n, 1 / x] // FullSimplify
```

$$(1-x)^n\,\text{Hypergeometric2F1}\left[-n, -n, 1, \frac{1+x}{1-x}\right] - 2^n x^n\,\text{LegendreP}\left[n, \frac{1}{x}\right]$$

Für konkrete Werte von n dagegen wird bereits beim Einsetzen sowohl **Hypergeometric2F1** als auch **LegendreP** durch die entsprechende rationale Funktion ersetzt, so dass ein einfaches **Together** zum Nachweis der Identität ausreicht.

```
(res - (2 x)^n LegendreP[n, 1 / x]) /. n -> # & /@ Range[10]
% // Together
```

$$\Big\{-2-\frac{2\,(1-x)}{-1+x},\ -2\left(3-x^2\right)-\frac{2\,(1-x)^2\,(-3+x^2)}{(-1+x)^2},$$
$$-4\left(5-3\,x^2\right)+\frac{4\,(1-x)^3\,(-5+3\,x^2)}{(-1+x)^3},\ -2\left(35-30\,x^2+3\,x^4\right)+\frac{2\,(1-x)^4\,(35-30\,x^2+3\,x^4)}{(-1+x)^4},$$
$$-4\left(63-70\,x^2+15\,x^4\right)-\frac{4\,(1-x)^5\,(63-70\,x^2+15\,x^4)}{(-1+x)^5},$$
$$-4\left(231-315\,x^2+105\,x^4-5\,x^6\right)-\frac{4\,(1-x)^6\,(-231+315\,x^2-105\,x^4+5\,x^6)}{(-1+x)^6},$$
$$-8\left(429-693\,x^2+315\,x^4-35\,x^6\right)+\frac{8\,(1-x)^7\,(-429+693\,x^2-315\,x^4+35\,x^6)}{(-1+x)^7},$$
$$-2\left(6435-12\,012\,x^2+6930\,x^4-1260\,x^6+35\,x^8\right)+\frac{2\,(1-x)^8\,(6435-12\,012\,x^2+6930\,x^4-1260\,x^6+35\,x^8)}{(-1+x)^8},$$
$$-4\left(12\,155-25\,740\,x^2+18\,018\,x^4-4620\,x^6+315\,x^8\right)-\frac{4\,(1-x)^9\,(12\,155-25\,740\,x^2+18\,018\,x^4-4620\,x^6+315\,x^8)}{(-1+x)^9},$$
$$-4\left(46\,189-109\,395\,x^2+90\,090\,x^4-30\,030\,x^6+3465\,x^8-63\,x^{10}\right)-\frac{4\,(1-x)^{10}\,(-46\,189+109\,395\,x^2-90\,090\,x^4+30\,030\,x^6-3465\,x^8+63\,x^{10})}{(-1+x)^{10}}\Big\}$$

$$\{0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0\}$$

Dieselbe Simplifikationsfrage an **FullSImplify** mit einer logischen Restriktion als zweitem Parameter kann *Mathematica* dagegen nicht vereinfachen, selbst wenn die Restriktion mit **Reduce** in eine explizite endliche Alternative umgewandelt wird.

```
FullSimplify[res - (2 x)^n LegendreP[n, 1 / x], n ∈ Integers && 0 < n < 5]
```

$$(1-x)^n\ \text{Hypergeometric2F1}\left[-n,\,-n,\,1,\,\frac{1+x}{1-x}\right]-2^n\,x^n\ \text{LegendreP}\left[n,\,\frac{1}{x}\right]$$

```
cc = Reduce[n ∈ Integers && 0 < n < 5]
FullSimplify[res - (2 x)^n LegendreP[n, 1 / x], cc]
```

$$n == 1\ ||\ n == 2\ ||\ n == 3\ ||\ n == 4$$

$$(1-x)^n\ \text{Hypergeometric2F1}\left[-n,\,-n,\,1,\,\frac{1+x}{1-x}\right]-2^n\,x^n\ \text{LegendreP}\left[n,\,\frac{1}{x}\right]$$

Der Zusammenhang zur ursprünglichen Aufgabe.

```
res /. x → 0
```

```
Hypergeometric2F1[-n, -n, 1, 1]
```

```
htest = (Binomial[2 n, n] - res /. x → 0) // FullSimplify
```

```
Binomial[2 n, n] - Hypergeometric2F1[-n, -n, 1, 1]
```

```
Table[htest, {n, 1, 10}]
```

```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

```
ltest = (Limit[(2 x)^n LegendreP[n, 1 / x], x -> 0] - (res /. x -> 0)) // FullSimplify
```

$$-\text{Hypergeometric2F1}[-n,\,-n,\,1,\,1]+\text{Limit}\left[2^{n}\,x^{n}\,\text{LegendreP}\left[n,\,\frac{1}{x}\right],\,x\to 0\right]$$

Für konkrete Werte von n stehen auf beiden Seiten konkrete Zahlen, die *Mathematica* natürlich automatisch berechnet.

```
Table[htest, {n, 1, 10}]
```

```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

```
Table[ltest, {n, 1, 10}]
```

```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Einige weitere Beispiele

$$t[k\_] := \frac{k\,k!}{n^{k}}\,\text{Binomial}[n,\,k]$$

```
Sum[t[k], {k, 1, n}]
```

```
n
```

$$s[k\_] := -\frac{n\,t[k]}{k}$$

```
s[k + 1] - s[k] == t[k] // FullSimplify
```

```
True
```

$$t[k\_] := (-1)^{k}\,\frac{\text{Binomial}[4\,n,\,2\,k]}{\text{Binomial}[2\,n,\,k]}$$

```
Sum[t[k], {k, 0, 2 n}]
```

$$\sum_{k=0}^{2\,n}\frac{(-1)^{k}\,\text{Binomial}[4\,n,\,2\,k]}{\text{Binomial}[2\,n,\,k]}$$

$$s[k\_] := \frac{(2\,k-1)\,t[k]}{2\,(1-2\,n)}$$

```
s[k + 1] - s[k] == t[k] // FullSimplify
```

```
True
```

Unendliche Summen und Integrale

```
Sum[1/n^k, {k, 1, ∞}]
```

$$\frac{1}{-1+n}$$

```
Sum[1/n^k, {n, 1, ∞}]
```

Zeta[k]

```
Table[Zeta[k], {k, 2, 10}]
```

$$\left\{\frac{\pi^2}{6}, \text{Zeta}[3], \frac{\pi^4}{90}, \text{Zeta}[5], \frac{\pi^6}{945}, \text{Zeta}[7], \frac{\pi^8}{9450}, \text{Zeta}[9], \frac{\pi^{10}}{93\,555}\right\}$$

```
% // N
```

{1.64493, 1.20206, 1.08232, 1.03693, 1.01734, 1.00835, 1.00408, 1.00201, 1.00099}

Summation rationaler Ausdrücke

```
res = Sum[1/(n^4 + 1), {n, 1, ∞}]
```

$$-\frac{1}{4}\,\text{RootSum}\left[2 + 4\,\#1 + 6\,\#1^2 + 4\,\#1^3 + \#1^4\,\&,\ \frac{\text{PolyGamma}[0, -\#1]}{1 + 3\,\#1 + 3\,\#1^2 + \#1^3}\,\&\right]$$

Etwas Theorie

Integration rationaler Funktionen.

```
ClearAll["Global`*"]
g1 = (x + 1) (x + 2);
g2 = x^3 - 2 x + 1;
g3 = x^3 + x + 1;
```

$$\int \frac{1}{g1}\, dx$$

Log[1 + x] − Log[2 + x]

```
Factor[g2]
```

$$(-1 + x)\,(-1 + x + x^2)$$

$$\int \frac{1}{g2}\, dx$$

$$\frac{1}{10}\left(-5-3\sqrt{5}\right)\text{Log}\left[-1+\sqrt{5}-2x\right]+\text{Log}[-1+x]+\frac{1}{10}\left(-5+3\sqrt{5}\right)\text{Log}\left[1+\sqrt{5}+2x\right]$$

sol = Solve[g2 == 0, x]

$$\left\{\{x\to 1\},\left\{x\to\frac{1}{2}\left(-1-\sqrt{5}\right)\right\},\left\{x\to\frac{1}{2}\left(-1+\sqrt{5}\right)\right\}\right\}$$

$$\textbf{cc} = \frac{1}{3x^2-2}\ \text{/. sol // Simplify}$$

$$\left\{1,\frac{2}{5+3\sqrt{5}},\frac{2}{5-3\sqrt{5}}\right\}$$

$$\frac{1}{10}\left(-5+3\sqrt{5}\right)-\frac{2}{5+3\sqrt{5}}\ \text{// Together}$$

0

$$\textbf{pts} = \left\{x,\frac{1}{3x^2-2}\right\}\ \text{/. sol // Together}$$

$$\left\{\{1,1\},\left\{\frac{1}{2}\left(-1-\sqrt{5}\right),\frac{2}{5+3\sqrt{5}}\right\},\left\{\frac{1}{2}\left(-1+\sqrt{5}\right),-\frac{2}{-5+3\sqrt{5}}\right\}\right\}$$

InterpolatingPolynomial[pts, a] // Together
v = % // ExpandDenominator

$$-\frac{16\left(-16+9a+12a^2\right)}{\left(-3+\sqrt{5}\right)\left(3+\sqrt{5}\right)\left(-5+3\sqrt{5}\right)\left(5+3\sqrt{5}\right)}$$

$$\frac{1}{5}\left(-16+9a+12a^2\right)$$

$$\left(v * \left(3a^2-2\right)\ \text{// Expand}\right)\ \text{//. }\left\{a^3\ \text{->}\ 2a-1,\ a^4\ \text{->}\ 2a^2-a\right\}\ \text{// Expand}$$

1

$$\int \frac{1}{g3}\, dx$$

$$\text{RootSum}\left[1+\#1+\#1^3\ \&,\ \frac{\text{Log}[x-\#1]}{1+3\#1^2}\ \&\right]$$

▪ Uneigentliche Integrale rationaler Funktionen

$$\int_0^\infty \frac{1}{g1}\,dx$$

```
Log[2]
```

$$r1 = \int_2^\infty \frac{1}{g2}\,dx$$

$$\frac{1}{10}\left(-10\,\mathrm{Log}[2] + \left(5+3\sqrt{5}\right)\mathrm{Log}\left[5-\sqrt{5}\right] + \left(5-3\sqrt{5}\right)\mathrm{Log}\left[5+\sqrt{5}\right]\right)$$

```
r2 = Plus @@ (#[[2]] Log[x - #[[1]]] & /@ pts)
```

$$\mathrm{Log}[-1+x] - \frac{2\,\mathrm{Log}\left[\frac{1}{2}\left(1-\sqrt{5}\right)+x\right]}{-5+3\sqrt{5}} + \frac{2\,\mathrm{Log}\left[\frac{1}{2}\left(1+\sqrt{5}\right)+x\right]}{5+3\sqrt{5}}$$

```
r1 + (r2 /. x → 2) // Simplify
```

```
0
```

```
Plus @@ cc // Together
```

```
0
```

$$\int_2^\infty \frac{x+1}{g2}\,dx$$

$$-\frac{4\,\mathrm{ArcCoth}\left[\sqrt{5}\right]}{\sqrt{5}} + \mathrm{Log}[5]$$

```
FindRoot[g3 == 0, {x, 0}]
```

```
{x → -0.682328}
```

$$\int_0^\infty \frac{1}{g3}\,dx$$

$$-\mathrm{RootSum}\left[1 + \#1 + \#1^3\,\&,\ \frac{\mathrm{Log}[-\#1]}{1+3\,\#1^2}\,\&\right]$$

▪ Unendliche Summen rationaler Ausdrücke

$$\mathrm{Sum}\left[\frac{1}{g1},\ \{x,\ 1,\ n\}\right]$$

$$\frac{n}{2\,(2+n)}$$

$$\frac{1}{x+1} - \frac{1}{x+2} \text{ // Together}$$

$$\frac{1}{(1+x)\,(2+x)}$$

$$\frac{1}{2} - \frac{1}{n+2} \text{ // Together}$$

$$\frac{n}{2\,(2+n)}$$

$$\text{Sum}\left[\frac{1}{g1}, \{x, 1, n\}\right]$$

$$\frac{n}{2\,(2+n)}$$

$$\text{Sum}\left[\frac{1}{g1}, \{x, 1, \infty\}\right]$$

$$\frac{1}{2}$$

$$\text{s1} = \text{Sum}\left[\frac{1}{g2}, \{x, 2, \infty\}\right]$$

$$-\text{RootSum}\left[5 + 10\,\#1 + 6\,\#1^2 + \#1^3\,\&,\ \frac{\text{PolyGamma}[0, -\#1]}{10 + 12\,\#1 + 3\,\#1^2}\,\&\right]$$

% // FullSimplify

$$-\text{RootSum}\left[5 + 10\,\#1 + 6\,\#1^2 + \#1^3\,\&,\ \frac{\text{PolyGamma}[0, -\#1]}{10 + 12\,\#1 + 3\,\#1^2}\,\&\right]$$

$$\text{s2} = \text{Sum}\left[\frac{1}{g2}, \{x, 2, n\}\right]$$

$$\frac{1}{10}\left(-\left(-5+\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{5}{2} - \frac{\sqrt{5}}{2}\right] + \left(5+\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{1}{2}\left(5+\sqrt{5}\right)\right] + 10\,\text{PolyGamma}[0, n] - \left(5+3\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{3}{2} - \frac{\sqrt{5}}{2} + n\right] + \left(-5+3\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{3}{2} + \frac{\sqrt{5}}{2} + n\right]\right) - \frac{16\left(5\,\text{EulerGamma} - 2\sqrt{5}\,\pi\,\text{Tan}\left[\frac{\sqrt{5}\,\pi}{2}\right]\right)}{5\left(-3+\sqrt{5}\right)\left(-1+\sqrt{5}\right)\left(1+\sqrt{5}\right)\left(3+\sqrt{5}\right)}$$

s3 = Limit[s2, n → ∞] // FullSimplify

$$\frac{1}{10}\left(10\,\text{EulerGamma} - \left(-5+\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{1}{2}\left(5-\sqrt{5}\right)\right] + \left(5+\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{1}{2}\left(5+\sqrt{5}\right)\right] - 4\sqrt{5}\,\pi\,\text{Tan}\left[\frac{\sqrt{5}\,\pi}{2}\right]\right)$$

s1 - s3 // FullSimplify

0

s4 = Sum$\left[\frac{1}{g2}, \{x, 2, 9\}\right]$ + Sum$\left[\frac{1}{g2}, \{x, 10, \infty\}\right]$

$$\frac{124\,404\,198\,307}{439\,677\,789\,320} - \text{RootSum}\left[981 + 298\,\#1 + 30\,\#1^2 + \#1^3\,\&, \frac{\text{PolyGamma}[0, -\#1]}{298 + 60\,\#1 + 3\,\#1^2}\,\&\right]$$

s1 - s4 // Normal // FullSimplify

0

s3 - s4 // Normal // FullSimplify

0

s1 // Normal // Simplify

$$\frac{1}{10}\left(10\,\text{EulerGamma} + \left(5+3\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{5}{2}-\frac{\sqrt{5}}{2}\right] + \left(5-3\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{1}{2}\left(5+\sqrt{5}\right)\right]\right)$$

s4 // Normal // Simplify

$$-\frac{19\,117\,450\,392}{7\,851\,389\,095} + \text{EulerGamma} + \frac{1}{10}\left(5+3\sqrt{5}\right)\text{PolyGamma}\left[0, \frac{1}{2}\left(21-\sqrt{5}\right)\right] + \frac{1}{2}\,\text{PolyGamma}\left[0, \frac{1}{2}\left(21+\sqrt{5}\right)\right] - \frac{3\,\text{PolyGamma}\left[0, \frac{1}{2}\left(21+\sqrt{5}\right)\right]}{2\sqrt{5}}$$

s1 = Sum$\left[\frac{1}{g3}, \{x, 0, \infty\}\right]$

$$-\text{RootSum}\left[1 + \#1 + \#1^3\,\&, \frac{\text{PolyGamma}[0, -\#1]}{1 + 3\,\#1^2}\,\&\right]$$

s2 = Sum[1/g3, {x, 0, n}]

$$-\text{RootSum}\left[1+\#1+\#1^3\,\&,\ \frac{\text{PolyGamma}[0,-\#1]}{1+3\,\#1^2}\,\&\right]+$$
$$\text{RootSum}\left[3+4\,n+3\,n^2+n^3+4\,\#1+6\,n\,\#1+3\,n^2\,\#1+3\,\#1^2+3\,n\,\#1^2+\#1^3\,\&,\ \frac{\text{PolyGamma}[0,-\#1]}{4+6\,n+3\,n^2+6\,\#1+6\,n\,\#1+3\,\#1^2}\,\&\right]$$

s3 = Sum[1/g3, {x, 0, 2}] + Sum[1/g3, {x, 3, ∞}]

$$\frac{47}{33}-\text{RootSum}\left[31+28\,\#1+9\,\#1^2+\#1^3\,\&,\ \frac{\text{PolyGamma}[0,-\#1]}{28+18\,\#1+3\,\#1^2}\,\&\right]$$

Die folgenden beiden Rechnungen schafft *Mathematica* nicht.

s4 = Limit[s2, n → ∞] // FullSimplify

s1 − s4 // Normal // FullSimplify

s1 − s3 // Normal // FullSimplify

Noch einmal unser Ausgangsbeispiel

res = Sum[1/(n^4 + 1), {n, 1, ∞}]

$$-\frac{1}{4}\,\text{RootSum}\left[2+4\,\#1+6\,\#1^2+4\,\#1^3+\#1^4\,\&,\ \frac{\text{PolyGamma}[0,-\#1]}{1+3\,\#1+3\,\#1^2+\#1^3}\,\&\right]$$

res1 = res // Normal // FullSimplify

$$\frac{1}{4}\left(-2+(-1)^{1/4}\,\pi\left(\text{Cot}\left[(-1)^{1/4}\,\pi\right]+\text{Coth}\left[\frac{(1+i)\,\pi}{\sqrt{2}}\right]\right)\right)$$

res2 = res1 // TrigExpand // Simplify

$$-\frac{2\left(\text{Cos}\left[\sqrt{2}\,\pi\right]-\text{Cosh}\left[\sqrt{2}\,\pi\right]\right)+\sqrt{2}\,\pi\left(\text{Sin}\left[\sqrt{2}\,\pi\right]+\text{Sinh}\left[\sqrt{2}\,\pi\right]\right)}{4\left(\text{Cos}\left[\sqrt{2}\,\pi\right]-\text{Cosh}\left[\sqrt{2}\,\pi\right]\right)}$$

res1 // N

$0.578478-6.17995\times10^{-17}\,i$

res2 // N

0.578478

```
NSum[1/(n^4 + 1), {n, 1, ∞}]
```

0.578478

Die PolyGamma−Funktion

```
PolyGamma[x + 1] - PolyGamma[x] // FullSimplify
```

$\frac{1}{x}$

```
Sum[1/k, {k, 1, n - 1}] - PolyGamma[n] // FullSimplify
```

EulerGamma

```
% // N
```

0.577216