Umformung und Vereinfachung mathematischer Ausdrücke

Die Simplifikationsproblematik

```
ClearAll["Global`*"]
```

Was Sie erwarten können

```
Simplify[576 + 816 x + 460 x^2 + 129 x^3 + 18 x^4 + x^5]
```

$(3+x)^2\,(4+x)^3$

```
% - 1 // Simplify
```

$-1+(3+x)^2\,(4+x)^3$

```
% // Expand // Simplify
```

$575+816\,x+460\,x^2+129\,x^3+18\,x^4+x^5$

```
% // FullSimplify
```

$575+x\,(816+x\,(460+x\,(129+x\,(18+x))))$

Vereinfachen mit Simplify

$$D\left[\frac{(1+x^2)}{\sqrt{1-x^2}},\ \{x,\ 4\}\right]$$

$$(1+x^2)\left(\frac{105\,x^4}{(1-x^2)^{9/2}}+\frac{90\,x^2}{(1-x^2)^{7/2}}+\frac{9}{(1-x^2)^{5/2}}\right)+8\,x\left(\frac{15\,x^3}{(1-x^2)^{7/2}}+\frac{9\,x}{(1-x^2)^{5/2}}\right)+12\left(\frac{3\,x^2}{(1-x^2)^{5/2}}+\frac{1}{(1-x^2)^{3/2}}\right)$$

```
% // Together
```

$$\frac{3\,(7+51\,x^2+12\,x^4)}{\sqrt{1-x^2}\,(-1+x^2)^4}$$

```
% // Simplify
```

$$\frac{3\,(7+51\,x^2+12\,x^4)}{(1-x^2)^{9/2}}$$

```
Simplify[Sin[x]^2 + Cos[x]^2]
```

1

```
Simplify[Sin[a^2 - b^2]^2 + Cos[(a - b) (a + b)]^2]
```

1

```
Simplify[(a - b)^2 - (a + b)^2]
```

$-4\,a\,b$

Gezielte Umformung von Ausdrücken

$$\int \mathrm{Log}\left[x^2 - 1\right] \, dx$$

$$-2\,x - \mathrm{Log}[-1 + x] + \mathrm{Log}[1 + x] + x\,\mathrm{Log}\left[-1 + x^2\right]$$

```
s1 = % // Simplify
```

$$-\mathrm{Log}[-1 + x] + \mathrm{Log}[1 + x] + x\left(-2 + \mathrm{Log}\left[-1 + x^2\right]\right)$$

$$\int (\mathrm{Log}[x - 1] + \mathrm{Log}[x + 1]) \, dx$$

$$-2\,x - \mathrm{Log}[-1 + x] + x\,\mathrm{Log}[-1 + x] + \mathrm{Log}[1 + x] + x\,\mathrm{Log}[1 + x]$$

```
s2 = % // Simplify
```

$$-2\,x + (-1 + x)\,\mathrm{Log}[-1 + x] + (1 + x)\,\mathrm{Log}[1 + x]$$

```
FullSimplify[s1 == s2]
```

$$x\left(\mathrm{Log}[-1 + x] + \mathrm{Log}[1 + x] - \mathrm{Log}\left[-1 + x^2\right]\right) == 0$$

```
Solve[Log[x^2 - 1] == 0, x]
```

$$\left\{\left\{x \to -\sqrt{2}\right\}, \left\{x \to \sqrt{2}\right\}\right\}$$

```
Solve[Log[x - 1] + Log[x + 1] == 0, x]
```

$$\left\{\left\{x \to \sqrt{2}\right\}\right\}$$

```
Simplify[s1 == s2, x > 0]
```

True

Vereinfachen und mathematische Exaktheit. Zusatzannahmen

```
{Sqrt[x]^2, Exp[3 Log[x]], Tan[ArcSin[x]]}
```

$\left\{x, x^3, \frac{x}{\sqrt{1-x^2}}\right\}$

```
{ArcTan[Tan[x]], Log[Exp[x]], Sqrt[x^2]} // FullSimplify
```

$\left\{\text{ArcTan}[\text{Tan}[x]], \text{Log}[e^x], \sqrt{x^2}\right\}$

```
f = ArcTan[Tan[x]];
Plot[f, {x, -5, 5}, PlotRange → {-2, 2}]
```




```
f /. x → 12 π/5
```

$\frac{2\pi}{5}$

```
Simplify[f, -1 < x < 1]
```

x

```
f = Log[Exp[x]];
Plot[f, {x, -50, 50}]
```




```
f /. x → 10 I
```

$10\,i - 4\,i\,\pi$

```
Simplify[f, x ∈ Reals]
```

x

f = Sqrt$[x^2]$

$\sqrt{x^2}$

```
Simplify[f, x ∈ Reals]
```

Abs[x]

```
Simplify[f, x ≥ 0]
```

x

w1 = $\sqrt{x}\,\sqrt{y} - \sqrt{x\,y}$ **// Simplify**

$\sqrt{x}\,\sqrt{y} - \sqrt{x\,y}$

```
w1 /. {{x → 1, y → 1}, {x → -1, y → -1}}
```

{0, -2}

```
Simplify[w1, x > 0]
```

0

$$\texttt{w2} = \frac{1}{\sqrt{x}} - \sqrt{\frac{1}{x}} \;\texttt{// Simplify}$$

$$-\sqrt{\frac{1}{x}} + \frac{1}{\sqrt{x}}$$

```
w2 /. {{x → 1}, {x → -1}}
```

```
{0, -2 ⅈ}
```

```
w2 /. x → -1 + 0.001 I // Chop
```

```
0
```

```
Limit[w2 /. {x → -1 + a I}, a → 0]
```

```
0
```

$$\texttt{l} = \text{Log}[-1+x] + \text{Log}[1+x] - \text{Log}[-1+x^2]$$

$$\text{Log}[-1+x] + \text{Log}[1+x] - \text{Log}[-1+x^2]$$

```
l /. x → -2
```

$$2\,i\,\pi$$

```
Simplify[l, x > -1]
```

```
0
```

Eigene Umformungsregeln vereinbaren

$$\texttt{sqrtRules} = \left\{\sqrt{x\_}\,\sqrt{y\_} \to \sqrt{x\,y},\; \frac{1}{\sqrt{y\_}} \to \sqrt{\frac{1}{y}}\right\}$$

$$\left\{\sqrt{x\_}\,\sqrt{y\_} \to \sqrt{x\,y},\; \frac{1}{\sqrt{y\_}} \to \sqrt{\frac{1}{y}}\right\}$$

$$\left\{\texttt{w1},\ \texttt{w2},\ \frac{\sqrt{x}}{\sqrt{y}} - \sqrt{\frac{x}{y}}\right\} \;\texttt{//. sqrtRules}$$

```
{0, 0, 0}
```

$$\left\{\texttt{w1},\ \texttt{w2},\ \frac{\sqrt{x}}{\sqrt{y}} - \sqrt{\frac{x}{y}}\right\} \;\texttt{// PowerExpand}$$

```
{0, 0, 0}
```

Polynomiale, rationale und pseudorationale Ausdrücke

```
ClearAll["Global`*"]
```

Ausmultiplizieren polynomialer Ausdrücke

```
u = x (x - a)^2 (x - b)^3
```

$x\,(-a+x)^2\,(-b+x)^3$

```
Expand[u]
```

$-a^2 b^3 x + 3 a^2 b^2 x^2 + 2 a b^3 x^2 - 3 a^2 b x^3 - 6 a b^2 x^3 - b^3 x^3 + a^2 x^4 + 6 a b x^4 + 3 b^2 x^4 - 2 a x^5 - 3 b x^5 + x^6$

```
Expand[(x + a) (x + b) (a + b)^2, x]
```

$a\,b\,(a+b)^2 + a\,(a+b)^2\,x + b\,(a+b)^2\,x + (a+b)^2\,x^2$

```
Expand[(x + a) (x + b) (a + b)^2, a + b]
```

$(a^2 + 2\,a\,b + b^2)\,(a+x)\,(b+x)$

```
Expand[(Sin[x] + Cos[x])^3]
```

$\mathrm{Cos}[x]^3 + 3\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x] + 3\,\mathrm{Cos}[x]\,\mathrm{Sin}[x]^2 + \mathrm{Sin}[x]^3$

```
ExpandNumerator[(x + 1)^3/(x - 1)^4]
```

$$\frac{1 + 3\,x + 3\,x^2 + x^3}{(-1+x)^4}$$

```
ExpandDenominator[(x + 1)^3/(x - 1)^4] Mfx■
```

$$\frac{\mathrm{Mfx}\blacksquare\;(1+x)^3}{1 - 4\,x + 6\,x^2 - 4\,x^3 + x^4}$$

```
Expand[(Sin[(x + 1)^2] + 1)^2]
```

$1 + 2\,\mathrm{Sin}[(1+x)^2] + \mathrm{Sin}[(1+x)^2]^2$

```
ExpandAll[(Sin[(x + 1)^2] + 1)^2]
```

$1 + 2\,\mathrm{Sin}[1 + 2\,x + x^2] + \mathrm{Sin}[1 + 2\,x + x^2]^2$

Ausdrücke ordnen und nach Variablen zusammenfassen

```
f = Expand[(a + x^2 y + b y)^2 + (x - 2 y + c y^2)^2 + x^3 + x^4 y^2]
```

$a^2 + x^2 + x^3 + 2\,a\,b\,y - 4\,x\,y + 2\,a\,x^2\,y + 4\,y^2 + b^2\,y^2 + 2\,c\,x\,y^2 + 2\,b\,x^2\,y^2 + 2\,x^4\,y^2 - 4\,c\,y^3 + c^2\,y^4$

Das Polynom wird als Polynom in x mit polynomialen Koeffizienten umsortiert, was eine gewisse Klammersetzung induziert.
Das macht sich leider in der Ausgabe nur teilweise bemerkbar, weil die Summanden einer Summe (als Auswirkung des Attributs **Orderless** von **Plus**) defaultmäßig nach einer inneren Ordnung sortiert werden, also nicht nach x−Potenzen.

```
u = Collect[f, x]
```

$a^2 + x^3 + 2\,a\,b\,y + 4\,y^2 + b^2\,y^2 + 2\,x^4\,y^2 - 4\,c\,y^3 + c^2\,y^4 + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + x\,(-4\,y + 2\,c\,y^2)$

```
Depth /@ List @@ u
```

{2, 2, 2, 3, 3, 3, 3, 3, 5, 5}

```
Depth /@ u
```

31

Selbst Sortieren hilft nichts, weil *Mathematica* die "falsche" Ordnung sofort erkennt und durch die seiner Meinung nach korrekte ersetzt.

```
Sort[List @@ u, Exponent[#1, x] > Exponent[#2, x] &]
```

$\{2\,x^4\,y^2,\ x^3,\ x^2\,(1 + 2\,a\,y + 2\,b\,y^2),\ x\,(-4\,y + 2\,c\,y^2),\ c^2\,y^4,\ -4\,c\,y^3,\ b^2\,y^2,\ 4\,y^2,\ 2\,a\,b\,y,\ a^2\}$

```
Sort[u, Exponent[#1, x] > Exponent[#2, x] &]
```

$a^2 + x^3 + 2\,a\,b\,y + 4\,y^2 + b^2\,y^2 + 2\,x^4\,y^2 - 4\,c\,y^3 + c^2\,y^4 + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + x\,(-4\,y + 2\,c\,y^2)$

Das Ganze ändert sich erst, wenn das Attribut **Orderless** zurückgesetzt wird. Nun ist die Ausgabe sowohl von **Collect** als auch von **Sort** nach x−Potenzen geordnet.

```
ClearAttributes[Plus, Orderless];
u = Collect[f, x]
Depth /@ List @@ u
Sort[u, Exponent[#1, x] > Exponent[#2, x] &]
SetAttributes[Plus, Orderless];
```

$a^2 + 2\,a\,b\,y + 4\,y^2 + b^2\,y^2 - 4\,c\,y^3 + c^2\,y^4 + x\,(-4\,y + 2\,c\,y^2) + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + x^3 + 2\,x^4\,y^2$

{2, 2, 3, 3, 3, 3, 5, 5, 2, 3}

$2\,x^4\,y^2 + x^3 + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + x\,(-4\,y + 2\,c\,y^2) + c^2\,y^4 - 4\,c\,y^3 + b^2\,y^2 + 4\,y^2 + 2\,a\,b\,y + a^2$

```
Exponent[f, x]
```

4

```
CoefficientList[f, x]
```

$\{a^2 + 2\,a\,b\,y + 4\,y^2 + b^2\,y^2 - 4\,c\,y^3 + c^2\,y^4,\ -4\,y + 2\,c\,y^2,\ 1 + 2\,a\,y + 2\,b\,y^2,\ 1,\ 2\,y^2\}$

```
Coefficient[f, x^2]
```

$1 + 2\,a\,y + 2\,b\,y^2$

```
Coefficient[f, x, 1]
```

$-4\,y + 2\,c\,y^2$

```
Coefficient[f, y, {0, 1, 2}]
```

$\{a^2 + x^2 + x^3,\ 2\,a\,b - 4\,x + 2\,a\,x^2,\ 4 + b^2 + 2\,c\,x + 2\,b\,x^2 + 2\,x^4\}$

```
Collect[f, {y, a}]
```

$a^2 + x^2 + x^3 + (-4\,x + a\,(2\,b + 2\,x^2))\,y + (4 + b^2 + 2\,c\,x + 2\,b\,x^2 + 2\,x^4)\,y^2 - 4\,c\,y^3 + c^2\,y^4$

```
CoefficientList[f, {x, y}]
```

$\{\{a^2,\ 2\,a\,b,\ 4 + b^2,\ -4\,c,\ c^2\},\ \{0,\ -4,\ 2\,c,\ 0,\ 0\},$
$\{1,\ 2\,a,\ 2\,b,\ 0,\ 0\},\ \{1,\ 0,\ 0,\ 0,\ 0\},\ \{0,\ 0,\ 2,\ 0,\ 0\}\}$

```
Collect[f, x, Simplify]
```

$a^2 + x^3 + 2\,a\,b\,y + 2\,x^4\,y^2 + 2\,x\,y\,(-2 + c\,y) + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + y^2\,(b^2 + (-2 + c\,y)^2)$

Collect ist ein komplexes Kommando, welches auf **Coefficient** aufbaut. So hätte man das vorige Kommando auch so anschreiben können:

```
Plus @@ Table[Simplify[Coefficient[f, x, i]] x^i, {i, 0, Exponent[f, x]}]
```

$a^2 + x^3 + 2\,a\,b\,y + 2\,x^4\,y^2 + 2\,x\,y\,(-2 + c\,y) + x^2\,(1 + 2\,a\,y + 2\,b\,y^2) + y^2\,(b^2 + (-2 + c\,y)^2)$

Und dies ist eine Alternative zu **Collect**[f,{y,a}]

```
Collect[f, y, Collect[#, a] &]
```

$a^2 + x^2 + x^3 + (-4\,x + a\,(2\,b + 2\,x^2))\,y + (4 + b^2 + 2\,c\,x + 2\,b\,x^2 + 2\,x^4)\,y^2 - 4\,c\,y^3 + c^2\,y^4$

Collect kann natürlich auch mit verallgemeinerten Kernen aufgreufen werden.

```
u = Expand[(Sin[x] - Cos[x]^2)^2 + (1 - Cos[x])^2]
```

$1 - 2\,\mathrm{Cos}[x] + \mathrm{Cos}[x]^2 + \mathrm{Cos}[x]^4 - 2\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x] + \mathrm{Sin}[x]^2$

```
Collect[u, Cos[x]]
```

$1 - 2\,\mathrm{Cos}[x] + \mathrm{Cos}[x]^4 + \mathrm{Cos}[x]^2\,(1 - 2\,\mathrm{Sin}[x]) + \mathrm{Sin}[x]^2$

Zerlegen in Faktoren

Factor[1 - 4 x + 6 x^2 - 4 x^3 + x^4]

$(-1+x)^4$

Factor[x^4 - y^4]

$(x-y)(x+y)(x^2+y^2)$

s1 = Integrate[Sin[x]^3 + Cos[x]^3, x]

$-\frac{3\,\mathrm{Cos}[x]}{4}+\frac{1}{12}\mathrm{Cos}[3\,x]+\frac{3\,\mathrm{Sin}[x]}{4}+\frac{1}{12}\mathrm{Sin}[3\,x]$

Factor funktioniert auch für pseudorationale Ausdrücke in verallgemeinerten Kernen. Mit vier Kernen ist keine (wesentliche) Faktorisierung möglich.

s1 // Factor

$\frac{1}{12}(-9\,\mathrm{Cos}[x]+\mathrm{Cos}[3\,x]+9\,\mathrm{Sin}[x]+\mathrm{Sin}[3\,x])$

Reduziert man die Zahl der Kerne aber durch **TrigExpand** auf Sin[x] und Cos[x], so lässt sich der Ausdruck in zwei Faktoren zerlegen.

s2 = s1 // TrigExpand

$-\frac{3\,\mathrm{Cos}[x]}{4}+\frac{\mathrm{Cos}[x]^3}{12}+\frac{3\,\mathrm{Sin}[x]}{4}+\frac{1}{4}\mathrm{Cos}[x]^2\,\mathrm{Sin}[x]-\frac{1}{4}\mathrm{Cos}[x]\,\mathrm{Sin}[x]^2-\frac{\mathrm{Sin}[x]^3}{12}$

s3 = s2 // Factor

$\frac{1}{12}(\mathrm{Cos}[x]-\mathrm{Sin}[x])\left(-9+\mathrm{Cos}[x]^2+4\,\mathrm{Cos}[x]\,\mathrm{Sin}[x]+\mathrm{Sin}[x]^2\right)$

s3 // Simplify

$\frac{1}{6}(\mathrm{Cos}[x]-\mathrm{Sin}[x])(-4+\mathrm{Sin}[2\,x])$

Factor$\left[\frac{x^2+x-6}{x^2-1}\right]$

$\frac{(-2+x)(3+x)}{(-1+x)(1+x)}$

Factor$\left[x^2-\frac{1}{4}\right]$

$\frac{1}{4}(-1+2\,x)(1+2\,x)$

```
Factor[1 + x^2]
```

$1 + x^2$

```
Factor[1 + x^2, GaussianIntegers → True]
```

$(-i + x)\ (i + x)$

```
Factor[x^2 + 3 x + 1, Extension → Sqrt[5]]
```

$-\frac{1}{4}\left(-3 + \sqrt{5} - 2\,x\right)\left(3 + \sqrt{5} + 2\,x\right)$

```
Factor[x^2 + 2 Sqrt[2] x + 2, Extension → Automatic]
```

$\left(\sqrt{2} + x\right)^2$

Arbeiten mit rationalen Funktionen

```
Together[x + y/z]
```

$$\frac{y + x\,z}{z}$$

```
Together[x/(x (x + 1) (x - 1)) + 1/(x + 1) + x^2/(x - 1)]
```

$$\frac{x + x^2 + x^3}{(-1 + x)\ (1 + x)}$$

```
Together[1/(x + 3) + 1/(x - 1) + 1/(x^2 - 1)]
```

$$\frac{5 + 5\,x + 2\,x^2}{(-1 + x)\ (1 + x)\ (3 + x)}$$

```
(1 - x^10)/(1 - x^9) // Together
```

$$\frac{1 + x + x^2 + x^3 + x^4 + x^5 + x^6 + x^7 + x^8 + x^9}{1 + x + x^2 + x^3 + x^4 + x^5 + x^6 + x^7 + x^8}$$

```
f = (x + 1)^3/(x - 5)^2; {Numerator[f], Denominator[f]}
```

$\left\{(1 + x)^3,\ (-5 + x)^2\right\}$

Beide Kommandos sind reine Selektoren und bilden vorher keinen gemeinsamen Nenner.

$$\mathbf{Numerator}\left[\frac{a\,x}{1-x^2}+\frac{b}{x}\right]$$

$$\frac{b}{x}+\frac{a\,x}{1-x^2}$$

$$\mathbf{Numerator}\left[\mathbf{Together}\left[\frac{a\,x}{1-x^2}+\frac{b}{x}\right]\right]$$

$$-b-a\,x^2+b\,x^2$$

$$\mathbf{Cancel}\left[\frac{12-8\,x-x^2+x^3}{-18-3\,x+4\,x^2+x^3}\right]$$

$$\frac{-2+x}{3+x}$$

$$f=\frac{1-x^3}{(x-1)^6}$$

$$\frac{1-x^3}{(-1+x)^6}$$

f // Simplify

$$\frac{1-x^3}{(-1+x)^6}$$

f // Cancel

$$\frac{-1-x-x^2}{(-1+x)^5}$$

$$f=\frac{x^{20}-1}{x^7-1}$$

$$\frac{-1+x^{20}}{-1+x^7}$$

f // Simplify

$$\frac{-1+x^{20}}{-1+x^7}$$

f // Cancel

$$\left(1+x+x^2+x^3+x^4+x^5+x^6+x^7+x^8+x^9+x^{10}+x^{11}+x^{12}+x^{13}+x^{14}+x^{15}+x^{16}+x^{17}+x^{18}+x^{19}\right)/\left(1+x+x^2+x^3+x^4+x^5+x^6\right)$$

$$\mathbf{Apart}\left[\frac{x}{x^4-1}\right]$$

$$\frac{1}{4\,(-1+x)}+\frac{1}{4\,(1+x)}-\frac{x}{2\,(1+x^2)}$$

```
Apart[1/(1 + Sin[x]^3)]
```

$$\frac{1}{3\,(1+\mathrm{Sin}[x])}+\frac{2-\mathrm{Sin}[x]}{3\,\left(1-\mathrm{Sin}[x]+\mathrm{Sin}[x]^2\right)}$$

```
Apart[x y/(x - y)^2]
```

$$\frac{x^2}{(-x+y)^2}+\frac{x}{-x+y}$$

```
Apart[x y/(x - y)^2, x]
```

$$\frac{y}{x-y}+\frac{y^2}{(x-y)^2}$$

```
Apart[x y/(x - y)^2, y]
```

$$\frac{x^2}{(-x+y)^2}+\frac{x}{-x+y}$$

```
Apart[x y/(x - y)^2, {x, y}]
```

$$\left\{\frac{y}{x-y}+\frac{y^2}{(x-y)^2},\ \frac{x^2}{(-x+y)^2}+\frac{x}{-x+y}\right\}$$

```
Apart[(x^5 + 1)/(x^2 + x + 1)]
```

$$1-x^2+x^3-\frac{x}{1+x+x^2}$$

Trigonometrische Umformungen

```
ClearAll["Global`*"]
```

Einführung

```
{Sin[x]/Cos[x], 1/Cos[x], Cos[x]/Sin[x], 1/Sin[x]}
```

```
{Tan[x], Sec[x], Cot[x], Csc[x]}
```

```
u1 = Sin[x] + Cot[x] // TrigToExp
```

$$\frac{1}{2}\,i\,e^{-i\,x}-\frac{1}{2}\,i\,e^{i\,x}-\frac{i\,e^{-i\,x}}{e^{-i\,x}-e^{i\,x}}-\frac{i\,e^{i\,x}}{e^{-i\,x}-e^{i\,x}}$$

Das ist nun noch komplizierter als vorher, weil gleich im ersten Term Grad 4 statt Grad 3 entsteht.

```
u2 = u1 // Together
```

$$-\frac{i\,e^{-i\,x}\left(1-2\,e^{i\,x}-2\,e^{2\,i\,x}-2\,e^{3\,i\,x}+e^{4\,i\,x}\right)}{2\left(-1+e^{2\,i\,x}\right)}$$

```
u3 = u2 // ExpToTrig
```

```
-(i (Cos[x] - i Sin[x]) (1 - 2 Cos[x] - 2 Cos[2 x] - 2 Cos[3 x] + Cos[4 x] - 2 i Sin[x] -
      2 i Sin[2 x] - 2 i Sin[3 x] + i Sin[4 x])) / (2 (-1 + Cos[2 x] + i Sin[2 x]))
```

```
u3 // Simplify
```

$$-\frac{1}{2}\,(-1-2\,\mathrm{Cos}[x]+\mathrm{Cos}[2\,x])\,\mathrm{Csc}[x]$$

```
u3 - Sin[x] // Simplify
```

```
Cot[x]
```

▪ Die Standardkommandos

```
ClearAll["Global`*"]
```

TrigExpand

```
TrigExpand[Cos[α + β]]
```

```
Cos[α] Cos[β] - Sin[α] Sin[β]
```

$$\mathtt{TrigExpand}\left[\mathrm{Cos}\left[2\,x+\frac{\pi}{4}\right]\right]$$

$$\frac{\mathrm{Cos}[x]^2}{\sqrt{2}}-\sqrt{2}\,\mathrm{Cos}[x]\,\mathrm{Sin}[x]-\frac{\mathrm{Sin}[x]^2}{\sqrt{2}}$$

```
TrigExpand[Cos[4 x]]
```

$$\mathrm{Cos}[x]^4-6\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x]^2+\mathrm{Sin}[x]^4$$

$$\mathtt{TrigExpand}\left[\mathrm{Sin}[x]^2+\mathrm{Cot}[x]^2\right]$$

$$-\frac{3}{4}\,\mathrm{Cos}[x]^2+\frac{1}{8}\,\mathrm{Cos}[x]^2\,\mathrm{Cot}[x]^2+\frac{7\,\mathrm{Csc}[x]^2}{8}+\frac{\mathrm{Sin}[x]^2}{8}$$

TrigReduce

```
TrigReduce[2 Cos[α] Cos[β]]
```

$\mathrm{Cos}[\alpha-\beta]+\mathrm{Cos}[\alpha+\beta]$

```
TrigReduce[Sin[x]^5]
```

$\frac{1}{16}\,(10\,\mathrm{Sin}[x]-5\,\mathrm{Sin}[3\,x]+\mathrm{Sin}[5\,x])$

```
TrigReduce[Sin[x]^2 + Cot[x]^2]
```

$\frac{1}{8}\left(7\,\mathrm{Csc}[x]^2+\mathrm{Cos}[4\,x]\,\mathrm{Csc}[x]^2\right)$

TrigFactor

```
TrigFactor[Cos[4 x]]
```

$2\,\mathrm{Sin}\left[\frac{\pi}{4}-2\,x\right]\,\mathrm{Sin}\left[\frac{\pi}{4}+2\,x\right]$

```
TrigFactor[Cos[3 x]]
```

$\mathrm{Cos}[x]\,(-1+2\,\mathrm{Cos}[2\,x])$

$4\,\mathrm{Cos}\left[x+\frac{\pi}{3}\right]\,\mathrm{Cos}\left[x-\frac{\pi}{3}\right]$ `// TrigReduce`

$-1+2\,\mathrm{Cos}[2\,x]$

```
TrigFactor[Cos[{2, 3, 4, 5, 6} x]]
```

$\left\{2\,\mathrm{Sin}\left[\frac{\pi}{4}-x\right]\,\mathrm{Sin}\left[\frac{\pi}{4}+x\right],\ \mathrm{Cos}[x]\,(-1+2\,\mathrm{Cos}[2\,x]),\ 2\,\mathrm{Sin}\left[\frac{\pi}{4}-2\,x\right]\,\mathrm{Sin}\left[\frac{\pi}{4}+2\,x\right],\ \mathrm{Cos}[x]\,(1-2\,\mathrm{Cos}[2\,x]+2\,\mathrm{Cos}[4\,x]),\ -2\,\mathrm{Sin}\left[\frac{\pi}{4}-x\right]\,(-1+2\,\mathrm{Sin}[2\,x])\,(1+2\,\mathrm{Sin}[2\,x])\,\mathrm{Sin}\left[\frac{\pi}{4}+x\right]\right\}$

■ Ein trigonometrisches Regelsystem

```
s1 = Sin[x_ + y_] → Sin[x] Cos[y] + Cos[x] Sin[y]
c1 = Cos[x_ + y_] → Cos[x] Cos[y] - Sin[x] Sin[y]
```

```
Sin[x_ + y_] → Cos[y] Sin[x] + Cos[x] Sin[y]

Cos[x_ + y_] → Cos[x] Cos[y] - Sin[x] Sin[y]
```

```
Sin[x + y + z] //. {s1, c1}
```

```
Cos[x] (Cos[z] Sin[y] + Cos[y] Sin[z]) + Sin[x] (Cos[y] Cos[z] - Sin[y] Sin[z])
```

Die Regeln wurden mehrfach angewendet, aber Klammern nicht aufgelöst. **Expand** erst im Nachhinein anzuwenden ist sinnvoll; besser wäre es aber, den Effekt des Zusammenfassens so früh wie möglich in der Rechnung zu berücksichtigen.
Natürlich ist :→ statt → zu verwenden, denn Expand[a] soll ja zur Zeiit der Regelanwendung und nicht zur Zeit der Regeldefinition ausgeführt werden. Die rechte Seite der Regel darf also während der Definition nicht ausgewertet werden.
Der folgende Ansatz scheitert trotzdem zunächst.

```
d1 = a_ :→ Expand[a]

a_ :→ Expand[a]

Sin[x - y + z] //. {s1, c1, d1}

Cos[y - z] Sin[x] - Cos[x] Sin[y - z]
```

Der Grund ist simpel: **d1** wurde auf Teilausdrücken gar nicht erst versucht, weil die Regel ja auch auf den Gesamtausdruck angewendet werden kann und dort keine Änderung hervorruft. Um "unendliche Schleifen des Nichtstuns" zu vermeiden bricht *Mathematica* in solchen Fällen das **ReplaceRepeated** ab.
Die Regel d1 darf also nur angewendet werden, wenn sich wirklich etwas ändert, was hier mit =!= (**UnsameQ**) geprüft wird.

```
d1 = a_ /; Expand[a] =!= a :> Expand[a]

a_ /; Expand[a] =!= a :→ Expand[a]

Sin[x - y + z] //. {s1, c1, d1}

Cos[y] Cos[z] Sin[x] - Cos[x] Cos[z] Sin[y] + Cos[x] Cos[y] Sin[z] + Sin[x] Sin[y] Sin[z]

Sin[2 x] //. {s1, c1, d1}

Sin[2 x]

s2 = Sin[n_Integer x_] /; n > 1 → Sin[x] Cos[(n - 1) x] + Cos[x] Sin[(n - 1) x]
c2 = Cos[n_Integer x_] /; n > 1 → Cos[x] Cos[(n - 1) x] - Sin[x] Sin[(n - 1) x]

Sin[n_Integer x_] /; n > 1 → Cos[(-1 + n) x] Sin[x] + Cos[x] Sin[(-1 + n) x]

Cos[n_Integer x_] /; n > 1 → Cos[x] Cos[(-1 + n) x] - Sin[x] Sin[(-1 + n) x]

Sin[12 x] //. {s2, c2, d1}
```

$$12\,\mathrm{Cos}[x]^{11}\,\mathrm{Sin}[x] - 220\,\mathrm{Cos}[x]^{9}\,\mathrm{Sin}[x]^{3} + 792\,\mathrm{Cos}[x]^{7}\,\mathrm{Sin}[x]^{5} - 792\,\mathrm{Cos}[x]^{5}\,\mathrm{Sin}[x]^{7} + 220\,\mathrm{Cos}[x]^{3}\,\mathrm{Sin}[x]^{9} - 12\,\mathrm{Cos}[x]\,\mathrm{Sin}[x]^{11}$$

$$\mathbf{sc = Sin[x\_]^{n\_Integer} /; n > 1 \to \left(1 - Cos[x]^2\right) Sin[x]^{n-2}}$$

$$\mathrm{Sin}[x\_]^{n\_Integer} /; n > 1 \to \left(1 - \mathrm{Cos}[x]^2\right) \mathrm{Sin}[x]^{-2+n}$$

```
Sin[5 x] //. {s2, c2, d1, sc}
```

$$\mathrm{Sin}[x] - 12\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x] + 16\,\mathrm{Cos}[x]^4\,\mathrm{Sin}[x]$$

```
Collect[%, Sin[x]]
```

$$\left(1 - 12\,\mathrm{Cos}[x]^2 + 16\,\mathrm{Cos}[x]^4\right) \mathrm{Sin}[x]$$

```
% // Factor
```

$\left(-1-2\,\text{Cos}[x]+4\,\text{Cos}[x]^2\right)\left(-1+2\,\text{Cos}[x]+4\,\text{Cos}[x]^2\right)\text{Sin}[x]$

```
Sin[5 x] // TrigFactor
```

$(1+2\,\text{Cos}[2\,x]+2\,\text{Cos}[4\,x])\,\text{Sin}[x]$

Eine Testserie

Mit dieser Funktion werden die Ergebnisse verschiedene Simplifikationsstrategien für trigonometrische Ausdrücke aufgesammelt.

```
tests[h_] := #[h] & /@ {Simplify, TrigReduce, TrigExpand, TrigFactor}
```

Ein erstes einfaches Beispiel: Ein polynomialer Ausdruck in **Sin** mit Mehrfachwinkeln.

```
d1 = Sin[3 x] Sin[5 x]
i1 = Integrate[d1, x]
h1 = d1 - D[i1, x]
```

$\text{Sin}[3\,x]\,\text{Sin}[5\,x]$

$\frac{1}{4}\text{Sin}[2\,x]-\frac{1}{16}\text{Sin}[8\,x]$

$-\frac{1}{2}\text{Cos}[2\,x]+\frac{1}{2}\text{Cos}[8\,x]+\text{Sin}[3\,x]\,\text{Sin}[5\,x]$

```
h1 // tests
```

$\{0,\ 0,\ 0,\ 0\}$

Hier sehen Sie noch einmal die Wirkung der vier Simplifikationskommandos auf einen trigonometrischen Ausdruck. Das Ergebnis von **TrigFactor** ist wegen der komplexen Zahlen nicht zufriedenstellend.

```
i1 // tests
```

$$\Big\{\frac{1}{16}(4\,\text{Sin}[2\,x]-\text{Sin}[8\,x]),\ \frac{1}{16}(4\,\text{Sin}[2\,x]-\text{Sin}[8\,x]),$$
$$\frac{1}{2}\text{Cos}[x]\,\text{Sin}[x]-\frac{1}{2}\text{Cos}[x]^7\,\text{Sin}[x]+\frac{7}{2}\text{Cos}[x]^5\,\text{Sin}[x]^3-\frac{7}{2}\text{Cos}[x]^3\,\text{Sin}[x]^5+$$
$$\frac{1}{2}\text{Cos}[x]\,\text{Sin}[x]^7,\ -i\,\text{Cos}[x]\,(i+(1+i)\,\text{Cos}[2\,x])\,(1+(1+i)\,\text{Cos}[2\,x])\,\text{Sin}[x]^3\Big\}$$

Im zweiten Beispiel muss der gemeinsame Kern $\frac{x}{6}$ erst gefunden werden.

```
d2 = Cos[x/2] Cos[x/3]
i2 = Integrate[d2, x]
h2 = d2 - D[i2, x]
```

$\mathrm{Cos}\left[\frac{x}{3}\right]\,\mathrm{Cos}\left[\frac{x}{2}\right]$

$3\,\mathrm{Sin}\left[\frac{x}{6}\right]+\frac{3}{5}\,\mathrm{Sin}\left[\frac{5\,x}{6}\right]$

$-\frac{1}{2}\,\mathrm{Cos}\left[\frac{x}{6}\right]+\mathrm{Cos}\left[\frac{x}{3}\right]\,\mathrm{Cos}\left[\frac{x}{2}\right]-\frac{1}{2}\,\mathrm{Cos}\left[\frac{5\,x}{6}\right]$

```
Cos[5 x/6] // FullForm
```

Cos[Times[Rational[5, 6], x]]

```
Cos[5 x/6] // TrigExpand
```

$\mathrm{Cos}\left[\frac{x}{6}\right]^5-10\,\mathrm{Cos}\left[\frac{x}{6}\right]^3\,\mathrm{Sin}\left[\frac{x}{6}\right]^2+5\,\mathrm{Cos}\left[\frac{x}{6}\right]\,\mathrm{Sin}\left[\frac{x}{6}\right]^4$

```
h2 // tests
```

{0, 0, 0, 0}

```
i2 // tests
```

$\left\{\frac{3}{5}\left(5\,\mathrm{Sin}\left[\frac{x}{6}\right]+\mathrm{Sin}\left[\frac{5\,x}{6}\right]\right),\ \frac{3}{5}\left(5\,\mathrm{Sin}\left[\frac{x}{6}\right]+\mathrm{Sin}\left[\frac{5\,x}{6}\right]\right),\right.$
$3\,\mathrm{Sin}\left[\frac{x}{6}\right]+3\,\mathrm{Cos}\left[\frac{x}{6}\right]^4\,\mathrm{Sin}\left[\frac{x}{6}\right]-6\,\mathrm{Cos}\left[\frac{x}{6}\right]^2\,\mathrm{Sin}\left[\frac{x}{6}\right]^3+\frac{3}{5}\,\mathrm{Sin}\left[\frac{x}{6}\right]^5,$
$\left.\frac{6}{5}\left(3+\mathrm{Cos}\left[\frac{x}{3}\right]+\mathrm{Cos}\left[\frac{2\,x}{3}\right]\right)\mathrm{Sin}\left[\frac{x}{6}\right]\right\}$

Das folgende Beispiel bereitete *Mathematica* vor Version 6 große Schwierigkeiten, weil der einfache Weg der Integration nicht gefunden wurde. Relikte dieser "Blindheit" haben noch beim Ergebnis von **TrigFactor** auf diesem Ausdruck überlebt.

```
d3 = Sin[2 x - π/6] Cos[3 x + π/4];
i3 = Integrate[d3, x]
```

$-\frac{1}{10}\,\mathrm{Cos}\left[\frac{\pi}{12}+5\,x\right]+\frac{1}{2}\,\mathrm{Sin}\left[\frac{\pi}{12}-x\right]$

```
d3a = d3 // TrigReduce
Integrate[d3a, x]
```

$$\frac{1}{2}\left(-\mathrm{Cos}\left[\frac{\pi}{12}-x\right]+\mathrm{Sin}\left[\frac{\pi}{12}+5\,x\right]\right)$$

$$-\frac{1}{10}\,\mathrm{Cos}\left[\frac{\pi}{12}+5\,x\right]+\frac{1}{2}\,\mathrm{Sin}\left[\frac{\pi}{12}-x\right]$$

```
d3 - D[i3, x] // tests
```

$$\left\{0,\,0,\,0,\,-\frac{1}{2}\,i\left(i+(-1)^{1/6}\right)\left(-1+(-1)^{1/3}-(-1)^{2/3}\right)\left(2\,i+(-1)^{5/6}\right)\mathrm{Sin}\left[\frac{\pi}{12}-x\right]\right\}$$

```
d3 // TrigFactor
```

$$2\,(-1)^{5/6}\,\mathrm{Cos}\left[\frac{\pi}{12}-x\right]\mathrm{Cos}\left[\frac{\pi}{12}+x\right]\mathrm{Sin}\left[\frac{\pi}{12}-x\right]\left(-(-1)^{1/6}+\left(1+(-1)^{1/3}\right)\mathrm{Cos}[2\,x]+\left(-i+(-1)^{5/6}\right)\mathrm{Sin}[2\,x]\right)$$

Und nun noch einige Beispiele, die nicht im Buch besprochen sind.

$$\mathbf{d4} = \frac{1}{\mathrm{Cos}[x]^4}$$

```
i4 = Integrate[d4, x]
h4 = d4 - D[i4, x]
```

$$\mathrm{Sec}[x]^4$$

$$\frac{2\,\mathrm{Tan}[x]}{3}+\frac{1}{3}\,\mathrm{Sec}[x]^2\,\mathrm{Tan}[x]$$

$$-\frac{2}{3}\,\mathrm{Sec}[x]^2+\frac{2\,\mathrm{Sec}[x]^4}{3}-\frac{2}{3}\,\mathrm{Sec}[x]^2\,\mathrm{Tan}[x]^2$$

```
h4 // tests
```

{0, 0, 0, 0}

Hier die Wirkung der 4 Simplifikationsbefehle, wenn auch andere Winkelfunktionen mit im Spiel sind.

```
i4 // tests
```

$$\left\{\frac{1}{3}\left(2+\mathrm{Sec}[x]^2\right)\mathrm{Tan}[x],\;\frac{1}{6}\left(\mathrm{Sec}[x]^3\,\mathrm{Sin}[3\,x]+3\,\mathrm{Sec}[x]^2\,\mathrm{Tan}[x]\right),\;\frac{\mathrm{Tan}[x]}{2}+\frac{1}{2}\,\mathrm{Sec}[x]^2\,\mathrm{Tan}[x]-\frac{\mathrm{Tan}[x]^3}{6},\;\frac{1}{3}\,(2+\mathrm{Cos}[2\,x])\,\mathrm{Sec}[x]^2\,\mathrm{Tan}[x]\right\}$$

d5 = $\frac{1}{\text{Sin}[x]^2\,(1-\text{Cos}[x])}$
i5 = Integrate[d5, x] // Simplify
h5 = d5 - D[i5, x]

$$\frac{\text{Csc}[x]^2}{1-\text{Cos}[x]}$$

$$\frac{-(-2+\text{Cos}[x])\,\text{Cot}[x]+\text{Sin}[x]}{3\,(-1+\text{Cos}[x])}$$

$$\frac{\text{Csc}[x]^2}{1-\text{Cos}[x]}-\frac{2\,\text{Cos}[x]+(-2+\text{Cos}[x])\,\text{Csc}[x]^2}{3\,(-1+\text{Cos}[x])}-\frac{\text{Sin}[x]\,(-(-2+\text{Cos}[x])\,\text{Cot}[x]+\text{Sin}[x])}{3\,(-1+\text{Cos}[x])^2}$$

h5 // tests

$$\Big\{0,\ -\frac{1}{6\,(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]^2}{1-\text{Cos}[x]}+\frac{2\,\text{Csc}[x]^2}{3\,(-1+\text{Cos}[x])}+\frac{\text{Csc}[x]^2\left(-\frac{\text{Cos}[x]}{6}-\frac{1}{6}\,i\,\text{Sin}[x]\right)}{-1+\text{Cos}[x]}+\frac{\text{Csc}[x]^2\left(-\frac{\text{Cos}[x]}{6}+\frac{1}{6}\,i\,\text{Sin}[x]\right)}{-1+\text{Cos}[x]}+\frac{-\frac{\text{Cos}[x]}{3}-\frac{1}{3}\,i\,\text{Sin}[x]}{-1+\text{Cos}[x]}+\frac{-\frac{\text{Cos}[x]}{3}+\frac{1}{3}\,i\,\text{Sin}[x]}{-1+\text{Cos}[x]}+\frac{\text{Csc}[x]\left(-\frac{1}{24}\,i\,\text{Cos}[x]+\frac{\text{Sin}[x]}{24}\right)}{(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]\left(\frac{1}{24}\,i\,\text{Cos}[x]+\frac{\text{Sin}[x]}{24}\right)}{(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]\left(-\frac{1}{6}\,i\,\text{Cos}[2\,x]-\frac{1}{6}\,\text{Sin}[2\,x]\right)}{(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]\left(\frac{1}{6}\,i\,\text{Cos}[2\,x]-\frac{1}{6}\,\text{Sin}[2\,x]\right)}{(-1+\text{Cos}[x])^2}+\frac{\frac{1}{12}\,\text{Cos}[2\,x]-\frac{1}{12}\,i\,\text{Sin}[2\,x]}{(-1+\text{Cos}[x])^2}+\frac{\frac{1}{12}\,\text{Cos}[2\,x]+\frac{1}{12}\,i\,\text{Sin}[2\,x]}{(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]\left(-\frac{1}{24}\,i\,\text{Cos}[3\,x]+\frac{1}{24}\,\text{Sin}[3\,x]\right)}{(-1+\text{Cos}[x])^2}+\frac{\text{Csc}[x]\left(\frac{1}{24}\,i\,\text{Cos}[3\,x]+\frac{1}{24}\,\text{Sin}[3\,x]\right)}{(-1+\text{Cos}[x])^2},\ -\frac{1}{12\,(-1+\text{Cos}[x])^2}-\frac{2\,\text{Cos}[x]}{3\,(-1+\text{Cos}[x])^2}-\frac{2\,\text{Cos}[x]}{3\,(-1+\text{Cos}[x])}+\frac{5\,\text{Cos}[x]^2}{12\,(-1+\text{Cos}[x])^2}-\frac{\text{Cot}[x]\,\text{Csc}[x]}{3\,(-1+\text{Cos}[x])}+\frac{\text{Csc}[x]^2}{1-\text{Cos}[x]}+\frac{2\,\text{Csc}[x]^2}{3\,(-1+\text{Cos}[x])}-\frac{\text{Sin}[x]^2}{4\,(-1+\text{Cos}[x])^2},\ 0\Big\}$$

i5 // tests

$$\Big\{\frac{-(-2+\text{Cos}[x])\,\text{Cot}[x]+\text{Sin}[x]}{3\,(-1+\text{Cos}[x])},\ \frac{4\,\text{Cot}[x]-\text{Csc}[x]-\text{Cos}[2\,x]\,\text{Csc}[x]+2\,\text{Sin}[x]}{6\,(-1+\text{Cos}[x])},\ \frac{2\,\text{Cot}[x]}{3\,(-1+\text{Cos}[x])}-\frac{\text{Cos}[x]\,\text{Cot}[x]}{6\,(-1+\text{Cos}[x])}-\frac{\text{Csc}[x]}{6\,(-1+\text{Cos}[x])}+\frac{\text{Sin}[x]}{2\,(-1+\text{Cos}[x])},\ \frac{1}{12}\,(-2\,\text{Cos}[x]+\text{Cos}[2\,x])\,\text{Csc}\Big[\frac{x}{2}\Big]^3\,\text{Sec}\Big[\frac{x}{2}\Big]\Big\}$$

```
d6 = 1/(Sin[2 x] (3 Tan[x] + 5))
i6 = Integrate[d6, x] // Simplify
h6 = d6 - D[i6, x]
```

$$\frac{\mathrm{Csc}[2\,x]}{5+3\,\mathrm{Tan}[x]}$$

$$-\frac{1}{5}\,\mathrm{ArcTanh}\left[1+\frac{6\,\mathrm{Tan}[x]}{5}\right]$$

$$\frac{\mathrm{Csc}[2\,x]}{5+3\,\mathrm{Tan}[x]}+\frac{6\,\mathrm{Sec}[x]^2}{25\left(1-\left(1+\frac{6\,\mathrm{Tan}[x]}{5}\right)^2\right)}$$

```
h6 // tests
```

{0, 0, 0, 0}

```
i6 // tests
```

$$\left\{-\frac{1}{5}\,\mathrm{ArcTanh}\left[1+\frac{6\,\mathrm{Tan}[x]}{5}\right],\ -\frac{1}{5}\,\mathrm{ArcTanh}\left[1+\frac{6\,\mathrm{Tan}[x]}{5}\right],\ -\frac{1}{5}\,\mathrm{ArcTanh}\left[1+\frac{6\,\mathrm{Tan}[x]}{5}\right],\ -\frac{1}{5}\,\mathrm{ArcTanh}\left[1+\frac{6\,\mathrm{Tan}[x]}{5}\right]\right\}$$